Vanable Conts

Introduction to

Ada

Raphaël Amiard Gustavo A. Hoffmann



LEARN.

ADACORE.COM





Introduction to Ada

Release 2025-04

Raphaël Amiard and Gustavo A. Hoffmann

CONTENTS:

1	1.1 History	3 3 4 4
2	Imperative language 2.1 Hello world	5 5 6
		-
	 2.4 Imperative language - Case statement	2 4 6
3	2.6.2 Case expressions	7 9
	3.1 Subprograms 1 3.1.1 Subprogram calls 2 3.1.2 Nested subprograms 2 3.1.3 Function calls 2	0 1
	3.2 Parameter modes 2 3.3 Subprogram calls 2 3.3.1 In parameters 2 3.3.2 In out parameters 2	4 4 5
	3.3.3 Out parameters	7
4	Modular programming 3 4.1 Packages 3 4.2 Using a package 3 4.3 Package body 3 4.4 Child packages 3 4.4.1 Child of a child package 3 4.4.2 Multiple children 3 4.4.3 Visibility 3 4.5 Renaming 4	1 3 5 7 8
5	Strongly typed language 4. 5.1 What is a type? 4 5.2 Integers 4	3

6	5.3 5.4 5.5 5.6 5.7 5.8	Unsigned types Enumerations Floating-point types 5.5.1 Basic properties 5.5.2 Precision of floating-point types 5.5.3 Range of floating-point types Strong typing Derived types Subtypes 5.8.1 Subtypes as type aliases	45 46 47 48 48 49 50 52 54 56 58
0	6.1 6.2 6.3 6.4	Record type declaration	61 62 62 63
7		Array type declaration Indexing Simpler array declarations Range attribute Unconstrained arrays Predefined array type: String Restrictions Returning unconstrained arrays Declaring arrays (2) Array slices	
8	More 8.1 8.2 8.3	Aggregates: A primer	86
9	9.1 9.2 9.3 9.4	Overview	93 94 94
10	10.1 10.2	e about records Dynamically sized record types	98
11	11.1	d-point types Decimal fixed-point types	
12	12.2 12.3	Basic encapsulation	110 112
13	Gen	erics 1	19

	13.1 Introduction 13.2 Formal type declaration 13.3 Formal object declaration 13.4 Generic body definition 13.5 Generic instantiation 13.6 Generic packages 13.7 Formal subprograms 13.8 Example: I/O instances 13.9 Example: ADTs 13.1Œxample: Swap 13.1Œxample: Reversing 13.1Æxample: Test application	119 120 121 122 124 125 127 129 132
14		L39
	14.2 Raising an exception	139 140
15	Tasking 1 15.1 Tasks	L43 143
	15.1.1 Simple task	
	15.1.2 Simple synchronization	144
	15.1.3 Delay	
	15.1.4 Synchronization: rendezvous	
	15.1.5 Select loop	
	15.2 Protected objects	
	15.2.1 Simple object	
	15.2.2 Entries	
	15.3 Task and protected types	
	15.3.1 Task types	
	15.3.2 Protected types	158
16		L61
	16.1 Pre- and postconditions	
	16.2 Predicates	
	16.3 Type invariants	108
17		L71
	17.1 Multi-language project	
	17.2 Type convention	
	17.3 Foreign subprograms	
	17.3.1 Calling C subprograms in Ada	
	17.3.2 Calling Add subprograms in C	
	17.4.1 Using C global variables in Ada	
	17.4.2 Using Ada variables in C	
	17.5 Generating bindings	
	17.5.1 Adapting bindings	179
1 9	Object-oriented programming	L85
	18.1 Derived types	
	18.2 Tagged types	
	18.3 Classwide types	
	18.4 Dispatching operations	190
	18.5 Dot notation	
	18.6 Private & Limited	
	18.7 Classwide access types	т95

19Standard library: Containers	199
19.1 Vectors	. 199
19.1.1 Instantiation	. 199
19.1.2 Initialization	. 200
19.1.3 Appending and prepending elements	. 200
19.1.4 Accessing first and last elements	
19.1.5 Iterating	
19.1.6 Finding and changing elements	
19.1.7 Inserting elements	
19.1.8 Removing elements	
19.1.9 Other Operations	
19.2 Sets	
19.2.1 Initialization and iteration	
19.2.2 Operations on elements	
19.2.3 Other Operations	
19.3 Indefinite maps	
19.3.1 Hashed maps	
19.3.2 Ordered maps	
19.3.3 Complexity	
19.5.5 Complexity	. 224
20Standard library: Dates & Times	227
20.1 Date and time handling	
20.1.1 Delaying using date	
20.2 Real-time	
20.2.1 Benchmarking	
20.2.1 Benefittationing 1.1.1.1.1.1.1.1.1.1.1.1.1.1.1.1.1.1.1.	. 231
21Standard library: Strings	235
21.1 String operations	. 235
21.2 Limitation of fixed-length strings	
21.3 Bounded strings	
21.4 Unbounded strings	
2211 01120411404 0411190 111111111111111111111111	
22Standard library: Files and streams	245
22.1 Text I/O	. 245
22.2 Sequential I/O	
22.3 Direct I/O	
22.4 Stream I/O	
23Standard library: Numerics	255
23.1 Elementary Functions	. 255
23.2 Random Number Generation	. 256
23.3 Complex Types	. 258
23.4 Vector and Matrix Manipulation	
·	
24Appendices	265
24.1 Appendix A: Generic Formal Types	
24.1.1 Indefinite version	
24.2 Appendix R: Containers	267

Copyright © 2018 - 2024, AdaCore

This book is published under a CC BY-SA license, which means that you can copy, redistribute, remix, transform, and build upon the content for any purpose, even commercially, as long as you give appropriate credit, provide a link to the license, and indicate if changes were made. If you remix, transform, or build upon the material, you must distribute your contributions under the same license as the original. You can find license details on this page¹



This course will teach you the basics of the Ada programming language and is intended for those who already have a basic understanding of programming techniques. You will learn how to apply those techniques to programming in Ada.

This document was written by Raphaël Amiard and Gustavo A. Hoffmann, with review from Richard Kenner.

0 Note

The code examples in this course use a 50-column limit, which greatly improves the readability of the code on devices with a small screen size. This constraint, however, leads to an unusual coding style. For instance, instead of calling Put_Line in a single line, we have this:

Note that typical Ada code uses a limit of at least 79 columns. Therefore, please don't take the coding style from this course as a reference!

1 Note

Each code example from this book has an associated "code block metadata", which contains the name of the "project" and an MD5 hash value. This information is used to identify a single code example.

You can find all code examples in a zip file, which you can download from the learn website². The directory structure in the zip file is based on the code block metadata. For example, if you're searching for a code example with this metadata:

- Project: Courses.Intro_To_Ada.Imperative_Language.Greet
- MD5: cba89a34b87c9dfa71533d982d05e6ab

you will find it in this directory:

CONTENTS: 1

¹ http://creativecommons.org/licenses/by-sa/4.0

projects/Courses/Intro_To_Ada/Imperative_Language/Greet/ cba89a34b87c9dfa71533d982d05e6ab/

In order to use this code example, just follow these steps:

- 1. Unpack the zip file;
- 2. Go to target directory;
- 3. Start GNAT Studio on this directory;
- 4. Build (or compile) the project;
- 5. Run the application (if a main procedure is available in the project).

2 CONTENTS:

² https://learn.adacore.com/zip/learning-ada_code.zip

INTRODUCTION

1.1 History

In the 1970s the United States Department of Defense (DOD) suffered from an explosion of the number of programming languages, with different projects using different and non-standard dialects or language subsets / supersets. The DOD decided to solve this problem by issuing a request for proposals for a common, modern programming language. The winning proposal was one submitted by Jean Ichbiah from CII Honeywell-Bull.

The first Ada standard was issued in 1983; it was subsequently revised and enhanced in 1995, 2005 and 2012, with each revision bringing useful new features.

This tutorial will focus on Ada 2012 as a whole, rather than teaching different versions of the language.

1.2 Ada today

Today, Ada is heavily used in embedded real-time systems, many of which are safety critical. While Ada is and can be used as a general-purpose language, it will really shine in low-level applications:

- Embedded systems with low memory requirements (no garbage collector allowed).
- Direct interfacing with hardware.
- Soft or hard real-time systems.
- · Low-level systems programming.

Specific domains seeing Ada usage include Aerospace & Defense, civil aviation, rail, and many others. These applications require a high degree of safety: a software defect is not just an annoyance, but may have severe consequences. Ada provides safety features that detect defects at an early stage — usually at compilation time or using static analysis tools. Ada can also be used to create applications in a variety of other areas, such as:

- Video game programming³
- Real-time audio⁴
- Kernel modules⁵

This is a non-comprehensive list that hopefully sheds light on which kind of programming Ada is good at.

³ https://github.com/AdaDoom3/AdaDoom3

⁴ http://www.electronicdesign.com/embedded-revolution/assessing-ada-language-audio-applications

⁵ http://www.nihamkin.com/tag/kernel.html

In terms of modern languages, the closest in terms of targets and level of abstraction are probably $C++^6$ and $Rust^7$.

1.3 Philosophy

Ada's philosophy is different from most other languages. Underlying Ada's design are principles that include the following:

- Readability is more important than conciseness. Syntactically this shows through the fact that keywords are preferred to symbols, that no keyword is an abbreviation, etc.
- Very strong typing. It is very easy to introduce new types in Ada, with the benefit of preventing data usage errors.
 - It is similar to many functional languages in that regard, except that the programmer has to be much more explicit about typing in Ada, because there is almost no type inference.
- Explicit is better than implicit. Although this is a Python⁸ commandment, Ada takes it way further than any language we know of:
 - There is mostly no structural typing, and most types need to be explicitly named by the programmer.
 - As previously said, there is mostly no type inference.
 - Semantics are very well defined, and undefined behavior is limited to an absolute minimum.
 - The programmer can generally give a *lot* of information about what their program means to the compiler (and other programmers). This allows the compiler to be extremely helpful (read: strict) with the programmer.

During this course, we will explain the individual language features that are building blocks for that philosophy.

1.4 SPARK

While this class is solely about the Ada language, it is worth mentioning that another language, extremely close to and interoperable with Ada, exists: the SPARK language.

SPARK is a subset of Ada, designed so that the code written in SPARK is amenable to automatic proof. This provides a level of assurance with regard to the correctness of your code that is much higher than with a regular programming language.

There is a dedicated course for the SPARK language but keep in mind that every time we speak about the specification power of Ada during this course, it is power that you can leverage in SPARK to help proving the correctness of program properties ranging from absence of run-time errors to compliance with formally specified functional requirements.

⁶ https://en.wikipedia.org/wiki/C%2B%2B

⁷ https://www.rust-lang.org/en-US/

⁸ https://www.python.org

IMPERATIVE LANGUAGE

Ada is a multi-paradigm language with support for object orientation and some elements of functional programming, but its core is a simple, coherent procedural/imperative language akin to C or Pascal.

1 In other languages

One important distinction between Ada and a language like C is that statements and expressions are very clearly distinguished. In Ada, if you try to use an expression where a statement is required then your program will fail to compile. This rule supports a useful stylistic principle: expressions are intended to deliver values, not to have side effects. It can also prevent some programming errors, such as mistakenly using the equality operator = instead of the assignment operation := in an assignment statement.

2.1 Hello world

Here's a very simple imperative Ada program:

Listing 1: greet.adb

```
with Ada.Text_IO;

procedure Greet is
begin
-- Print "Hello, World!" to the screen
Ada.Text_IO.Put_Line ("Hello, World!");
end Greet;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet
MD5: cba89a34b87c9dfa71533d982d05e6ab
```

Runtime output

```
Hello, World!
```

which we'll assume is in the source file greet.adb.

There are several noteworthy things in the above program:

- A subprogram in Ada can be either a procedure or a function. A procedure, as illustrated above, does not return a value when called.
- with is used to reference external modules that are needed in the procedure. This is similar to import in various languages or roughly similar to #include in C and C++. We'll see later how they work in detail. Here, we are requesting a standard library

module, the Ada.Text_IO package, which contains a procedure to print text on the screen: Put Line.

- Greet is a procedure, and the main entry point for our first program. Unlike in C or C++, it can be named anything you prefer. The builder will determine the entry point. In our simple example, **gprbuild**, GNAT's builder, will use the file you passed as parameter.
- Put_Line is a procedure, just like Greet, except it is declared in the Ada.Text_IO module. It is the Ada equivalent of C's printf.
- Comments start with -- and go to the end of the line. There is no multi-line comment syntax, that is, it is not possible to start a comment in one line and continue it in the next line. The only way to create multiple lines of comments in Ada is by using -- on each line. For example:

```
-- We start a comment in this line...
-- and we continue on the second line...
```

1 In other languages

Procedures are similar to functions in C or C++ that return **void**. We'll see later how to declare functions in Ada.

Here is a minor variant of the "Hello, World" example:

Listing 2: greet.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Greet is
begin
-- Print "Hello, World!" to the screen
Put_Line ("Hello, World!");
end Greet;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet_2
MD5: a58al193207df44aa6edaa4fe1c14280
```

Runtime output

```
Hello, World!
```

This version utilizes an Ada feature known as a **use** clause, which has the form **use** package-name. As illustrated by the call on Put_Line, the effect is that entities from the named package can be referenced directly, without the package-name. prefix.

2.2 Imperative language - If/Then/Else

This section describes Ada's **if** statement and introduces several other fundamental language facilities including integer I/O, data declarations, and subprogram parameter modes.

Ada's **if** statement is pretty unsurprising in form and function:

Listing 3: check_positive.adb

```
with Ada.Text I0;
                               use Ada. Text IO;
   with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
2
3
   procedure Check Positive is
4
      N : Integer:
5
   begin
6
       -- Put a String
7
      Put ("Enter an integer value: ");
8
9
       -- Read in an integer value
10
      Get (N);
11
12
      if N > 0 then
13
          -- Put an Integer
14
         Put (N):
15
         Put Line (" is a positive number");
16
      end if;
17
   end Check Positive;
18
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Check_Positive
MD5: 2e8b4b2f3f258fd9e02c2d65846af101
```

The **if** statement minimally consists of the reserved word **if**, a condition (which must be a Boolean value), the reserved word **then** and a non-empty sequence of statements (the **then** part) which is executed if the condition evaluates to True, and a terminating **end if**.

This example declares an integer variable N, prompts the user for an integer, checks if the value is positive and, if so, displays the integer's value followed by the string " is a positive number". If the value is not positive, the procedure does not display any output.

The type Integer is a predefined signed type, and its range depends on the computer architecture. On typical current processors Integer is 32-bit signed.

The example illustrates some of the basic functionality for integer input-output. The relevant subprograms are in the predefined package Ada.Integer_Text_IO and include the Get procedure (which reads in a number from the keyboard) and the Put procedure (which displays an integer value).

Here's a slight variation on the example, which illustrates an **if** statement with an **else** part:

Listing 4: check_positive.adb

```
use Ada.Text I0;
   with Ada.Text IO;
   with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
3
   procedure Check_Positive is
      N : Integer;
5
   begin
6
          Put a String
      Put ("Enter an integer value: ");
8
9
      -- Reads in an integer value
10
      Get (N);
11
12
      -- Put an Integer
13
      Put (N);
14
15
```

(continues on next page)

```
if N > 0 then
    Put_Line (" is a positive number");
else
    Put_Line (" is not a positive number");
end if;
end Check_Positive;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Check_Positive_2
MD5: 28fca0d7840d06d478e5933e8182d1db
```

In this example, if the input value is not positive then the program displays the value followed by the String " is not a positive number".

Our final variation illustrates an if statement with elsif sections:

Listing 5: check_direction.adb

```
with Ada.Text_IO;
                               use Ada. Text IO;
2
   with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
   procedure Check_Direction is
4
      N : Integer;
5
   begin
6
7
      Put ("Enter an integer value: ");
      Get (N);
8
      Put (N);
9
10
      if N = 0 or N = 360 then
11
         Put_Line (" is due north");
12
      elsif N in 1 .. 89 then
13
          Put Line (" is in the northeast quadrant");
14
      elsif N = 90 then
15
          Put Line (" is due east");
16
      elsif N in 91 .. 179 then
17
         Put_Line (" is in the southeast quadrant");
18
      elsif N = 180 then
19
         Put_Line (" is due south");
20
      elsif N in 181 .. 269 then
21
         Put_Line (" is in the southwest quadrant");
22
      elsif N = 270 then
23
         Put_Line (" is due west");
24
      elsif N in 271 .. 359 then
25
         Put Line (" is in the northwest quadrant");
26
      else
27
         Put_Line (" is not in the range 0..360");
28
      end if;
29
   end Check_Direction;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Check_Direction MD5: 7759d30c9bb0bfb88efdf12128f9c382
```

This example expects the user to input an integer between 0 and 360 inclusive, and displays which quadrant or axis the value corresponds to. The **in** operator in Ada tests whether a scalar value is within a specified range and returns a Boolean result. The effect of the program should be self-explanatory; later we'll see an alternative and more efficient style to accomplish the same effect, through a **case** statement.

Ada's elsif keyword differs from C or C++, where nested else .. if blocks would be

used instead. And another difference is the presence of the **end if** in Ada, which avoids the problem known as the "dangling else".

2.3 Imperative language - Loops

Ada has three ways of specifying loops. They differ from the C / Java / Javascript for-loop, however, with simpler syntax and semantics in line with Ada's philosophy.

2.3.1 For loops

The first kind of loop is the **for** loop, which allows iteration through a discrete range.

Listing 6: greet 5a.adb

```
with Ada. Text IO; use Ada. Text IO;
2
   procedure Greet 5a is
3
   beain
4
      for I in 1 .. 5 loop
5
          -- Put Line is a procedure call
6
         Put Line ("Hello, World!"
7
                     & Integer'Image (I));
8
                    ^ Procedure parameter
      end loop;
10
   end Greet 5a;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet_5a
MD5: 7f588b67947126f789333adfaaf1b638
```

Runtime output

```
Hello, World! 1
Hello, World! 2
Hello, World! 3
Hello, World! 4
Hello, World! 5
```

A few things to note:

- 1 .. 5 is a discrete range, from 1 to 5 inclusive.
- The loop parameter I (the name is arbitrary) in the body of the loop has a value within this range.
- I is local to the loop, so you cannot refer to I outside the loop.
- Although the value of I is incremented at each iteration, from the program's perspective it is constant. An attempt to modify its value is illegal; the compiler would reject the program.
- Integer' Image is a function that takes an Integer and converts it to a **String**. It is an example of a language construct known as an *attribute*, indicated by the 'syntax, which will be covered in more detail later.
- The & symbol is the concatenation operator for String values
- The end loop marks the end of the loop

The "step" of the loop is limited to 1 (forward direction) and -1 (backward). To iterate backwards over a range, use the **reverse** keyword:

Listing 7: greet_5a_reverse.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Greet_5a_Reverse is
begin
for I in reverse 1 .. 5 loop
    Put_Line ("Hello, World!"
    & Integer'Image (I));
end loop;
end Greet_5a_Reverse;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet_5a_Reverse
MD5: a0d5dcfc471fb1a107477c934fa527c2
```

Runtime output

```
Hello, World! 5
Hello, World! 4
Hello, World! 3
Hello, World! 2
Hello, World! 1
```

The bounds of a **for** loop may be computed at run-time; they are evaluated once, before the loop body is executed. If the value of the upper bound is less than the value of the lower bound, then the loop is not executed at all. This is the case also for **reverse** loops. Thus no output is produced in the following example:

Listing 8: greet no op.adb

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet_No_Op
MD5: 5070693fb0324d3e4e43a8c8c4f046e1
```

Build output

The **for** loop is more general than what we illustrated here; more on that later.

2.3.2 Bare loops

The simplest loop in Ada is the bare loop, which forms the foundation of the other kinds of Ada loops.

Listing 9: greet_5b.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Greet_5b is
3
      -- Variable declaration:
4
      I : Integer := 1;
5
          ^ Type
6
                      ^ Initial value
7
   begin
8
      loop
9
          Put_Line ("Hello, World!"
10
                    & Integer'Image (I));
11
12
          -- Exit statement:
13
          exit when I = 5;
14
                    ^ Boolean condition
15
16
          -- Assignment:
17
          I := I + 1;
18
          -- There is no I++ short form to
19
          -- increment a variable
20
      end loop;
21
   end Greet 5b;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet_5b
MD5: 5b218a64a07f64bd97774b574883c44a
```

Runtime output

```
Hello, World! 1
Hello, World! 2
Hello, World! 3
Hello, World! 4
Hello, World! 5
```

This example has the same effect as Greet_5a shown earlier.

It illustrates several concepts:

- We have declared a variable named I between the **is** and the **begin**. This constitutes a *declarative region*. Ada clearly separates the declarative region from the statement part of a subprogram. A declaration can appear in a declarative region but is not allowed as a statement.
- The bare loop statement is introduced by the keyword **loop** on its own and, like every kind of loop statement, is terminated by the combination of keywords **end loop**. On its own, it is an infinite loop. You can break out of it with an **exit** statement.
- The syntax for assignment is :=, and the one for equality is =. There is no way to confuse them, because as previously noted, in Ada, statements and expressions are distinct, and expressions are not valid statements.

2.3.3 While loops

The last kind of loop in Ada is the while loop.

Listing 10: greet_5c.adb

```
with Ada.Text IO; use Ada.Text IO;
   procedure Greet_5c is
3
      I : Integer := 1;
4
   begin
5
         Condition must be a Boolean value
6
          (no Integers).
7
      -- Operator "<=" returns a Boolean
8
      while I <= 5 loop
9
         Put_Line ("Hello, World!"
10
                    & Integer'Image (I));
11
12
         I := I + 1;
13
      end loop;
14
  end Greet_5c;
15
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet_5c
MD5: 5d1d099477795b226db43736c2810274
```

Runtime output

```
Hello, World! 1
Hello, World! 2
Hello, World! 3
Hello, World! 4
Hello, World! 5
```

The condition is evaluated before each iteration. If the result is false, then the loop is terminated.

This program has the same effect as the previous examples.

1 In other languages

Note that Ada has different semantics than C-based languages with respect to the condition in a while loop. In Ada the condition has to be a Boolean value or the compiler will reject the program; the condition is not an integer that is treated as either **True** or **False** depending on whether it is non-zero or zero.

2.4 Imperative language - Case statement

Ada's **case** statement is similar to the C and C++ **switch** statement, but with some important differences.

Here's an example, a variation of a program that was shown earlier with an if statement:

Listing 11: check direction.adb

(continues on next page)

```
loop
7
          Put ("Enter an integer value: ");
8
          Get (N);
          Put (N);
10
11
          case N is
12
              when 0 | 360 =>
13
                 Put_Line
14
                   (" is due north");
15
              when 1 ... 89 =>
16
                 Put Line
17
                   (\overline{\ } is in the northeast quadrant");
18
              when 90 =>
19
                 Put_Line
20
                   (" is due east");
              when 91 .. 179 =>
22
                 Put Line
23
                   (" is in the southeast quadrant");
24
              when 180 =>
25
                 Put Line
26
                   (" is due south");
27
              when 181 .. 269 =>
28
                 Put Line
29
                   (" is in the southwest quadrant");
30
              when 270 =>
31
                 Put Line
32
                   (" is due west");
33
              when 271 .. 359 =>
34
                 Put Line
35
                   (" is in the northwest quadrant");
36
              when others =>
37
                 Put Line
38
                   (" Au revoir");
39
                 exit;
40
          end case;
41
       end loop;
42
   end Check_Direction;
43
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Check_Direction_2
MD5: 1c758b76a2c3991cb4e2a0cf5e172ac3
```

This program repeatedly prompts for an integer value and then, if the value is in the range 0 ... 360, displays the associated quadrant or axis. If the value is an Integer outside this range, the loop (and the program) terminate after outputting a farewell message.

The effect of the case statement is similar to the if statement in an earlier example, but the case statement can be more efficient because it does not involve multiple range tests.

Notable points about Ada's case statement:

- The case expression (here the variable N) must be of a discrete type, i.e. either an integer type or an enumeration type. Discrete types will be covered in more detail later *discrete types* (page 43).
- Every possible value for the case expression needs to be covered by a unique branch of the case statement. This will be checked at compile time.
- A branch can specify a single value, such as 0; a range of values, such as 1 ... 89; or any combination of the two (separated by a |).
- As a special case, an optional final branch can specify others, which covers all values

not included in the earlier branches.

- Execution consists of the evaluation of the case expression and then a transfer of control to the statement sequence in the unique branch that covers that value.
- When execution of the statements in the selected branch has completed, control resumes after the **end case**. Unlike C, execution does not fall through to the next branch. So Ada doesn't need (and doesn't have) a **break** statement.

2.5 Imperative language - Declarative regions

As mentioned earlier, Ada draws a clear syntactic separation between declarations, which introduce names for entities that will be used in the program, and statements, which perform the processing. The areas in the program where declarations may appear are known as declarative regions.

In any subprogram, the section between the **is** and the **begin** is a declarative region. You can have variables, constants, types, inner subprograms, and other entities there.

We've briefly mentioned variable declarations in previous subsection. Let's look at a simple example, where we declare an integer variable X in the declarative region and perform an initialization and an addition on it:

Listing 12: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Main is
      X : Integer;
   begin
      X := 0;
6
      Put_Line ("The initial value of X is "
                 & Integer'Image (X));
      Put_Line ("Performing operation on X...");
10
      X := X + 1;
11
12
      Put Line ("The value of X now is "
13
                 & Integer'Image (X));
14
   end Main:
15
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Variable_Declaration MD5: cbb08d5e382fbfcc28e986bea80cd253
```

Runtime output

```
The initial value of X is \theta Performing operation on X... The value of X now is \theta
```

Let's look at an example of a nested procedure:

Listing 13: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Main is
procedure Nested is
begin
```

(continues on next page)

```
Put_Line ("Hello World");
end Nested;
begin
Nested;
-- Call to Nested
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Nested_Procedure MD5: 2e7fb267e31232196065febd5e35e6ef
```

Runtime output

```
Hello World
```

A declaration cannot appear as a statement. If you need to declare a local variable amidst the statements, you can introduce a new declarative region with a block statement:

Listing 14: greet.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Greet is
   begin
4
      loop
5
          Put_Line ("Please enter your name: ");
6
8
             Name : String := Get Line;
q
                                ^ Call to the
10
             - -
                                  Get Line function
11
          begin
12
             exit when Name = "";
13
             Put Line ("Hi " & Name & "!");
          end;
15
16
          -- Name is undefined here
17
      end loop;
18
19
     Put_Line ("Bye!");
20
   end Greet;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Greet_6
MD5: a9c0c14a1b3e2ebe07cd88f442787e3a
```

Attention

The Get_Line function allows you to receive input from the user, and get the result as a string. It is more or less equivalent to the scanf C function.

It returns a **String**, which, as we will see later, is an *Unconstrained array type* (page 74). For now we simply note that, if you wish to declare a **String** variable and do not know its size in advance, then you need to initialize the variable during its declaration.

2.6 Imperative language - conditional expressions

Ada 2012 introduced an expression analog for conditional statements (if and case).

2.6.1 If expressions

Here's an alternative version of an example we saw earlier; the **if** statement has been replaced by an **if** expression:

Listing 15: check_positive.adb

```
with Ada.Text I0;
                               use Ada.Text I0;
   with Ada. Integer Text IO; use Ada. Integer Text IO;
   procedure Check Positive is
4
      N : Integer;
5
   begin
6
      Put ("Enter an integer value: ");
7
      Get (N):
8
      Put (N);
9
10
      declare
11
         S : constant String :=
12
            (if N > 0)
13
               then " is a positive number"
14
               else " is not a positive number");
15
      begin
16
         Put_Line (S);
17
      end;
18
   end Check Positive;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Check_Positive
MD5: 01f23463b14774f750dbb21f6c65ea09
```

The **if** expression evaluates to one of the two Strings depending on N, and assigns that value to the local variable S.

Ada's if expressions are similar to if statements. However, there are a few differences that stem from the fact that it is an expression:

- · All branches' expressions must be of the same type
- It must be surrounded by parentheses if the surrounding expression does not already contain them
- An else branch is mandatory unless the expression following then has a Boolean value. In that case an else branch is optional and, if not present, defaults to else True.

Here's another example:

Listing 16: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Main is
begin
for I in 1 .. 10 loop
Put_Line (if I mod 2 = 0
then "Even"
```

(continues on next page)

```
else "Odd");
end loop;
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Even_Odd
MD5: c89c3233ab8822c828f7a7bba8fd3f1c
```

Runtime output

```
Odd
Even
```

This program produces 10 lines of output, alternating between "Odd" and "Even".

2.6.2 Case expressions

Analogous to **if** expressions, Ada also has **case** expressions. They work just as you would expect.

Listing 17: main.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Main is
3
   begin
4
       for I in 1 .. 10 loop
5
          Put Line
6
             (case I is
              when 1 \mid 3 \mid 5 \mid 7 \mid 9 \Rightarrow "0dd"
8
              when 2 | 4 | 6 | 8 | 10 => "Even");
       end loop;
10
   end Main;
11
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Imperative_Language.Case_Expression MD5: 6ce40efc987c2665960b1f08d30d780d
```

Runtime output

```
Odd
Even
```

Introduction to Ada

This program has the same effect as the preceding example.

The syntax differs from **case** statements, with branches separated by commas.

THREE

SUBPROGRAMS

3.1 Subprograms

So far, we have used procedures, mostly to have a main body of code to execute. Procedures are one kind of *subprogram*.

There are two kinds of subprograms in Ada, *functions* and *procedures*. The distinction between the two is that a function returns a value, and a procedure does not.

This example shows the declaration and definition of a function:

```
Listing 1: increment.ads
```

```
function Increment (I : Integer) return Integer;
```

Listing 2: increment.adb

```
-- We declare (but don't define) a function with
-- one parameter, returning an integer value

function Increment (I : Integer) return Integer is
-- We define the Increment function
begin
return I + 1;
end Increment;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Increment
MD5: 582fe283730a130cec071c455a0ce3d4
```

Subprograms in Ada can, of course, have parameters. One syntactically important note is that a subprogram which has no parameters does not have a parameter section at all, for example:

```
procedure Proc;
function Func return Integer;
```

Here's another variation on the previous example:

Listing 3: increment_by.ads

```
function Increment_By
(I : Integer := 0;
Incr : Integer := 1) return Integer;
/ Default value for parameters
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Increment_By MD5: 5728b915789beee0b5546ea7b36a1cc2
```

In this example, we see that parameters can have default values. When calling the subprogram, you can then omit parameters if they have a default value. Unlike C/C++, a call to a subprogram without parameters does not include parentheses.

This is the implementation of the function above:

Listing 4: increment by adb

```
function Increment_By
(I : Integer := 0;
Incr : Integer := 1) return Integer is
begin
return I + Incr;
end Increment_By;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Increment_By
MD5: 07c85e5c1272ea396bf4dbc0cefcdce7
```

1 In the GNAT toolchain

The Ada standard doesn't mandate in which file the specification or the implementation of a subprogram must be stored. In other words, the standard doesn't require a specific file structure or specific file name extensions. For example, we could save both the specification and the implementation of the Increment function above in a file called increment.txt. (We could even store the entire source code of a system in a single file.) From the standard's perspective, this would be completely acceptable.

The GNAT toolchain, however, requires the following file naming scheme:

- files with the .ads extension contain the specification, while
- files with the .adb extension contain the implementation.

Therefore, in the GNAT toolchain, the specification of the Increment function must be stored in the increment.ads file, while its implementation must be stored in the increment.adb file. This rule always applies to packages, which we discuss *later* (page 31). (Note, however, that it's possible to circumvent this rule.) For more details, you may refer to the Introduction to GNAT Toolchain course or the GPRbuild User's Guide⁹.

3.1.1 Subprogram calls

We can then call our subprogram this way:

Listing 5: show_increment.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Increment_By;

procedure Show_Increment is
    A, B, C : Integer;
begin
    C := Increment_By;

(continues on next page)
```

⁹ https://docs.adacore.com/gprbuild-docs/html/gprbuild_ug.html

```
^ Parameterless call,
8
                           value of I is 0
9
                           and Incr is 1
10
11
       Put_Line ("Using defaults for Increment_By is "
12
                  & Integer'Image (C));
13
14
       A := 10;
15
       B := 3;
16
       C := Increment_By (A, B);
17
                             ` Regular parameter passing
18
19
       Put_Line ("Increment of "
20
                  & Integer'Image (A)
21
                  & " with "
22
                  & Integer'Image (B)
23
                  & " is "
24
                  & Integer'Image (C));
25
26
       A := 20;
27
       B := 5;
28
       C := Increment_By (I
                                 => A,
29
                            Incr => B);
30
                            ^ Named parameter passing
31
32
       Put_Line ("Increment of "
33
                  & Integer'Image (A)
34
                  & " with "
35
                  & Integer'Image (B)
36
                  & " is "
37
                  & Integer'Image (C));
38
   end Show Increment;
39
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Increment_By MD5: dcb501c8c6815b03c6841fc8b80d6911
```

Runtime output

```
Using defaults for Increment_By is 1
Increment of 10 with 3 is 13
Increment of 20 with 5 is 25
```

Ada allows you to name the parameters when you pass them, whether they have a default or not. There are some rules:

- Positional parameters come first.
- A positional parameter cannot follow a named parameter.

As a convention, people usually name parameters at the call site if the function's corresponding parameters has a default value. However, it is also perfectly acceptable to name every parameter if it makes the code clearer.

3.1.2 Nested subprograms

As briefly mentioned earlier, Ada allows you to declare one subprogram inside another.

This is useful for two reasons:

- It lets you organize your programs in a cleaner fashion. If you need a subprogram only as a "helper" for another subprogram, then the principle of localization indicates that the helper subprogram should be declared nested.
- It allows you to share state easily in a controlled fashion, because the nested subprograms have access to the parameters, as well as any local variables, declared in the outer scope.

For the previous example, we can move the duplicated code (call to Put_Line) to a separate procedure. This is a shortened version with the nested Display Result procedure.

Listing 6: show_increment.adb

```
with Ada. Text IO; use Ada. Text IO;
   with Increment By;
   procedure Show Increment is
      A, B, C : Integer;
6
      procedure Display_Result is
7
       begin
8
          Put_Line ("Increment of "
9
                     & Integer'Image (A)
10
                     & " with "
11
                     & Integer'Image (B)
12
                     & " is '
13
                     & Integer'Image (C));
14
      end Display_Result;
15
16
   begin
17
      A := 10;
18
      B := 3:
19
      C := Increment By (A, B);
20
      Display Result;
21
      A := 20;
22
      B := 5;
23
      C := Increment By (A, B);
24
      Display_Result;
   end Show Increment;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Increment_By MD5: 23ec8ae3080c042123a9e82ee6b3d9e3
```

Runtime output

```
Increment of 10 with 3 is 13
Increment of 20 with 5 is 25
```

3.1.3 Function calls

An important feature of function calls in Ada is that the return value at a call cannot be ignored; that is, a function call cannot be used as a statement.

If you want to call a function and do not need its result, you will still need to explicitly store it in a local variable.

Listing 7: quadruple.adb

```
function Quadruple (I : Integer)
return Integer is

(continues on next page)
```

```
3
       function Double (I : Integer)
4
                          return Integer is
       begin
          return I * 2;
7
       end Double;
8
9
      Res : Integer := Double (Double (I));
10
                         ^ Calling the Double
11
                           function
12
   begin
13
      Double (I);
14
      -- ERROR: cannot use call to function
15
                  "Double" as a statement
16
17
      return Res;
18
  end Quadruple;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Quadruple
MD5: 44326f12a9d797ea13ffe52ea48fc36f
```

Build output

```
quadruple.adb:14:04: error: cannot use call to function "Double" as a statement quadruple.adb:14:04: error: return value of a function call cannot be ignored gprbuild: *** compilation phase failed
```

1 In the GNAT toolchain

In GNAT, with all warnings activated, it becomes even harder to ignore the result of a function, because unused variables will be flagged. For example, this code would not be valid:

```
function Read_Int
   (Stream : Network_Stream;
   Result : out Integer) return Boolean;

procedure Main is
   Stream : Network_Stream := Get_Stream;
   My_Int : Integer;

   -- Warning: in the line below, B is
   -- never read.
   B : Boolean := Read_Int (Stream, My_Int);
begin
   null;
end Main;
```

You then have two solutions to silence this warning:

• Either annotate the variable with pragma Unreferenced , e.g.:

```
B : Boolean := Read_Int (Stream, My_Int);
pragma Unreferenced (B);
```

• Or give the variable a name that contains any of the strings discard dummy ignore junk unused (case insensitive)

3.2 Parameter modes

So far we have seen that Ada is a safety-focused language. There are many ways this is realized, but two important points are:

- Ada makes the user specify as much as possible about the behavior expected for the program, so that the compiler can warn or reject if there is an inconsistency.
- Ada provides a variety of techniques for achieving the generality and flexibility of pointers and dynamic memory management, but without the latter's drawbacks (such as memory leakage and dangling references).

Parameter modes are a feature that helps achieve the two design goals above. A subprogram parameter can be specified with a mode, which is one of the following:

in	Parameter can only be read, not written
out	Parameter can be written to, then read
in out	Parameter can be both read and written

The default mode for parameters is **in**; so far, most of the examples have been using **in** parameters.

Historically

Functions and procedures were originally more different in philosophy. Before Ada 2012, functions could only take **in** parameters.

3.3 Subprogram calls

3.3.1 In parameters

The first mode for parameters is the one we have been implicitly using so far. Parameters passed using this mode cannot be modified, so that the following program will cause an error:

Listing 8: swap.adb

```
procedure Swap (A, B : Integer) is
      Tmp : Integer;
   begin
3
      Tmp := A;
4
5
      -- Error: assignment to "in" mode
6
                 parameter not allowed
7
      A := B:
8
      -- Error: assignment to "in" mode
                 parameter not allowed
      B := Tmp;
12
  end Swap;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Swap
MD5: 478ac23f878934aae820e4b9c056d939
```

Build output

```
swap.adb:8:04: error: assignment to "in" mode parameter not allowed
swap.adb:12:04: error: assignment to "in" mode parameter not allowed
gprbuild: *** compilation phase failed
```

The fact that **in** is the default mode is very important. It means that a parameter will not be modified unless you explicitly specify a mode in which modification is allowed.

3.3.2 In out parameters

To correct our code above, we can use an **in out** parameter.

Listing 9: in_out_params.adb

```
with Ada.Text_IO; use Ada.Text_IO;
2
   procedure In Out Params is
3
      procedure Swap (A, B : in out Integer) is
4
         Tmp : Integer;
5
      begin
6
         Tmp := A;
7
8
              := B;
             := Tmp;
9
         В
      end Swap;
10
11
      A : Integer := 12;
12
      B : Integer := 44;
13
   beain
14
        Swap (A, B);
15
16
        -- Prints 44
17
        Put Line (Integer'Image (A));
18
   end In Out Params;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.In_Out_Params
MD5: 319358e479449c115cf2b3cbb4ff3a6b
```

Runtime output

```
44
```

An **in out** parameter will allow read and write access to the object passed as parameter, so in the example above, we can see that A is modified after the call to Swap.

A Attention

While **in out** parameters look a bit like references in C++, or regular parameters in Java that are passed by-reference, the Ada language standard does not mandate "by reference" passing for in out parameters except for certain categories of types as will be explained later.

In general, it is better to think of modes as higher level than by-value versus by-reference semantics. For the compiler, it means that an array passed as an **in** parameter might be passed by reference, because it is more efficient (which does not change anything for the user since the parameter is not assignable). However, a parameter of a discrete type will always be passed by copy, regardless of its mode (which is more efficient on most architectures).

3.3.3 Out parameters

The **out** mode applies when the subprogram needs to write to a parameter that might be uninitialized at the point of call. Reading the value of an **out** parameter is permitted, but it should only be done after the subprogram has assigned a value to the parameter. Out parameters behave a bit like return values for functions. When the subprogram returns, the actual parameter (a variable) will have the value of the out parameter at the point of return.

1 In other languages

Ada doesn't have a tuple construct and does not allow returning multiple values from a subprogram (except by declaring a full-fledged record type). Hence, a way to return multiple values from a subprogram is to use **out** parameters.

For example, a procedure reading integers from the network could have one of the following specifications:

```
procedure Read_Int
  (Stream : Network_Stream;
  Success : out Boolean;
  Result : out Integer);

function Read_Int
  (Stream : Network_Stream;
  Result : out Integer) return Boolean;
```

While reading an out variable before writing to it should, ideally, trigger an error, imposing that as a rule would cause either inefficient run-time checks or complex compile-time rules. So from the user's perspective an out parameter acts like an uninitialized variable when the subprogram is invoked.

1 In the GNAT toolchain

GNAT will detect simple cases of incorrect use of out parameters. For example, the compiler will emit a warning for the following program:

Listing 10: outp.adb

```
procedure Outp is
1
      procedure Foo (A : out Integer) is
2
         B : Integer := A;
3
                         ^ Warning on reference
4
                           to uninitialized A
5
      begin
6
         A := B;
      end Foo;
8
   begin
      null;
10
   end Outp;
```

Code block metadata

Project: Courses.Intro_To_Ada.Subprograms.Out_Params
MD5: 36bdb4e541297d7fb0b075816cb6e73a

Build output

outp.adb:3:22: warning: "A" may be referenced before it has a value [enabled by_default]

3.3.4 Forward declaration of subprograms

As we saw earlier, a subprogram can be declared without being fully defined, This is possible in general, and can be useful if you need subprograms to be mutually recursive, as in the example below:

Listing 11: mutually_recursive_subprograms.adb

```
procedure Mutually_Recursive_Subprograms is
1
       procedure Compute_A (V : Natural);
2
        -- Forward declaration of Compute A
3
4
       procedure Compute B (V : Natural) is
5
       begin
6
           if V > 5 then
7
              Compute_A (V - 1);
8
              -- Call to Compute_A
9
           end if;
10
       end Compute B;
11
12
       procedure Compute_A (V : Natural) is
13
       begin
14
           if V > 2 then
15
              Compute B (V - 1);
16
                 Call to Compute B
17
           end if;
18
       end Compute A;
19
20
      Compute A (15);
21
   end Mutually_Recursive_Subprograms;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Mutually_Recursive_Subprograms MD5: 5ee030cdecc6c4aea8916cbb763e8526
```

3.4 Renaming

Subprograms can be renamed by using the **renames** keyword and declaring a new name for a subprogram:

```
procedure New_Proc renames Original_Proc;
```

This can be useful, for example, to improve the readability of your application when you're using code from external sources that cannot be changed in your system. Let's look at an example:

Listing 12: a_procedure_with_very_long_name_that_cannot_be_changed.ads

```
procedure A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed
(A_Message : String);
```

Listing 13: a_procedure_with_very_long_name_that_cannot_be_changed.adb

3.4. Renaming 27

(continues on next page)

```
Put_Line (A_Message);
end A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Proc_Renaming
MD5: 6d4952e9dee8ef69a9e3c3e185c635f1
```

As the wording in the name of procedure above implies, we cannot change its name. We can, however, rename it to something like Show in our test application and use this shorter name. Note that we also have to declare all parameters of the original subprogram — we may rename them, too, in the declaration. For example:

Listing 14: show_renaming.adb

```
with A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;

procedure Show_Renaming is

procedure Show (S : String) renames
    A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;

begin
    Show ("Hello World!");
end Show_Renaming;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Proc_Renaming
MD5: 5b3b550f8a1cbeb7d9cfd3673f6d42b3
```

Runtime output

```
Hello World!
```

Note that the original name (A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed) is still visible after the declaration of the Show procedure.

We may also rename subprograms from the standard library. For example, we may rename Integer'Image to Img:

Listing 15: show image renaming.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Show_Image_Renaming is

function Img (I : Integer) return String
    renames Integer'Image;

begin
    Put_Line (Img (2));
    Put_Line (Img (3));
end Show_Image_Renaming;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Subprograms.Integer_Image_Renaming
MD5: 9843b9d5967679c4fe8bd83a5213829f
```

Runtime output

```
2
3
```

Renaming also allows us to introduce default expressions that were not available in the original declaration. For example, we may specify "Hello World!" as the default for the **String** parameter of the Show procedure:

```
with A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;
procedure Show_Renaming_Defaults is

procedure Show (S : String := "Hello World!")
    renames
    A_Procedure_With_Very_Long_Name_That_Cannot_Be_Changed;

begin
    Show;
end Show_Renaming_Defaults;
```

3.4. Renaming 29

MODULAR PROGRAMMING

So far, our examples have been simple standalone subprograms. Ada is helpful in that regard, since it allows arbitrary declarations in a declarative part. We were thus able to declare our types and variables in the bodies of main procedures.

However, it is easy to see that this is not going to scale up for real-world applications. We need a better way to structure our programs into modular and distinct units.

Ada encourages the separation of programs into multiple packages and sub-packages, providing many tools to a programmer on a quest for a perfectly organized code-base.

4.1 Packages

Here is an example of a package declaration in Ada:

Listing 1: week.ads

```
package Week is
2
       Mon : constant String := "Monday";
3
       Tue : constant String := "Tuesday";
Wed : constant String := "Wednesday";
4
5
       Thu : constant String := "Thursday";
6
       Fri : constant String := "Friday";
       Sat : constant String := "Saturday";
       Sun : constant String := "Sunday";
9
10
   end Week;
11
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Week
MD5: 0fa033dc8fe2b9741483de273354e7ee
```

And here is how you use it:

Listing 2: main.adb

Code block metadata

Project: Courses.Intro To Ada.Modular Programming.Week

MD5: 03e17a75620de6a397b1d3c5a3e22f6a

Runtime output

First day of the week is Monday

Packages let you make your code modular, separating your programs into semantically significant units. Additionally the separation of a package's specification from its body (which we will see below) can reduce compilation time.

While the with clause indicates a dependency, you can see in the example above that you still need to prefix the referencing of entities from the Week package by the name of the package. (If we had included a use Week clause, then such a prefix would not have been necessary.)

Accessing entities from a package uses the dot notation, A.B, which is the same notation as the one used to access record fields.

A with clause can only appear in the prelude of a compilation unit (i.e., before the reserved word, such as procedure, that marks the beginning of the unit). It is not allowed anywhere else. This rule is only needed for methodological reasons: the person reading your code should be able to see immediately which units the code depends on.

1 In other languages

Packages look similar to, but are semantically very different from, header files in C/C++.

- The first and most important distinction is that packages are a language-level mechanism. This is in contrast to a #include'd header file, which is a functionality of the C preprocessor.
- An immediate consequence is that the with construct is a semantic inclusion mechanism, not a text inclusion mechanism. Hence, when you with a package, you are saying to the compiler "I'm depending on this semantic unit", and not "include this bunch of text in place here".
- The effect of a package thus does not vary depending on where it has been withed from. Contrast this with C/C++, where the meaning of the included text depends on the context in which the #include appears.

This allows compilation/recompilation to be more efficient. It also allows tools like IDEs to have correct information about the semantics of a program. In turn, this allows better tooling in general, and code that is more analyzable, even by humans.

An important benefit of Ada with clauses when compared to #include is that it is stateless. The order of with and use clauses does not matter, and can be changed without side effects.

1 In the GNAT toolchain

The Ada language standard does not mandate any particular relationship between source files and packages; for example, in theory you can put all your code in one file, or use your own file naming conventions. In practice, however, an implementation will have specific rules. With GNAT, each top-level compilation unit needs to go into a separate file. In the example above, the Week package will be in an .ads file (for Ada specification), and the Main procedure will be in an .adb file (for Ada body).

4.2 Using a package

As we have seen above, the **with** clause indicates a dependency on another package. However, every reference to an entity coming from the Week package had to be prefixed by the full name of the package. It is possible to make every entity of a package visible directly in the current scope, using the **use** clause.

In fact, we have been using the **use** clause since almost the beginning of this tutorial.

Listing 3: main.adb

```
with Ada.Text IO; use Ada.Text IO;
                      ^ Make every entity of the
2
                        Ada.Text_IO package
   - -
3
                        directly visible.
   with Week;
5
6
   procedure Main is
7
      use Week;
8
      -- Make every entity of the Week
9
          package directly visible.
10
  begin
      Put Line ("First day of the week is " & Mon);
12
  end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Week
MD5: ea54077d4ae165b28ae8facfe8ba2db7
```

Runtime output

```
First day of the week is Monday
```

As you can see in the example above:

- Put_Line is a subprogram that comes from the Ada.Text_IO package. We can reference it directly because we have **used** the package at the top of the Main unit.
- Unlike with clauses, a use clause can be placed either in the prelude, or in any declarative region. In the latter case the use clause will have an effect in its containing lexical scope.

4.3 Package body

In the simple example above, the Week package only has declarations and no body. That's not a mistake: in a package specification, which is what is illustrated above, you cannot declare bodies. Those have to be in the package body.

Listing 4: operations.ads

```
package Operations is

-- Declaration
function Increment_By
(I : Integer;
Incr : Integer := 0) return Integer;

function Get_Increment_Value return Integer;
end Operations;
```

Listing 5: operations.adb

```
package body Operations is
1
2
      Last_Increment : Integer := 1;
3
4
      function Increment By
5
              : Integer;
         (I
6
         Incr : Integer := 0) return Integer is
      begin
          if Incr /= 0 then
9
             Last_Increment := Incr;
10
         end if;
11
12
          return I + Last_Increment;
13
      end Increment_By;
14
15
      function Get Increment Value return Integer is
16
      begin
17
          return Last_Increment;
18
      end Get_Increment_Value;
19
20
   end Operations;
21
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Operations
MD5: 2adfb64e825605c74fecf6c9d45c8437
```

Here we can see that the body of the Increment_By function has to be declared in the body. Coincidentally, introducing a body allows us to put the Last_Increment variable in the body, and make them inaccessible to the user of the Operations package, providing a first form of encapsulation.

This works because entities declared in the body are *only* visible in the body.

This example shows how Last Increment is used indirectly:

Listing 6: main.adb

```
with Ada. Text IO; use Ada. Text IO;
1
   with Operations;
2
3
   procedure Main is
4
      use Operations;
5
6
      I : Integer := 0;
7
      R : Integer;
8
      procedure Display_Update_Values is
10
          Incr : constant Integer :=
11
                    Get_Increment_Value;
12
      begin
13
          Put_Line (Integer'Image (I)
14
                     & " incremented by "
15
                     & Integer'Image (Incr)
16
                     & " is
17
                     & Integer'Image (R));
18
          I := R;
19
      end Display_Update_Values;
20
   beain
21
      R := Increment_By (I);
22
```

```
Display_Update_Values;
23
      R := Increment_By (I);
24
      Display_Update_Values;
25
      R := Increment_By (I, 5);
27
      Display_Update_Values;
28
      R := Increment_By (I);
29
      Display_Update_Values;
30
31
      R := Increment_By (I, 10);
32
      Display_Update_Values;
33
       R := Increment_By (I);
34
      Display_Update_Values;
35
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Operations
MD5: 76190b1261a9652cfb7986ecec191e37
```

Runtime output

```
0 incremented by 1 is 1
1 incremented by 1 is 2
2 incremented by 5 is 7
7 incremented by 5 is 12
12 incremented by 10 is 22
22 incremented by 10 is 32
```

4.4 Child packages

Packages can be used to create hierarchies. We achieve this by using child packages, which extend the functionality of their parent package. One example of a child package that we've been using so far is the Ada.Text_IO package. Here, the parent package is called Ada, while the child package is called Text_IO. In the previous examples, we've been using the Put_Line procedure from the Text_IO child package.

1 Important

Ada also supports nested packages. However, since they can be more complicated to use, the recommendation is to use child packages instead. Nested packages will be covered in the advanced course.

Let's begin our discussion on child packages by taking our previous Week package:

Listing 7: week.ads

```
package Week is

Mon : constant String := "Monday";
   Tue : constant String := "Tuesday";
   Wed : constant String := "Wednesday";
   Thu : constant String := "Thursday";
   Fri : constant String := "Friday";
   Sat : constant String := "Saturday";
   Sun : constant String := "Sunday";
   (continues on next page)
```

```
end Week;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: 0fa033dc8fe2b9741483de273354e7ee
```

If we want to create a child package for Week, we may write:

Listing 8: week-child.ads

```
package Week.Child is

function Get_First_Of_Week return String;

end Week.Child;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: a7db38e772cf6153b5eb95069517e833
```

Here, Week is the parent package and Child is the child package. This is the corresponding package body of Week. Child:

Listing 9: week-child.adb

```
package body Week.Child is

function Get_First_Of_Week return String is
begin
return Mon;
end Get_First_Of_Week;

end Week.Child;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: 04dad82685ad9f0231c3084266b0af83
```

In the implementation of the Get_First_Of_Week function, we can use the Mon string directly, even though it was declared in the parent package Week. We don't write with Week here because all elements from the specification of the Week package — such as Mon, Tue and so on — are visible in the child package Week.Child.

Now that we've completed the implementation of the Week.Child package, we can use elements from this child package in a subprogram by simply writing with Week.Child. Similarly, if we want to use these elements directly, we write use Week.Child in addition. For example:

Listing 10: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Week.Child; use Week.Child;

procedure Main is
begin
Put_Line ("First day of the week is "
```

```
% Get_First_Of_Week);
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: e2f5c6ad3a92da4cb04ee7ec12293df4
```

Runtime output

```
First day of the week is Monday
```

4.4.1 Child of a child package

So far, we've seen a two-level package hierarchy. But the hierarchy that we can potentially create isn't limited to that. For instance, we could extend the hierarchy of the previous source code example by declaring a Week.Child.Grandchild package. In this case, Week. Child would be the parent of the Grandchild package. Let's consider this implementation:

Listing 11: week-child-grandchild.ads

```
package Week.Child.Grandchild is

function Get_Second_Of_Week return String;

end Week.Child.Grandchild;
```

Listing 12: week-child-grandchild.adb

```
package body Week.Child.Grandchild is

function Get_Second_Of_Week return String is
begin
return Tue;
end Get_Second_Of_Week;

end Week.Child.Grandchild;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: 03ee5932a68212b2e501370212508ab1
```

We can use this new Grandchild package in our test application in the same way as before: we can reuse the previous test application and adapt the with and use, and the function call. This is the updated code:

Listing 13: main.adb

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: 29ee409c8131bd9529c6bf6e366bb390
```

Runtime output

```
Second day of the week is Tuesday
```

Again, this isn't the limit for the package hierarchy. We could continue to extend the hierarchy of the previous example by implementing a Week.Child.Grandchild.Grandchild.Grandchild.gran

4.4.2 Multiple children

So far, we've seen a single child package of a parent package. However, a parent package can also have multiple children. We could extend the example above and implement a Week.Child 2 package. For example:

Listing 14: week-child_2.ads

```
package Week.Child_2 is

function Get_Last_Of_Week return String;

end Week.Child_2;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: bd3f63cacd142d9885600f4000b4573b
```

Here, Week is still the parent package of the Child package, but it's also the parent of the Child_2 package. In the same way, Child_2 is obviously one of the child packages of Week.

This is the corresponding package body of Week.Child_2:

Listing 15: week-child_2.adb

```
package body Week.Child_2 is

function Get_Last_Of_Week return String is
begin
return Sun;
end Get_Last_Of_Week;

end Week.Child_2;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: c2c03e4cb1daff02dd6076c2956ef2aa
```

We can now reference both children in our test application:

Listing 16: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Week.Child; use Week.Child;
with Week.Child_2; use Week.Child_2;

procedure Main is
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Child_Packages
MD5: 6a91f239fb2a2d8c702409c22467a424
```

Runtime output

```
First day of the week is Monday
Last day of the week is Sunday
```

4.4.3 Visibility

In the previous section, we've seen that elements declared in a parent package specification are visible in the child package. This is, however, not the case for elements declared in the package body of a parent package.

Let's consider the package Book and its child Additional_Operations:

Listing 17: book.ads

```
package Book is

Title : constant String :=
    "Visible for my children";

function Get_Title return String;

function Get_Author return String;

end Book;
```

Listing 18: book-additional_operations.ads

```
package Book.Additional_Operations is

function Get_Extended_Title return String;

function Get_Extended_Author return String;

end Book.Additional_Operations;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Visibility MD5: a0d67cff9aeff288709391d16306df00
```

This is the body of both packages:

Listing 19: book.adb

```
"Author not visible for my children";
4
5
       function Get_Title return String is
6
      beain
          return Title;
8
      end Get_Title;
9
10
       function Get_Author return String is
11
       begin
12
          return Author;
13
      end Get Author;
14
15
   end Book;
16
```

Listing 20: book-additional operations.adb

```
package body Book.Additional Operations is
1
      function Get_Extended_Title return String is
3
      begin
         return "Book Title: " & Title;
5
      end Get_Extended_Title;
6
7
      function Get_Extended_Author return String is
8
      begin
9
              "Author" string declared in the body
10
             of the Book package is not visible
11
             here. Therefore, we cannot write:
12
13
             return "Book Author: " & Author;
14
15
         return "Book Author: Unknown";
16
      end Get_Extended_Author;
17
18
   end Book.Additional_Operations;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Visibility
MD5: 68b7490da12bafae0aa6fe0ab76c6b1c
```

In the implementation of the Get_Extended_Title, we're using the Title constant from the parent package Book. However, as indicated in the comments of the Get_Extended_Author function, the Author string — which we declared in the body of the Book package — isn't visible in the Book.Additional_Operations package. Therefore, we cannot use it to implement the Get Extended Author function.

We can, however, use the Get_Author function from Book in the implementation of the Get_Extended_Author function to retrieve this string. Likewise, we can use this strategy to implement the Get_Extended_Title function. This is the adapted code:

Listing 21: book-additional_operations.adb

```
package body Book.Additional_Operations is

function Get_Extended_Title return String is
begin
return "Book Title: " & Get_Title;
end Get_Extended_Title;
```

```
function Get_Extended_Author return String is
begin
return "Book Author: " & Get_Author;
end Get_Extended_Author;

end Book.Additional_Operations;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Visibility
MD5: b00c187cb54d3fcb9574726028c1efc6
```

This is a simple test application for the packages above:

Listing 22: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;

with Book.Additional_Operations;

use Book.Additional_Operations;

procedure Main is
begin
    Put_Line (Get_Extended_Title);
    Put_Line (Get_Extended_Author);
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Visibility
MD5: bdc75987fe6le9401b400f8704890ebe
```

Runtime output

```
Book Title: Visible for my children
Book Author: Author not visible for my children
```

By declaring elements in the body of a package, we can implement encapsulation in Ada. Those elements will only be visible in the package body, but nowhere else. This isn't, however, the only way to achieve encapsulation in Ada: we'll discuss other approaches in the *Privacy* (page 109) chapter.

4.5 Renaming

Previously, we've mentioned that *subprograms can be renamed* (page 27). We can rename packages, too. Again, we use the **renames** keyword for that. The following example renames the Ada.Text_IO package as TIO:

Listing 23: main.adb

```
with Ada.Text_IO;

procedure Main is
    package TIO renames Ada.Text_IO;
begin
    TIO.Put_Line ("Hello");
end Main;
```

Code block metadata

4.5. Renaming 41

```
Project: Courses.Intro_To_Ada.Modular_Programming.Rename_Text_IO MD5: 33652dd004ef33d95c168ab8893cd412
```

Runtime output

```
Hello
```

We can use renaming to improve the readability of our code by using shorter package names. In the example above, we write TIO.Put_Line instead of the longer version (Ada. Text_IO.Put_Line). This approach is especially useful when we don't **use** packages and want to avoid that the code becomes too verbose.

Note we can also rename subprograms and objects inside packages. For instance, we could have just renamed the Put_Line procedure in the source code example above:

Listing 24: main.adb

```
with Ada.Text_IO;

procedure Main is
procedure Say (Something : String)
renames Ada.Text_IO.Put_Line;
begin
Say ("Hello");
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Modular_Programming.Rename_Put_Line MD5: f30174ff29eb01f33bc95f1787f9f1dc
```

Runtime output

```
Hello
```

In this example, we rename the Put_Line procedure to Say.

STRONGLY TYPED LANGUAGE

Ada is a strongly typed language. It is interestingly modern in that respect: strong static typing has been increasing in popularity in programming language design, owing to factors such as the growth of statically typed functional programming, a big push from the research community in the typing domain, and many practical languages with strong type systems.

5.1 What is a type?

In statically typed languages, a type is mainly (but not only) a *compile time* construct. It is a construct to enforce invariants about the behavior of a program. Invariants are unchangeable properties that hold for all variables of a given type. Enforcing them ensures, for example, that variables of a data type never have invalid values.

A type is used to reason about the *objects* a program manipulates (an object is a variable or a constant). The aim is to classify objects by what you can accomplish with them (i.e., the operations that are permitted), and this way you can reason about the correctness of the objects' values.

5.2 Integers

A nice feature of Ada is that you can define your own integer types, based on the requirements of your program (i.e., the range of values that makes sense). In fact, the definitional mechanism that Ada provides forms the semantic basis for the predefined integer types. There is no "magical" built-in type in that regard, which is unlike most languages, and arguably very elegant.

Listing 1: integer_type_example.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Integer Type Example is
      -- Declare a signed integer type,
      -- and give the bounds
      type My_Int is range -1 .. 20;
6
                                   ^ High bound
7
                             ^ Low bound
8
9
       -- Like variables, type declarations can
10
          only appear in declarative regions.
11
   begin
12
      for I in My Int loop
13
         Put_Line (My_Int'Image (I));
14
                          ^ 'Image attribute
15
                             converts a value
16
                             to a String.
17
```

```
end loop;
end Integer_Type_Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Integer_Type_Example MD5: 1d82fa54b604944fdd8652cbf84f4ff2
```

Runtime output

```
- 1
 0
 1
 2
 3
 4
 5
 6
 7
 8
 9
 10
 11
 12
 13
 14
 15
 16
 17
 18
 19
 20
```

This example illustrates the declaration of a signed integer type, and several things we can do with them.

Every type declaration in Ada starts with the **type** keyword (except for *task types* (page 156)). After the type, we can see a range that looks a lot like the ranges that we use in **for** loops, that defines the low and high bound of the type. Every integer in the inclusive range of the bounds is a valid value for the type.

Ada integer types

In Ada, an integer type is not specified in terms of its machine representation, but rather by its range. The compiler will then choose the most appropriate representation.

Another point to note in the above example is the My_Int'Image (I) expression. The Name'Attribute (optional params) notation is used for what is called an attribute in Ada. An attribute is a built-in operation on a type, a value, or some other program entity. It is accessed by using a 'symbol (the ASCII apostrophe).

Ada has several types available as "built-ins"; **Integer** is one of them. Here is how **Integer** might be defined for a typical processor:

```
type Integer is
  range - (2 ** 31) .. +(2 ** 31 - 1);
```

** is the exponent operator, which means that the first valid value for **Integer** is -2^{31} , and the last valid value is $2^{31} - 1$.

Ada does not mandate the range of the built-in type **Integer**. An implementation for a 16-bit target would likely choose the range -2^{15} through 2^{15} - 1.

5.2.1 Operational semantics

Unlike some other languages, Ada requires that operations on integers should be checked for overflow.

Listing 2: main.adb

```
procedure Main is
A : Integer := Integer'Last;
B : Integer;
begin
B := A + 5;
-- This operation will overflow, eg. it
-- will raise an exception at run time.
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Overflow_Check MD5: bddd15b394f043442024899d12b982fb
```

Build output

```
main.adb:5:11: warning: value not in range of type "Standard.Integer" [enabled by default]
main.adb:5:11: warning: Constraint_Error will be raised at run time [enabled by default]
```

Runtime output

```
raised CONSTRAINT_ERROR : main.adb:5 overflow check failed
```

There are two types of overflow checks:

- Machine-level overflow, when the result of an operation exceeds the maximum value (or is less than the minimum value) that can be represented in the storage reserved for an object of the type, and
- Type-level overflow, when the result of an operation is outside the range defined for the type.

Mainly for efficiency reasons, while machine-level overflow always results in an exception, type-level overflows will only be checked at specific boundaries, like assignment:

Listing 3: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
1
2
   procedure Main is
3
      type My_Int is range 1 .. 20;
4
      A : My_Int := 12;
5
      B : My_Int := 15;
6
      M : My_Int := (A + B) / 2;
      -- No overflow here, overflow checks
8
       -- are done at specific boundaries.
9
   begin
10
      for I in 1 .. M loop
11
         Put_Line ("Hello, World!");
12
```

5.2. Integers 45

```
end loop;
-- Loop body executed 13 times
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Overflow_Check_2
MD5: d24283cbb42c0be5b5fa215eb16ad2e7
```

Runtime output

```
Hello, World!
```

Type-level overflow will only be checked at specific points in the execution. The result, as we see above, is that you might have an operation that overflows in an intermediate computation, but no exception will be raised because the final result does not overflow.

5.3 Unsigned types

Ada also features unsigned Integer types. They're called *modular* types in Ada parlance. The reason for this designation is due to their behavior in case of overflow: They simply "wrap around", as if a modulo operation was applied.

For machine sized modular types, for example a modulus of 2^{32} , this mimics the most common implementation behavior of unsigned types. However, an advantage of Ada is that the modulus is more general:

Listing 4: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Main is
3
      type Mod_Int is mod 2 ** 5;
4
                       ^ Range is 0 .. 31
5
6
      A : constant Mod Int := 20;
      B : constant Mod_Int := 15;
9
      M : constant Mod Int := A + B;
10
      -- No overflow here,
11
      -- M = (20 + 15) \mod 32 = 3
12
   beain
13
      for I in 1 .. M loop
14
         Put Line ("Hello, World!");
15
      end loop;
16
  end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Unsigned_Types
MD5: df4efee4eb29e7ea15a0cf961b600dd5
```

Runtime output

```
Hello, World!
Hello, World!
Hello, World!
```

Unlike in C/C++, since this wraparound behavior is guaranteed by the Ada specification, you can rely on it to implement portable code. Also, being able to leverage the wrapping on arbitrary bounds is very useful — the modulus does not need to be a power of 2 — to implement certain algorithms and data structures, such as ring buffers¹⁰.

5.4 Enumerations

Enumeration types are another nicety of Ada's type system. Unlike C's enums, they are *not* integers, and each new enumeration type is incompatible with other enumeration types. Enumeration types are part of the bigger family of discrete types, which makes them usable in certain situations that we will describe later but one context that we have already seen is a case statement.

Listing 5: enumeration_example.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Enumeration Example is
3
       type Days is (Monday, Tuesday, Wednesday,
4
                     Thursday, Friday,
5
                     Saturday, Sunday);
6
          An enumeration type
   begin
      for I in Days loop
9
         case I is
10
             when Saturday .. Sunday =>
11
                Put_Line ("Week end!");
12
13
             when Monday .. Friday =>
14
                Put Line ("Hello on "
15
                          & Days'Image (I));
16
                    'Image attribute, works on
17
                    enums too
18
19
          end case;
      end loop;
20
   end Enumeration_Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Enumeration_Example MD5: 45d6c83992af4fb6d5015d5f22cb7113
```

Runtime output

```
Hello on MONDAY
Hello on TUESDAY
Hello on WEDNESDAY
Hello on THURSDAY
Hello on FRIDAY

(continues on next page)
```

5.4. Enumerations 47

¹⁰ https://en.wikipedia.org/wiki/Circular_buffer

```
Week end!
Week end!
```

Enumeration types are powerful enough that, unlike in most languages, they're used to define the standard Boolean type:

```
type Boolean is (False, True);
```

As mentioned previously, every "built-in" type in Ada is defined with facilities generally available to the user.

5.5 Floating-point types

5.5.1 Basic properties

Like most languages, Ada supports floating-point types. The most commonly used floating-point type is **Float**:

Listing 6: floating_point_demo.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Floating_Point_Demo is
    A : constant Float := 2.5;
begin
    Put_Line ("The value of A is "
    & Float'Image (A));
end Floating_Point_Demo;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Floating_Point_Demo MD5: 06998775497b68b742700138faecbb6a
```

Runtime output

```
The value of A is 2.50000E+00
```

The application will display 2.5 as the value of A.

The Ada language does not specify the precision (number of decimal digits in the mantissa) for Float; on a typical 32-bit machine the precision will be 6.

All common operations that could be expected for floating-point types are available, including absolute value and exponentiation. For example:

Listing 7: floating_point_operations.adb

```
6 Float'Image (A));
end Floating_Point_Operations;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Floating_Point_Operations MD5: c280e0f23e020aaee1a8777e7fb4c242
```

Runtime output

```
The value of A is 2.00000E+00
The value of A is 5.00000E+00
```

The value of A is 2.0 after the first operation and 5.0 after the second operation.

In addition to **Float**, an Ada implementation may offer data types with higher precision such as **Long_Float** and **Long_Long_Float**. Like **Float**, the standard does not indicate the exact precision of these types: it only guarantees that the type **Long_Float**, for example, has at least the precision of **Float**. In order to guarantee that a certain precision requirement is met, we can define custom floating-point types, as we will see in the next section.

5.5.2 Precision of floating-point types

Ada allows the user to specify the precision for a floating-point type, expressed in terms of decimal digits. Operations on these custom types will then have at least the specified precision. The syntax for a simple floating-point type declaration is:

```
type T is digits <number_of_decimal_digits>;
```

The compiler will choose a floating-point representation that supports the required precision. For example:

Listing 8: custom_floating_types.adb

```
with Ada. Text IO; use Ada. Text IO;
2
   procedure Custom_Floating_Types is
3
       type T3 is digits 3;
4
       type T15 is digits 15;
5
       type T18 is digits 18;
6
   begin
7
      Put Line ("T3 requires "
8
                 & Integer'Image (T3'Size)
                 & " bits");
10
      Put Line ("T15 requires "
11
                 & Integer'Image (T15'Size)
12
                 & " bits");
13
      Put_Line ("T18 requires "
14
                 & Integer'Image (T18'Size)
15
                 & " bits");
16
   end Custom Floating Types;
17
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Custom_Floating_Types
MD5: 3c23738f13e081038996c533da8fb723
```

Runtime output

```
T3 requires 32 bits
T15 requires 64 bits
T18 requires 128 bits
```

In this example, the attribute 'Size is used to retrieve the number of bits used for the specified data type. As we can see by running this example, the compiler allocates 32 bits for T3, 64 bits for T15 and 128 bits for T18. This includes both the mantissa and the exponent.

The number of digits specified in the data type is also used in the format when displaying floating-point variables. For example:

Listing 9: display_custom_floating_types.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Display_Custom_Floating_Types is
3
      type T3 is digits 3;
4
      type T18 is digits 18;
5
6
      C1 : constant := 1.0e-4;
7
8
      A : constant T3 := 1.0 + C1;
9
      B : constant T18 := 1.0 + C1;
10
11
   begin
      Put Line ("The value of A is "
12
                 & T3'Image (A));
13
      Put_Line ("The value of B is "
14
                 & T18'Image (B));
15
   end Display_Custom_Floating_Types;
16
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Display_Custom_Floating_Types MD5: 58ec2660388a7f05el39f73e94303cf1
```

Runtime output

```
The value of A is 1.00E+00
The value of B is 1.000100000000000E+00
```

As expected, the application will display the variables according to specified precision (1.00E+00 and 1.000100000000000E+00).

5.5.3 Range of floating-point types

In addition to the precision, a range can also be specified for a floating-point type. The syntax is similar to the one used for integer data types — using the **range** keyword. This simple example creates a new floating-point type based on the type **Float**, for a normalized range between -1.0 and 1.0:

Listing 10: floating point range.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Floating_Point_Range is
    type T_Norm is new Float range -1.0 .. 1.0;
    A : T_Norm;

begin
    A := 1.0;
    Put_Line ("The value of A is "
```

```
% T_Norm'Image (A));
end Floating_Point_Range;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Floating_Point_Range MD5: b43d596682aa0fa11124a3a3d0596abc
```

Runtime output

```
The value of A is 1.00000E+00
```

The application is responsible for ensuring that variables of this type stay within this range; otherwise an exception is raised. In this example, the exception Constraint_Error is raised when assigning 2.0 to the variable A:

Listing 11: floating point range exception.adb

Code block metadata

Build output

```
floating_point_range_exception.adb:7:09: warning: value not in range of type "T_
    Norm" defined at line 4 [enabled by default]
floating_point_range_exception.adb:7:09: warning: Constraint_Error will be raised_
    at run time [enabled by default]
```

Runtime output

```
raised CONSTRAINT_ERROR : floating_point_range_exception.adb:7 range check failed
```

Ranges can also be specified for custom floating-point types. For example:

Listing 12: custom_range_types.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Ada.Numerics; use Ada.Numerics;

procedure Custom_Range_Types is
type T6_Inv_Trig is
digits 6 range -Pi / 2.0 .. Pi / 2.0;
begin
null;
end Custom_Range_Types;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Custom_Range_Types MD5: 7b62abc869290a30e351163f670059e0
```

In this example, we are defining a type called T6_Inv_Trig, which has a range from - π / 2 to π / 2 with a minimum precision of 6 digits. (Pi is defined in the predefined package Ada.Numerics.)

5.6 Strong typing

As noted earlier, Ada is strongly typed. As a result, different types of the same family are incompatible with each other; a value of one type cannot be assigned to a variable from the other type. For example:

Listing 13: illegal example.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Illegal Example is
      -- Declare two different floating point types
      type Meters is new Float;
      type Miles is new Float;
      Dist_Imperial : Miles;
8
9
      -- Declare a constant
10
      Dist Metric : constant Meters := 1000.0;
11
12
         Not correct: types mismatch
13
      Dist_Imperial := Dist_Metric * 621.371e-6;
14
      Put Line (Miles'Image (Dist_Imperial));
15
   end Illegal Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Imperial_Metric_Error MD5: e28e341c5eda9b3b4cef691fa24b7f7e
```

Build output

```
illegal_example.adb:14:33: error: expected type "Miles" defined at line 6
illegal_example.adb:14:33: error: found type "Meters" defined at line 5
gprbuild: *** compilation phase failed
```

A consequence of these rules is that, in the general case, a "mixed mode" expression like 2 * 3.0 will trigger a compilation error. In a language like C or Python, such expressions are made valid by implicit conversions. In Ada, such conversions must be made explicit:

Listing 14: conv.adb

```
-- Type conversion, from Meters to Miles
-- Now the code is correct

Put_Line (Miles'Image (Dist_Imperial));
end Conv;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Imperial_Metric MD5: e455641e86227e80e5f920b5af6315d4
```

Runtime output

```
6.21371E-01
```

Of course, we probably do not want to write the conversion code every time we convert from meters to miles. The idiomatic Ada way in that case would be to introduce conversion functions along with the types.

Listing 15: conv.adb

```
with Ada.Text_IO; use Ada.Text_IO;
1
2
   procedure Conv is
3
      type Meters is new Float;
4
      type Miles is new Float;
5
6
      -- Function declaration, like procedure
      -- but returns a value.
8
      function To_Miles (M : Meters) return Miles is
9
                                       ^ Return type
10
      begin
11
         return Miles (M) * 621.371e-6;
12
      end To Miles;
13
14
      Dist Imperial : Miles;
15
      Dist Metric : constant Meters := 1000.0;
16
   begin
17
      Dist Imperial := To Miles (Dist Metric);
18
      Put Line (Miles'Image (Dist Imperial));
   end Conv;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Imperial_Metric_Func MD5: 661737fa9f130ac3070210bbf6f08214
```

Runtime output

```
6.21371E-01
```

If you write a lot of numeric code, having to explicitly provide such conversions might seem painful at first. However, this approach brings some advantages. Notably, you can rely on the absence of implicit conversions, which will in turn prevent some subtle errors.

1 In other languages

In C, for example, the rules for implicit conversions may not always be completely obvious. In Ada, however, the code will always do exactly what it seems to do. For example:

```
int a = 3, b = 2;
float f = a / b;
```

This code will compile fine, but the result of f will be 1.0 instead of 1.5, because the compiler will generate an integer division (three divided by two) that results in one. The software developer must be aware of data conversion issues and use an appropriate casting:

```
int a = 3, b = 2;
float f = (float)a / b;
```

In the corrected example, the compiler will convert both variables to their corresponding floating-point representation before performing the division. This will produce the expected result.

This example is very simple, and experienced C developers will probably notice and correct it before it creates bigger problems. However, in more complex applications where the type declaration is not always visible — e.g. when referring to elements of a **struct** — this situation might not always be evident and quickly lead to software defects that can be harder to find.

The Ada compiler, in contrast, will always reject code that mixes floating-point and integer variables without explicit conversion. The following Ada code, based on the erroneous example in C, will not compile:

Listing 16: main.adb

```
procedure Main is
A : Integer := 3;
B : Integer := 2;
F : Float;
begin
F := A / B;
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Implicit_Cast MD5: 38a8fcc6608c22e22940052ab8dd62f4
```

Build output

```
main.adb:6:11: error: expected type "Standard.Float"
main.adb:6:11: error: found type "Standard.Integer"
gprbuild: *** compilation phase failed
```

The offending line must be changed to F := Float (A) / Float (B); in order to be accepted by the compiler.

You can use Ada's strong typing to help enforce invariants in your code, as in the example above: Since Miles and Meters are two different types, you cannot mistakenly convert an instance of one to an instance of the other.

5.7 Derived types

In Ada you can create new types based on existing ones. This is very useful: you get a type that has the same properties as some existing type but is treated as a distinct type in the interest of strong typing.

Listing 17: main.adb

```
procedure Main is
1
      -- ID card number type,
2
       -- incompatible with Integer.
3
      type Social_Security_Number is new Integer
       range 0 .. 999_99_9999;
               ^ Since a SSN has 9 digits
                 max., and cannot be
      - -
                 negative, we enforce
8
       - -
                 a validity constraint.
9
10
      SSN : Social Security Number :=
11
        555_55_5555;
12
            You can put underscores as
13
              formatting in any number.
14
15
      I : Integer;
16
17
       -- The value -1 below will cause a
18
       -- runtime error and a compile time
19
       -- warning with GNAT.
20
      Invalid : Social_Security_Number := -1;
21
22
       -- Illegal, they have different types:
23
      I := SSN;
24
25
      -- Likewise illegal:
26
      SSN := I;
27
28
      -- OK with explicit conversion:
29
      I := Integer (SSN);
30
31
          Likewise OK:
32
      SSN := Social_Security_Number (I);
33
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Derived_Types
MD5: 63445601ddb5e52dceab095d3305623a
```

Build output

```
main.adb:21:40: warning: value not in range of type "Social_Security_Number" defined at line 4 [enabled by default]
main.adb:21:40: warning: Constraint_Error will be raised at run time [enabled by default]
main.adb:24:09: error: expected type "Standard.Integer"
main.adb:24:09: error: found type "Social_Security_Number" defined at line 4
main.adb:27:11: error: expected type "Social_Security_Number" defined at line 4
main.adb:27:11: error: found type "Standard.Integer"
gprbuild: *** compilation phase failed
```

The type Social_Security is said to be a derived type; its parent type is **Integer**.

As illustrated in this example, you can refine the valid range when defining a derived scalar type (such as integer, floating-point and enumeration).

The syntax for enumerations uses the **range** < **range** > syntax:

Listing 18: greet.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Greet is
3
      type Days is (Monday, Tuesday, Wednesday,
4
                      Thursday, Friday,
5
                     Saturday, Sunday);
6
7
      type Weekend Days is new
8
        Days range Saturday .. Sunday;
9
      -- New type, where only Saturday and Sunday
10
         are valid literals.
11
   begin
12
      null:
13
   end Greet;
14
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Days
MD5: 853b5c1576961c7c20d4306275122364
```

5.8 Subtypes

As we are starting to see, types may be used in Ada to enforce constraints on the valid range of values. However, we sometimes want to enforce constraints on some values while staying within a single type. This is where subtypes come into play. A subtype does not introduce a new type.

Listing 19: greet.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Greet is
3
      type Days is (Monday, Tuesday, Wednesday,
4
                     Thursday, Friday,
5
                     Saturday, Sunday);
6
      -- Declaration of a subtype
8
      subtype Weekend_Days is
9
        Days range Saturday .. Sunday;
10
              ^ Constraint of the subtype
11
12
      M : Days := Sunday;
13
14
      S : Weekend_Days := M;
15
      -- No error here, Days and Weekend_Days
16
          are of the same type.
17
   begin
18
      for I in Days loop
19
         case I is
20
                Just like a type, a subtype can
21
             -- be used as a range
22
             when Weekend_Days =>
23
               Put Line ("Week end!");
24
             when others =>
25
                Put_Line ("Hello on "
26
                           & Days'Image (I));
27
         end case;
28
```

```
end loop;
end Greet;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Days_Subtype
MD5: 8ee7127d152a8b2c9d0ac74d05fc2fc2
```

Runtime output

```
Hello on MONDAY
Hello on TUESDAY
Hello on WEDNESDAY
Hello on THURSDAY
Hello on FRIDAY
Week end!
Week end!
```

Several subtypes are predefined in the standard package in Ada, and are automatically available to you:

```
subtype Natural is Integer range 0 .. Integer'Last;
subtype Positive is Integer range 1 .. Integer'Last;
```

While subtypes of a type are statically compatible with each other, constraints are enforced at run time: if you violate a subtype constraint, an exception will be raised.

Listing 20: greet.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Greet is
      type Days is (Monday, Tuesday, Wednesday,
4
                     Thursday, Friday,
5
                     Saturday, Sunday);
6
      subtype Weekend Days is
8
        Days range Saturday .. Sunday;
10
               : Days := Saturday;
11
      Weekend : Weekend_Days;
12
   begin
13
      Weekend := Day;
14
                  ^ Correct: Same type, subtype
15
                    constraints are respected
16
      Weekend := Monday;
17
                  ^ Wrong value for the subtype
18
                    Compiles, but exception at runtime
19
   end Greet;
20
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Strongly_Typed_Language.Days_Subtype_Error MD5: 84d42d276d26544f35edab5870459378
```

Build output

```
greet.adb:17:15: warning: value not in range of type "Weekend_Days" defined at line 8 [enabled by default] greet.adb:17:15: warning: Constraint_Error will be raised at run time [enabled by default]
```

5.8. Subtypes 57

Runtime output

```
raised CONSTRAINT_ERROR : greet.adb:17 range check failed
```

5.8.1 Subtypes as type aliases

Previously, we've seen that we can create new types by declaring e.g. **type Miles is new Float**. We could also create type aliases, which generate alternative names — *aliases* — for known types. Note that type aliases are sometimes called *type synonyms*.

We achieve this in Ada by using subtypes without new constraints. In this case, however, we don't get all of the benefits of Ada's strong type checking. Let's rewrite an example using type aliases:

Listing 21: undetected_imperial_metric_error.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Undetected_Imperial_Metric_Error is
3
      -- Declare two type aliases
4
      subtype Meters is Float;
5
      subtype Miles is Float;
6
7
      Dist Imperial : Miles;
8
9
      -- Declare a constant
10
      Dist Metric : constant Meters := 100.0;
11
   beain
12
      -- No conversion to Miles type required:
13
      Dist_Imperial := (Dist_Metric * 1609.0)
14
                          / 1000.0;
15
16
      -- Not correct, but undetected:
17
      Dist_Imperial := Dist_Metric;
18
19
      Put_Line (Miles'Image (Dist_Imperial));
20
   end Undetected_Imperial_Metric_Error;
```

Code block metadata

Runtime output

```
1.00000E+02
```

In the example above, the fact that both Meters and Miles are subtypes of **Float** allows us to mix variables of both types without type conversion. This, however, can lead to all sorts of programming mistakes that we'd like to avoid, as we can see in the undetected error highlighted in the code above. In that example, the error in the assignment of a value in meters to a variable meant to store values in miles remains undetected because both Meters and Miles are subtypes of **Float**. Therefore, the recommendation is to use strong typing — via **type X is new** Y — for cases such as the one above.

There are, however, many situations where type aliases are useful. For example, in an application that uses floating-point types in multiple contexts, we could use type aliases to indicate additional meaning to the types or to avoid long variable names. For example, instead of writing:

```
Paid_Amount, Due_Amount : Float;
```

We could write:

```
subtype Amount is Float;
Paid, Due : Amount;
```

1 In other languages

In C, for example, we can use a **typedef** declaration to create a type alias. For example:

```
typedef float meters;
```

This corresponds to the declaration that we've seen above using subtypes. Other programming languages include this concept in similar ways. For example:

- C++: using meters = float;
- Swift: typealias Meters = Double
- Kotlin: typealias Meters = Double
- Haskell: type Meters = Float

Note, however, that subtypes in Ada correspond to type aliases if, and only if, they don't have new constraints. Thus, if we add a new constraint to a subtype declaration, we don't have a type alias anymore. For example, the following declaration *can't* be considered a type alias of **Float**:

```
subtype Meters is Float range 0.0 .. 1_000_000.0;
```

Let's look at another example:

```
subtype Degree_Celsius is Float;
subtype Liquid_Water_Temperature is
  Degree_Celsius range 0.0 .. 100.0;
subtype Running_Water_Temperature is
  Liquid_Water_Temperature;
```

In this example, Liquid_Water_Temperature isn't an alias of Degree_Celsius, since it adds a new constraint that wasn't part of the declaration of the Degree_Celsius. However, we do have two type aliases here:

- Degree_Celsius is an alias of Float;
- Running_Water_Temperature is an alias of Liquid_Water_Temperature, even if Liquid_Water_Temperature itself has a constrained range.

5.8. Subtypes 59

RECORDS

So far, all the types we have encountered have values that are not decomposable: each instance represents a single piece of data. Now we are going to see our first class of composite types: records.

Records allow composing a value out of instances of other types. Each of those instances will be given a name. The pair consisting of a name and an instance of a specific type is called a field, or a component.

6.1 Record type declaration

Here is an example of a simple record declaration:

```
type Date is record
    -- The following declarations are
    -- components of the record
Day : Integer range 1 .. 31;
Month : Months;
    -- You can add custom constraints
    -- on fields
Year : Integer range 1 .. 3000;
end record;
```

Fields look a lot like variable declarations, except that they are inside of a record definition. And as with variable declarations, you can specify additional constraints when supplying the subtype of the field.

Record components can have default values. When a variable having the record type is declared, a field with a default initialization will be automatically set to this value. The value can be any expression of the component type, and may be run-time computable.

In the remaining sections of this chapter, we see how to use record types. In addition to that, we discuss more about records in *another chapter* (page 97).

6.2 Aggregates

Records have a convenient notation for expressing values, illustrated above. This notation is called aggregate notation, and the literals are called aggregates. They can be used in a variety of contexts that we will see throughout the course, one of which is to initialize records.

An aggregate is a list of values separated by commas and enclosed in parentheses. It is allowed in any context where a value of the record is expected.

Values for the components can be specified positionally, as in Ada_Birthday example, or by name, as in Leap_Day_2020. A mixture of positional and named values is permitted, but you cannot use a positional notation after a named one.

6.3 Component selection

To access components of a record instance, you use an operation that is called component selection. This is achieved by using the dot notation. For example, if we declare a variable Some_Day of the Date record type mentioned above, we can access the Year component by writing Some_Day.Year.

Let's look at an example:

Listing 1: record selection.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Record_Selection is
3
      type Months is
5
         (January, February, March, April,
6
         May, June, July, August, September,
7
         October, November, December);
8
      type Date is record
10
         Day : Integer range 1 .. 31;
11
         Month: Months;
12
         Year : Integer range 1 .. 3000 := 2032;
13
      end record;
14
15
      procedure Display_Date (D : Date) is
16
      begin
17
         Put_Line ("Day:" & Integer'Image (D.Day)
18
                    & ", Month: '
19
                    & Months'Image (D.Month)
20
                    & ", Year:"
                    & Integer'Image (D.Year));
22
      end Display_Date;
23
24
      Some_Day : Date := (1, January, 2000);
25
26
```

```
begin
Display_Date (Some_Day);

Put_Line ("Changing year...");
Some_Day.Year := 2001;

Display_Date (Some_Day);
end Record_Selection;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Records.Record_Selection
MD5: 79602cf4d011ba7423d07772b13e2b5a
```

Runtime output

```
Day: 1, Month: JANUARY, Year: 2000
Changing year...
Day: 1, Month: JANUARY, Year: 2001
```

As you can see in this example, we can use the dot notation in the expression D.Year or Some_Day.Year to access the information stored in that component, as well as to modify this information in assignments. To be more specific, when we use D.Year in the call to Put_Line, we're retrieving the information stored in that component. When we write Some_Day.Year := 2001, we're overwriting the information that was previously stored in the Year component of Some Day.

6.4 Renaming

In previous chapters, we've discussed *subprogram* (page 27) and *package* (page 41) renaming. We can rename record components as well. Instead of writing the full component selection using the dot notation, we can declare an alias that allows us to access the same component. This is useful to simplify the implementation of a subprogram, for example.

We can rename record components by using the **renames** keyword in a variable declaration. For example:

```
Some_Day : Date;
Y : Integer renames Some_Day.Year;
```

Here, Y is an alias, so that every time we using Y, we are really using the Year component of Some Day.

Let's look at a complete example:

Listing 2: dates.ads

```
package Dates is
1
      type Months is
3
        (January, February, March, April,
         May, June, July, August, September,
         October, November, December);
      type Date is record
8
         Day : Integer range 1 .. 31;
9
         Month: Months;
10
         Year : Integer range 1 .. 3000 := 2032;
11
      end record;
12
                                                                        (continues on next page)
```

6.4. Renaming 63

```
procedure Increase_Month
(Some_Day : in out Date);

procedure Display_Month
(Some_Day : Date);

end Dates;
```

Listing 3: dates.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   package body Dates is
3
      procedure Increase Month
5
         (Some Day : in out Date)
6
7
          -- Renaming components from
          -- the Date record
9
          M : Months renames Some_Day.Month;
10
          Y : Integer renames Some_Day.Year;
11
12
          -- Renaming function (for Months
13
          -- enumeration)
14
          function Next (M : Months)
15
                          return Months
16
            renames Months'Succ;
17
      begin
18
          if M = December then
19
             M := January;
20
             Y := Y + 1;
21
          else
22
             M := Next (M);
23
          end if;
24
      end Increase_Month;
25
26
      procedure Display Month
27
         (Some_Day : Date)
28
29
          -- Renaming components from
30
          -- the Date record
31
          M : Months renames Some_Day.Month;
32
          Y : Integer renames Some_Day.Year;
33
      begin
34
          Put_Line ("Month: "
35
                    & Months'Image (M)
36
                    & ", Year:"
37
                    & Integer'Image (Y));
38
      end Display_Month;
39
40
   end Dates;
```

Listing 4: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Dates; use Dates;

procedure Main is
D: Date := (1, January, 2000);
```

```
begin
Display_Month (D);

Put_Line ("Increasing month...");
Increase_Month (D);

Display_Month (D);

end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Record_Component_Renaming
MD5: 905390bd02b8417039052218800975a3
```

Runtime output

```
Month: JANUARY, Year: 2000
Increasing month...
Month: FEBRUARY, Year: 2000
```

We apply renaming to two components of the Date record in the implementation of the Increase_Month procedure. Then, instead of directly using Some_Day.Month and Some_Day. Year in the next operations, we simply use the renamed versions M and Y.

Note that, in the example above, we also rename Months 'Succ — which is the function that gives us the next month — to Next.

6.4. Renaming 65

SEVEN

ARRAYS

Arrays provide another fundamental family of composite types in Ada.

7.1 Array type declaration

Arrays in Ada are used to define contiguous collections of elements that can be selected by indexing. Here's a simple example:

Listing 1: greet.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Greet is
       type My_Int is range 0 .. 1000;
       type Index is range 1 .. 5;
5
6
       type My_Int_Array is
7
        array (Index) of My_Int;
8
                            \overline{T}ype of elements
9
                ^ Bounds of the array
10
      Arr : My_Int_Array := (2, 3, 5, 7, 11);
11
                               ` Array literal
12
                                 (aggregate)
13
14
      V : My_Int;
15
   begin
16
       for I in Index loop
17
          V := Arr(I);
18
                     ^ Take the Ith element
19
          Put (My_Int'Image (V));
20
       end loop;
21
      New_Line;
   end Greet;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Greet
MD5: ffdd2ba2322b0946dfcac3a55bce5270
```

Runtime output

```
2 3 5 7 11
```

The first point to note is that we specify the index type for the array, rather than its size. Here we declared an integer type named Index ranging from 1 to 5, so each array instance will have 5 elements, with the initial element at index 1 and the last element at index 5.

Although this example used an integer type for the index, Ada is more general: any discrete type is permitted to index an array, including *Enum types* (page 47). We will soon see what that means.

Another point to note is that querying an element of the array at a given index uses the same syntax as for function calls: that is, the array object followed by the index in parentheses.

Thus when you see an expression such as A (B), whether it is a function call or an array subscript depends on what A refers to.

Finally, notice how we initialize the array with the (2, 3, 5, 7, 11) expression. This is another kind of aggregate in Ada, and is in a sense a literal expression for an array, in the same way that 3 is a literal expression for an integer. The notation is very powerful, with a number of properties that we will introduce later. A detailed overview appears in the notation of aggregate types (page 85).

Unrelated to arrays, the example also illustrated two procedures from Ada. Text IO:

- · Put, which displays a string without a terminating end of line
- New Line, which outputs an end of line

Let's now delve into what it means to be able to use any discrete type to index into the array.

1 In other languages

Semantically, an array object in Ada is the entire data structure, and not simply a handle or pointer. Unlike C and C++, there is no implicit equivalence between an array and a pointer to its initial element.

Listing 2: array_bounds_example.adb

```
with Ada. Text IO; use Ada. Text IO;
1
2
   procedure Array_Bounds_Example is
3
      type My Int is range 0 .. 1000;
4
5
      type Index is range 11 .. 15;
6
                             ^ Low bound can
7
                              be any value
8
9
      type My Int Array is
10
        array (Index) of My Int;
11
12
      Tab : constant My_Int_Array :=
13
               (2, 3, 5, 7, 11);
14
15
       for I in Index loop
16
         Put (My Int'Image (Tab (I)));
17
      end loop;
18
      New Line;
19
  end Array Bounds Example;
20
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Array_Bounds_Example
MD5: e5fe9e7b83055f3ae23dd890e29c22de
```

Runtime output

```
2 3 5 7 11
```

One effect is that the bounds of an array can be any values. In the first example we constructed an array type whose first index is 1, but in the example above we declare an array type whose first index is 11.

That's perfectly fine in Ada, and moreover since we use the index type as a range to iterate over the array indices, the code using the array does not need to change.

That leads us to an important consequence with regard to code dealing with arrays. Since the bounds can vary, you should not assume / hard-code specific bounds when iterating / using arrays. That means the code above is good, because it uses the index type, but a for loop as shown below is bad practice even though it works correctly:

```
for I in 11 .. 15 loop
    Tab (I) := Tab (I) * 2;
end loop;
```

Since you can use any discrete type to index an array, enumeration types are permitted.

Listing 3: month_example.adb

```
with Ada. Text IO; use Ada. Text IO;
2
   procedure Month Example is
3
      type Month_Duration is range 1 .. 31;
4
      type Month is (Jan, Feb, Mar, Apr,
5
                      May, Jun, Jul, Aug,
6
                       Sep, Oct, Nov, Dec);
7
8
9
      type My Int Array is
        array (Month) of Month Duration;
10
                ^ Can use an enumeration type
11
                 as the index
12
13
      Tab : constant My Int Array :=
14
             ^ constant is like a variable but
15
              cannot be modified
16
        (31, 28, 31, 30, 31, 30,
17
         31, 31, 30, 31, 30, 31);
18
       -- Maps months to number of days
19
       -- (ignoring leap years)
20
21
      Feb Days : Month Duration := Tab (Feb);
22
       -- Number of days in February
23
   beain
24
      for M in Month loop
25
         Put Line
26
            (Month'Image (M) & " has "
27
             & Month Duration'Image (Tab (M))
28
             & " days.");
29
            ^ Concatenation operator
30
      end loop;
31
   end Month Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Month_Example
MD5: 420bb8faa36d0efd3d071c76c2033d21
```

Runtime output

```
JAN has 31 days.
FEB has 28 days.
MAR has 31 days.

(continues on next page)
```

```
APR has 30 days.

MAY has 31 days.

JUN has 30 days.

JUL has 31 days.

AUG has 31 days.

SEP has 30 days.

OCT has 31 days.

NOV has 30 days.

DEC has 31 days.
```

In the example above, we are:

- Creating an array type mapping months to month durations in days.
- Creating an array, and instantiating it with an aggregate mapping months to their actual durations in days.
- Iterating over the array, printing out the months, and the number of days for each.

Being able to use enumeration values as indices is very helpful in creating mappings such as shown above one, and is an often used feature in Ada.

7.2 Indexing

We have already seen the syntax for selecting elements of an array. There are however a few more points to note.

First, as is true in general in Ada, the indexing operation is strongly typed. If you use a value of the wrong type to index the array, you will get a compile-time error.

Listing 4: greet.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Greet is
3
      type My_Int is range 0 .. 1000;
4
5
      type My Index
                       is range 1 .. 5;
6
      type Your_Index is range 1 .. 5;
      type My_Int_Array is
9
        array (My_Index) of My_Int;
10
11
      Tab : My_Int_Array := (2, 3, 5, 7, 11);
12
   begin
13
      for I in Your Index loop
14
         Put (My Int'Image (Tab (I)));
15
                                   ^ Compile time error
16
      end loop;
17
      New Line;
18
   end Greet;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Greet_2
MD5: 54543017e4ec69d24bf9e43d507b50e6
```

Build output

```
greet.adb:15:31: error: expected type "My_Index" defined at line 6
greet.adb:15:31: error: found type "Your_Index" defined at line 7
gprbuild: *** compilation phase failed
```

Second, arrays in Ada are bounds checked. This means that if you try to access an element outside of the bounds of the array, you will get a run-time error instead of accessing random memory as in unsafe languages.

Listing 5: greet.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Greet is
3
      type My_Int is range 0 .. 1000;
4
      type Index is range 1 .. 5;
5
6
      type My Int Array is
7
        array (Index) of My Int;
8
      Tab : My_Int_Array := (2, 3, 5, 7, 11);
10
11
      for I in Index range 2 .. 6 loop
12
         Put (My_Int'Image (Tab (I)));
13
                                    ^ Will raise an
14
                                      exception when
          - -
15
                                      I = 6
16
      end loop;
17
      New Line;
18
   end Greet;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Greet_3
MD5: 0102674d089be838f1dfbf0791d99fce
```

Build output

```
greet.adb:12:30: warning: static value out of range of type "Index" defined at line 5 [enabled by default]
greet.adb:12:30: warning: Constraint_Error will be raised at run time [enabled by default]
greet.adb:12:30: warning: suspicious loop bound out of range of loop subtype [enabled by default]
greet.adb:12:30: warning: loop executes zero times or raises Constraint_Error [enabled by default]
```

Runtime output

```
raised CONSTRAINT_ERROR : greet.adb:12 range check failed
```

7.3 Simpler array declarations

In the previous examples, we have always explicitly created an index type for the array. While this can be useful for typing and readability purposes, sometimes you simply want to express a range of values. Ada allows you to do that, too.

Listing 6: simple_array_bounds.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Simple_Array_Bounds is
3
      type My_Int is range 0 .. 1000;
4
5
      type My_Int_Array is
6
        array (1 .. 5) of My_Int;
7
                ^ Subtype of Integer
8
9
      Tab : constant My_Int_Array :=
10
               (2, 3, 5, 7, 11);
11
   begin
12
      for I in 1 .. 5 loop
13
                ^ Subtype of Integer
14
         Put (My_Int'Image (Tab (I)));
15
      end loop;
16
      New Line;
17
   end Simple Array Bounds;
18
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Simple_Array_Bounds
MD5: c337a7fe0dacccc5f60f7b234aa96d39
```

Runtime output

```
2 3 5 7 11
```

This example defines the range of the array via the range syntax, which specifies an anonymous subtype of Integer and uses it to index the array.

This means that the type of the index is **Integer**. Similarly, when you use an anonymous range in a for loop as in the example above, the type of the iteration variable is also **Integer**, so you can use I to index Tab.

You can also use a named subtype for the bounds for an array.

7.4 Range attribute

We noted earlier that hard coding bounds when iterating over an array is a bad idea, and showed how to use the array's index type/subtype to iterate over its range in a **for** loop. That raises the question of how to write an iteration when the array has an anonymous range for its bounds, since there is no name to refer to the range. Ada solves that via several attributes of array objects:

Listing 7: range_example.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Range_Example is
    type My_Int is range 0 .. 1000;

type My_Int_Array is
    array (1 .. 5) of My_Int;

Tab : constant My_Int_Array :=
    (2, 3, 5, 7, 11);

begin

(continues on next page)
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Range_Example
MD5: 8b0d7bf346cb59999dfd12dbaaf3e2a6
```

Runtime output

```
2 3 5 7 11
```

If you want more fine grained control, you can use the separate attributes 'First and 'Last.

Listing 8: array attributes example.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Array_Attributes_Example is
      type My_Int is range 0 .. 1000;
4
5
      type My_Int_Array is
6
        array (1 .. 5) of My_Int;
8
      Tab : My_Int_Array :=
q
               (2, 3, 5, 7, 11);
10
   begin
11
      for I in Tab'First .. Tab'Last - 1 loop
12
                  ^ Iterate on every index
13
                     except the last
14
         Put (My_Int'Image (Tab (I)));
15
      end loop;
16
      New_Line;
17
   end Array_Attributes_Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Array_Attributes_Example
MD5: 95cc407c8aadd936e050fe3505e8fb46
```

Runtime output

```
2 3 5 7
```

The 'Range, 'First and 'Last attributes in these examples could also have been applied to the array type name, and not just the array instances.

Although not illustrated in the above examples, another useful attribute for an array instance A is A'Length, which is the number of elements that A contains.

It is legal and sometimes useful to have a "null array", which contains no elements. To get this effect, define an index range whose upper bound is less than the lower bound.

7.5 Unconstrained arrays

Let's now consider one of the most powerful aspects of Ada's array facility.

Every array type we have defined so far has a fixed size: every instance of this type will have the same bounds and therefore the same number of elements and the same size.

However, Ada also allows you to declare array types whose bounds are not fixed: in that case, the bounds will need to be provided when creating instances of the type.

Listing 9: unconstrained_array_example.adb

```
with Ada.Text_IO; use Ada.Text_IO;
2
   procedure Unconstrained_Array_Example is
3
      type Days is (Monday, Tuesday, Wednesday,
4
                     Thursday, Friday,
5
                     Saturday, Sunday);
6
7
      type Workload Type is
8
        array (Days range <>) of Natural;
9
       -- Indefinite array type
10
               ^ Bounds are of type Days,
11
                 but not known
12
13
      Workload : constant
14
        Workload Type (Monday .. Friday) :=
15
                         ^ Specify the bounds
16
                          when declaring
17
          (Friday => 7, others => 8);
18
                        ^ Default value
19
      -- ^ Specify element by name of index
20
   begin
21
      for I in Workload'Range loop
22
         Put_Line (Integer'Image (Workload (I)));
23
      end loop:
24
   end Unconstrained_Array_Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Unconstrained_Array_Example
MD5: c84910e9b424cfabbbbe018ba0a6de59
```

Runtime output

```
8
8
8
8
7
```

The fact that the bounds of the array are not known is indicated by the Days range <> syntax. Given a discrete type Discrete_Type, if we use Discrete_Type for the index in an array type then Discrete_Type serves as the type of the index and comprises the range of index values for each array instance.

If we define the index as Discrete_Type range <> then Discrete_Type serves as the type of the index, but different array instances may have different bounds from this type.

An array type that is defined with the Discrete_Type range <> syntax for its index is referred to as an unconstrained array type, and, as illustrated above, the bounds need to be provided when an instance is created.

The above example also shows other forms of the aggregate syntax. You can specify associations by name, by giving the value of the index on the left side of an arrow association. $1 \Rightarrow 2$ thus means "assign value 2 to the element at index 1 in my array". **others** $\Rightarrow 8$ means "assign value 8 to every element that wasn't previously assigned in this aggregate".

A Attention

The so-called "box" notation (<>) is commonly used as a wildcard or placeholder in Ada. You will often see it when the meaning is "what is expected here can be anything".

1 In other languages

While unconstrained arrays in Ada might seem similar to variable length arrays in C, they are in reality much more powerful, because they're truly first-class values in the language. You can pass them as parameters to subprograms or return them from functions, and they implicitly contain their bounds as part of their value. This means that it is useless to pass the bounds or length of an array explicitly along with the array, because they are accessible via the 'First, 'Last, 'Range and 'Length attributes explained earlier.

Although different instances of the same unconstrained array type can have different bounds, a specific instance has the same bounds throughout its lifetime. This allows Ada to implement unconstrained arrays efficiently; instances can be stored on the stack and do not require heap allocation as in languages like Java.

7.6 Predefined array type: String

A recurring theme in our introduction to Ada types has been the way important built-in types like **Boolean** or **Integer** are defined through the same facilities that are available to the user. This is also true for strings: The **String** type in Ada is a simple array.

Here is how the string type is defined in Ada:

```
type String is
  array (Positive range <>) of Character;
```

The only built-in feature Ada adds to make strings more ergonomic is custom literals, as we can see in the example below.

String literals are a syntactic sugar for aggregates, so that in the following example, A and B have the same value.

Listing 10: string literals.ads

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.String_Literals
MD5: 8e5871c8ead4ff8da643539857e23b30
```

Listing 11: greet.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Greet is
      Message : String (1 .. 11) := "dlrow olleH";
                ^ Pre-defined array type.
                 Component type is Character
   begin
      for I in reverse Message'Range loop
8
             ^ Iterate in reverse order
9
         Put (Message (I));
10
      end loop:
11
      New Line;
12
  end Greet;
```

However, specifying the bounds of the object explicitly is a bit of a hassle; you have to manually count the number of characters in the literal. Fortunately, Ada gives you an easier way.

You can omit the bounds when creating an instance of an unconstrained array type if you supply an initialization, since the bounds can be deduced from the initialization expression.

Listing 12: greet.adb

```
with Ada.Text_IO; use Ada.Text_IO;
2
   procedure Greet is
3
      Message : constant String := "dlroW olleH";
4
                          ^ Bounds are automatically
5
                             computed from
6
                            initialization value
7
   begin
8
      for I in reverse Message'Range loop
9
        Put (Message (I));
10
      end loop;
11
      New Line;
12
   end Greet;
13
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Greet_5
MD5: 21448a1007a07ec9d434880628625c3f
```

Runtime output

```
Hello World
```

Listing 13: main.adb

Attention

As you can see above, the standard **String** type in Ada is an array. As such, it shares the advantages and drawbacks of arrays: a **String** value is stack allocated, it is accessed efficiently, and its bounds are immutable.

If you want something akin to C++'s std::string, you can use *Unbounded Strings* (page 243) from Ada's standard library. This type is more like a mutable, automatically managed string buffer to which you can add content.

7.7 Restrictions

A very important point about arrays: bounds *have* to be known when instances are created. It is for example illegal to do the following.

```
declare
   A : String;
begin
   A := "World";
end;
```

Also, while you of course can change the values of elements in an array, you cannot change the array's bounds (and therefore its size) after it has been initialized. So this is also illegal:

Also, while you can expect a warning for this kind of error in very simple cases like this one, it is impossible for a compiler to know in the general case if you are assigning a value of the correct length, so this violation will generally result in a run-time error.

4 Attention

While we will learn more about this later, it is important to know that arrays are not the only types whose instances might be of unknown size at compile-time.

Such objects are said to be of an *indefinite subtype*, which means that the subtype size is not known at compile time, but is dynamically computed (at run time).

7.7. Restrictions 77

```
Listing 14: indefinite subtypes.adb
   with Ada.Text_IO; use Ada.Text_IO;
2
   procedure Indefinite_Subtypes is
3
       function Get_Number return Integer is
4
       begin
5
            return Integer'Value (Get_Line);
6
       end Get_Number;
7
8
      A : String := "Hello";
9
          Indefinite subtype
10
11
       B : String (1 .. 5) := "Hello";
12
      -- Definite subtype
13
      C : String (1 .. Get_Number);
15
          Indefinite subtype
16
       -- (Get_Number's value is computed at
17
          run-time)
18
   begin
19
      null;
20
   end Indefinite_Subtypes;
        Code block metadata
        Project: Courses.Intro_To_Ada.Arrays.Indefinite_Subtypes
        MD5: a24235838511a94879f74757421a28f0
   Here, the 'Value attribute converts the string to an integer.
```

7.8 Returning unconstrained arrays

The return type of a function can be any type; a function can return a value whose size is unknown at compile time. Likewise, the parameters can be of any type.

For example, this is a function that returns an unconstrained **String**:

Listing 15: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Main is
      type Days is (Monday, Tuesday, Wednesday,
5
                     Thursday, Friday,
6
                     Saturday, Sunday);
8
      function Get_Day_Name (Day : Days := Monday)
9
                              return String is
10
      begin
11
         return
12
            (case Day is
13
                            => "Monday"
            when Monday
            when Tuesday => "Tuesday";
            when Wednesday => "Wednesday",
16
            when Thursday => "Thursday",
17
            when Friday
                            => "Friday",
18
            when Saturday => "Saturday",
19
            when Sunday
                            => "Sunday");
20
      end Get_Day_Name;
21
```

(continues on next page)

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Day_Name_1
MD5: 0b7c567c723ded52d8e95c4ef46bcecc
```

Runtime output

```
First day is Monday
```

(This example is for illustrative purposes only. There is a built-in mechanism, the 'Image attribute for scalar types, that returns the name (as a **String**) of any element of an enumeration type. For example Days' Image (Monday) is "MONDAY".)

In other languages

Returning variable size objects in languages lacking a garbage collector is a bit complicated implementation-wise, which is why C and C++ don't allow it, preferring to depend on explicit dynamic allocation / free from the user.

The problem is that explicit storage management is unsafe as soon as you want to collect unused memory. Ada's ability to return variable size objects will remove one use case for dynamic allocation, and hence, remove one potential source of bugs from your programs.

Rust follows the C/C++ model, but with safe pointer semantics. However, dynamic allocation is still used. Ada can benefit from a possible performance edge because it can use any model.

7.9 Declaring arrays (2)

While we can have array types whose size and bounds are determined at run time, the array's component type needs to be of a definite and constrained type.

Thus, if you need to declare, for example, an array of strings, the **String** subtype used as component will need to have a fixed size.

Listing 16: show days.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Show_Days is
       type Days is (Monday, Tuesday, Wednesday,
4
                      Thursday, Friday,
5
                      Saturday, Sunday):
6
7
      subtype Day Name is String (1 .. 2);
8
       -- Subtype of string with known size
9
10
      type Days_Name_Type is
11
        array (Days) of Day_Name;
12
                ^{\wedge} Type of the index
13
                           ^ Type of the element.
14
                            Must be definite
15
```

(continues on next page)

```
Names : constant Days_Name_Type :=

("Mo", "Tu", "We", "Th", "Fr", "Sa", "Su");

Initial value given by aggregate

begin

for I in Names'Range loop

Put_Line (Names (I));

end loop;

end Show_Days;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Day_Name_2
MD5: bc66303091c084f66abde72ae59f55a9
```

Runtime output

```
Mo
Tu
We
Th
Fr
Sa
Su
```

7.10 Array slices

One last feature of Ada arrays that we're going to cover is array slices. It is possible to take and use a slice of an array (a contiguous sequence of elements) as a name or a value.

Listing 17: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Main is
3
       Buf : String := "Hello ...";
5
       Full Name : String := "John Smith";
6
   begin
       Buf (7 ... 9) := "Bob";
       -- Careful! This works because the string
9
       -- on the right side is the same length as
10
       -- the replaced slice!
11
12
        -- Prints "Hello Bob"
13
       Put_Line (Buf);
14
15
        -- Prints "Hi John"
16
       Put_Line ("Hi " & Full_Name (1 .. 4));
17
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Slices
MD5: cdf582c6c9089658236f5c79b7be4c3f
```

Runtime output

```
Hello Bob
Hi John
```

As we can see above, you can use a slice on the left side of an assignment, to replace only part of an array.

A slice of an array is of the same type as the array, but has a different subtype, constrained by the bounds of the slice.

A Attention

Ada has multidimensional arrays 11 , which are not covered in this course. Slices will only work on one dimensional arrays.

7.11 Renaming

So far, we've seen that the following elements can be renamed: *subprograms* (page 27), *packages* (page 41), and *record components* (page 63). We can also rename objects by using the **renames** keyword. This allows for creating alternative names for these objects. Let's look at an example:

Listing 18: measurements.ads

```
package Measurements is

subtype Degree_Celsius is Float;

Current_Temperature : Degree_Celsius;

end Measurements;
```

Listing 19: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   with Measurements;
2
3
   procedure Main is
4
      subtype Degrees is
5
        Measurements.Degree Celsius;
6
      T : Degrees
             renames Measurements.Current_Temperature;
9
10
   begin
       T := 5.0;
11
12
       Put Line (Degrees'Image (T));
13
       Put Line (Degrees'Image
14
          (Measurements.Current_Temperature));
15
16
       T := T + 2.5;
17
18
       Put_Line (Degrees'Image (T));
19
       Put_Line (Degrees'Image
20
          (Measurements.Current_Temperature));
21
   end Main:
22
```

Code block metadata

7.11. Renaming

¹¹ http://www.ada-auth.org/standards/12rm/html/RM-3-6.html

```
Project: Courses.Intro_To_Ada.Arrays.Variable_Renaming
MD5: 4426aeaa364cb5cf10ff40e1bccb9757
```

Runtime output

```
5.00000E+00
5.00000E+00
7.50000E+00
7.50000E+00
```

In the example above, we declare a variable T by renaming the Current_Temperature object from the Measurements package. As you can see by running this example, both Current_Temperature and its alternative name T have the same values:

- first, they show the value 5.0
- after the addition, they show the value 7.5.

This is because they are essentially referring to the same object, but with two different names.

Note that, in the example above, we're using Degrees as an alias of Degree_Celsius. We discussed this method *earlier in the course* (page 58).

Renaming can be useful for improving the readability of more complicated array indexing. Instead of explicitly using indices every time we're accessing certain positions of the array, we can create shorter names for these positions by renaming them. Let's look at the following example:

Listing 20: colors.ads

```
package Colors is
1
2
       type Color is (Black,
3
                        Red,
4
                        Green,
5
                        Blue,
6
                       White);
7
8
       type Color_Array is
9
         array (Positive range <>) of Color;
10
11
       procedure Reverse_It (X : in out Color_Array);
12
13
   end Colors;
14
```

Listing 21: colors.adb

```
package body Colors is
1
2
      procedure Reverse_It (X : in out Color_Array)
3
       is
4
5
          for I in X'First ...
6
                    (X'Last + X'First) / 2
          loop
8
             declare
                          : Color;
                Tmp
10
                 X Left : Color
11
                   renames X (I);
12
                 X Right : Color
13
                   renames X (X'Last + X'First - I);
14
             begin
15
                                                                            (continues on next page)
```

83

(continued from previous page)

```
Tmp := X_Left;
X_Left := X_Right;
X_Right := Tmp;
end;
end loop;
end Reverse_It;

end Colors;
```

Listing 22: test_reverse_colors.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   with Colors; use Colors;
3
   procedure Test_Reverse_Colors is
5
      My Colors : Color Array (1 \dots 5) :=
7
         (Black, Red, Green, Blue, White);
   begin
10
       for C of My_Colors loop
11
          Put_Line ("My_Color: "
12
                    & Color'Image (C));
13
      end loop;
14
15
      New Line;
16
      Put_Line ("Reversing My_Color...");
17
18
      New_Line;
      Reverse_It (My_Colors);
19
20
      for C of My_Colors loop
21
          Put_Line ("My_Color: "
22
                     & Color'Image (C));
23
      end loop;
24
25
   end Test_Reverse_Colors;
26
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Arrays.Reverse_Colors
MD5: cd9fd7f64d1ec8967e340d57fd7afc0a
```

Runtime output

```
My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Reversing My_Color...

My_Color: WHITE
My_Color: BLUE
My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK
```

In the example above, package Colors implements the procedure Reverse_It by declaring new names for two positions of the array. The actual implementation becomes easy to read:

Compare this to the alternative version without renaming:

MORE ABOUT TYPES

8.1 Aggregates: A primer

So far, we have talked about aggregates quite a bit and have seen a number of examples. Now we will revisit this feature in some more detail.

An Ada aggregate is, in effect, a literal value for a composite type. It's a very powerful notation that helps you to avoid writing procedural code for the initialization of your data structures in many cases.

A basic rule when writing aggregates is that *every component* of the array or record has to be specified, even components that have a default value.

This means that the following code is incorrect:

Listing 1: incorrect.ads

```
package Incorrect is
type Point is record
X, Y : Integer := 0;
end record;

Origin : Point := (X => 0);
end Incorrect;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Types.Incorrect_Aggregate
MD5: 80a3475dece1c42cfb67b1d57b5bd464
```

Build output

```
incorrect.ads:6:22: error: no value supplied for component "Y"
gprbuild: *** compilation phase failed
```

There are a few shortcuts that you can use to make the notation more convenient:

- To specify the default value for a component, you can use the <> notation.
- You can use the | symbol to give several components the same value.
- You can use the **others** choice to refer to every component that has not yet been specified, provided all those fields have the same type.
- You can use the range notation . . to refer to specify a contiguous sequence of indices in an array.

However, note that as soon as you used a named association, all subsequent components likewise need to be specified with named associations.

Listing 2: points.ads

```
package Points is
       type Point is record
2
          X, Y : Integer := 0;
3
      end record;
4
5
      type Point Array is
6
        array (Positive range <>) of Point;
7
8
       -- use the default values
9
      Origin : Point := (X | Y => <>);
10
11
       -- likewise, use the defaults
12
      Origin_2 : Point := (others => <>);
13
14
      Points_1 : Point_Array := ((1, 2), (3, 4));
15
      Points_2 : Point_Array := (1 = (1, 2), 2 = (3, 4),
16
17
                                    3 \dots 20 \Rightarrow <>);
18
   end Points;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Types.Points
MD5: 48ea183a42f203325ed6190fbd8493d9
```

8.2 Overloading and qualified expressions

Ada has a general concept of name overloading, which we saw earlier in the section on *enumeration types* (page 47).

Let's take a simple example: it is possible in Ada to have functions that have the same name, but different types for their parameters.

Listing 3: pkg.ads

```
package Pkg is
function F (A : Integer) return Integer;
function F (A : Character) return Integer;
end Pkg;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Types.Overloading
MD5: defae85228ee183b536af395d077e71e
```

This is a common concept in programming languages, called overloading¹², or name overloading.

One of the novel aspects of Ada's overloading facility is the ability to resolve overloading based on the return type of a function.

Listing 4: pkg.ads

```
package Pkg is
type SSID is new Integer;

(continues on next page)
```

¹² https://en.wikipedia.org/wiki/Function_overloading

```
function Convert (Self : SSID)
return Integer;
function Convert (Self : SSID)
return String;
end Pkg;
```

Listing 5: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
  with Pkg;
                     use Pkg;
2
   procedure Main is
4
      S : String := Convert (123_145_299);
5
                    ^ Valid, will choose the
6
                      proper Convert
7
  begin
8
      Put Line (S);
  end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Types.Overloading
MD5: aa556b55ee89f9c5f8f7e138d84c27b8
```

A Attention

Note that overload resolution based on the type is allowed for both functions and enumeration literals in Ada - which is why you can have multiple enumeration literals with the same name. Semantically, an enumeration literal is treated like a function that has no parameters.

However, sometimes an ambiguity makes it impossible to resolve which declaration of an overloaded name a given occurrence of the name refers to. This is where a qualified expression becomes useful.

Listing 6: pkg.ads

```
package Pkg is
      type SSID is new Integer;
2
3
      function Convert (Self : SSID)
4
                         return Integer;
5
      function Convert (Self : SSID)
6
                         return String;
      function Convert (Self : Integer)
                         return String;
9
   end Pkg;
10
```

Listing 7: main.adb

(continues on next page)

```
S2 : String := Convert (SSID'(123_145_299));
9
                                ^ We specify that the
10
                                  type of the
11
                                  expression is SSID.
12
13
      -- We could also have declared a temporary
14
15
      I : SSID := 123 145 299;
16
17
      S3 : String := Convert (I);
18
   begin
19
      Put_Line (S);
20
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Types.Overloading_Error
MD5: 722660d8b692cde65a1c2b7800dd78c4
```

Syntactically the target of a qualified expression can be either any expression in parentheses, or an aggregate:

Listing 8: qual_expr.ads

```
package Qual_Expr is
type Point is record
A, B : Integer;
end record;

P : Point := Point'(12, 15);

A : Integer := Integer'(12);
end Qual_Expr;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Types.Qual_Expr
MD5: e71523eb441a28a4f6549d5f0418620a
```

This illustrates that qualified expressions are a convenient (and sometimes necessary) way for the programmer to make the type of an expression explicit, for the compiler of course, but also for other programmers.

Attention

While they look and feel similar, type conversions and qualified expressions are *not* the same.

A qualified expression specifies the exact type that the target expression will be resolved to, whereas a type conversion will try to convert the target and issue a run-time error if the target value cannot be so converted.

Note that you can use a qualified expression to convert from one subtype to another, with an exception raised if a constraint is violated.

```
X : Integer := Natural'(1);
```

8.3 Character types

As noted earlier, each enumeration type is distinct and incompatible with every other enumeration type. However, what we did not mention previously is that character literals are permitted as enumeration literals. This means that in addition to the language's strongly typed character types, user-defined character types are also permitted:

Listing 9: character example.adb

```
with Ada. Text IO; use Ada. Text IO;
1
   procedure Character Example is
3
      type My_Char is ('a', 'b', 'c');
      -- Our custom character type, an
      -- enumeration type with 3 valid values.
6
      C : Character;
8
      -- ^ Built-in character type
9
           (it's an enumeration type)
10
11
      M : My_Char;
12
   begin
13
      C := '?';
14
      -- ^ Character literal
15
             (enumeration literal)
16
17
      M := 'a';
18
19
      C := 65;
20
           ^ Invalid: 65 is not a
21
              Character value
22
23
      C := Character'Val (65);
24
      -- Assign the character at
25
      -- position 65 in the
26
      -- enumeration (which is 'A')
27
28
      M := C;
29
      -- ^ Invalid: C is of type Character,
30
           and M is a My_Char
31
32
      M := 'd';
33
      -- ^ Invalid: 'd' is not a valid
34
             literal for type My_Char
35
   end Character Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Types.Character_Example MD5: e4c5a07dbe8745749056f8c110d69fa3
```

Build output

```
character_example.adb:20:09: error: expected type "Standard.Character" character_example.adb:20:09: error: found type universal integer character_example.adb:29:09: error: expected type "My_Char" defined at line 4 character_example.adb:29:09: error: found type "Standard.Character" character_example.adb:33:09: error: character not defined for type "My_Char" defined at line 4 gprbuild: *** compilation phase failed
```

In this example, we're using characters in the definition of My Char.

ACCESS TYPES (POINTERS)

9.1 Overview

Pointers are a potentially dangerous construct, which conflicts with Ada's underlying philosophy.

There are two ways in which Ada helps shield programmers from the dangers of pointers:

- 1. One approach, which we have already seen, is to provide alternative features so that the programmer does not need to use pointers. Parameter modes, arrays, and varying size types are all constructs that can replace typical pointer usages in C.
- 2. Second, Ada has made pointers as safe and restricted as possible, but allows "escape hatches" when the programmer explicitly requests them and presumably will be exercising such features with appropriate care.

Here is how you declare a simple pointer type, or access type, in Ada:

Listing 1: dates.ads

```
package Dates is
1
      type Months is
2
        (January, February, March, April,
3
         May, June, July, August, September,
4
         October, November, December);
5
6
      type Date is record
7
         Day : Integer range 1 .. 31;
8
         Month: Months;
9
         Year : Integer;
10
      end record;
11
   end Dates;
12
```

Listing 2: access_types.ads

```
with Dates; use Dates;
1
2
   package Access Types is
3
        -- Declare an access type
4
        type Date_Acc is access Date;
5
                                   ` "Designated type"
6
                                  ^ Date_Acc values
        - -
                                    point to Date
        - -
8
        _ _
                                    objects
9
10
        D : Date_Acc := null;
11
                          ^ Literal for
12
                           "access to nothing"
13
```

(continues on next page)

```
-- ^ Access to date
end Access_Types;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Access_Types.Access_Types
MD5: d3421918c48c221836bdf03b9e68bfb5
```

This illustrates how to:

- Declare an access type whose values point to ("designate") objects from a specific type
- Declare a variable (access value) from this access type
- Give it a value of null

In line with Ada's strong typing philosophy, if you declare a second access type whose designated type is Date, the two access types will be incompatible with each other:

Listing 3: access_types.ads

```
with Dates; use Dates;
1
   package Access_Types is
3
       -- Declare an access type
       type Date_Acc is access Date;
5
       type Date_Acc_2 is access Date;
6
       D : Date Acc := null;
8
       D2 : Date_Acc_2 := D;
9
                           ^ Invalid! Different types
10
   end Access_Types;
11
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Access_Types.Access_Types
MD5: af0dff5a26cb16f0fe15c84286557a44
```

Build output

```
access_types.ads:9:24: error: expected type "Date_Acc_2" defined at line 6
access_types.ads:9:24: error: found type "Date_Acc" defined at line 5
gprbuild: *** compilation phase failed
```

In other languages

In most other languages, pointer types are structurally, not nominally typed, like they are in Ada, which means that two pointer types will be the same as long as they share the same target type and accessibility rules.

Not so in Ada, which takes some time getting used to. A seemingly simple problem is, if you want to have a canonical access to a type, where should it be declared? A commonly used pattern is that if you need an access type to a specific type you "own", you will declare it along with the type:

```
package Access_Types is
   type Point is record
    X, Y : Natural;
   end record;

   type Point_Access is access Point;
end Access_Types;
```

9.2 Allocation (by type)

Once we have declared an access type, we need a way to give variables of the types a meaningful value! You can allocate a value of an access type with the **new** keyword in Ada.

Listing 4: access types.ads

```
package Access_Types is
type Date_Acc is access Date;

D : Date_Acc := new Date;
Allocate a new Date record
end Access_Types;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Access_Types.Access_Types
MD5: e0be95b966e4aebaaf25db646d60c35c
```

If the type you want to allocate needs constraints, you can put them in the subtype indication, just as you would do in a variable declaration:

Listing 5: access_types.ads

```
with Dates; use Dates;
2
   package Access Types is
3
      type String_Acc is access String;
4
5
      -- Access to unconstrained array type
6
      Msg : String Acc;
7
             ^ Default value is null
8
9
      Buffer : String_Acc :=
10
        new String (1 .. 10);
11
                     ^ Constraint required
   end Access Types;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Access_Types.Access_Types
MD5: 83cf7a1074ff1b739658508098aa8208
```

In some cases, though, allocating just by specifying the type is not ideal, so Ada also allows you to initialize along with the allocation. This is done via the qualified expression syntax:

Listing 6: access types.ads

```
with Dates; use Dates;

package Access_Types is
    type Date_Acc is access Date;
    type String_Acc is access String;

D : Date_Acc :=
        new Date'(30, November, 2011);

(continues on next page)
```

```
Msg : String_Acc := new String'("Hello");
end Access_Types;
```

9.3 Dereferencing

The last important piece of Ada's access type facility is how to get from an access value to the object that is pointed to, that is, how to dereference the pointer. Dereferencing a pointer uses the .all syntax in Ada, but is often not needed — in many cases, the access value will be implicitly dereferenced for you:

Listing 7: access_types.ads

```
with Dates; use Dates;
   package Access Types is
      type Date Acc is access Date;
            : Date Acc :=
6
                 new Date'(30, November, 2011);
      Today : Date := D.all;
9
                       ^ Access value dereference
10
      1
             : Integer := D.Day;
11
                           ^ Implicit dereference
12
                            for record and array
13
                            components
14
                            Equivalent to D.all.day
15
  end Access Types;
16
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Access_Types.Access_Types
MD5: 5cd1c259da04010b0dc1b43e9bd93b55
```

9.4 Other features

As you might know if you have used pointers in C or C++, we are still missing features that are considered fundamental to the use of pointers, such as:

- Pointer arithmetic (being able to increment or decrement a pointer in order to point to the next or previous object)
- Manual deallocation what is called free or delete in C. This is a potentially unsafe operation. To keep within the realm of safe Ada, you need to never deallocate manually.

Those features exist in Ada, but are only available through specific standard library APIs.

Attention

The guideline in Ada is that most of the time you can avoid manual allocation, and you should.

There are many ways to avoid manual allocation, some of which have been covered (such as parameter modes). The language also provides library abstractions to avoid pointers:

- 1. One is the use of *containers* (page 199). Containers help users avoid pointers, because container memory is automatically managed.
- 2. A container to note in this context is the Indefinite holder¹³. This container allows you to store a value of an indefinite type such as String.
- 3. GNATCOLL has a library for smart pointers, called Refcount¹⁴ Those pointers' memory is automatically managed, so that when an allocated object has no more references to it, the memory is automatically deallocated.

9.5 Mutually recursive types

The linked list is a common idiom in data structures; in Ada this would be most naturally defined through two types, a record type and an access type, that are mutually dependent. To declare mutually dependent types, you can use an incomplete type declaration:

Listing 8: simple_list.ads

```
package Simple_List is
      type Node;
2
      -- This is an incomplete type declaration,
      -- which is completed in the same
4
      -- declarative region.
5
      type Node Acc is access Node;
7
8
      type Node is record
9
         Content
                  : Natural;
10
         Prev, Next : Node_Acc;
11
      end record;
   end Simple List;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Access_Types.Simple_List
MD5: 4929b89c1fc913da635fa02e48248271
```

In this example, the Node and Node_Acc types are mutually dependent.

¹³ http://www.ada-auth.org/standards/12rat/html/Rat12-8-5.html

¹⁴ https://github.com/AdaCore/gnatcoll-core/blob/master/src/gnatcoll-refcount.ads

MORE ABOUT RECORDS

10.1 Dynamically sized record types

We have previously seen *some simple examples of record types* (page 61). Let's now look at some of the more advanced properties of this fundamental language feature.

One point to note is that object size for a record type does not need to be known at compile time. This is illustrated in the example below:

Listing 1: runtime length.ads

```
package Runtime_Length is
function Compute_Max_Len return Natural;
end Runtime_Length;
```

Listing 2: var size record.ads

```
with Runtime_Length; use Runtime_Length;
   package Var_Size_Record is
       Max Len : constant Natural :=
                    Compute Max Len;
                    ^ Not known at compile time
       type Items_Array is
         array (Positive range <>) of Integer;
9
10
       type Growable_Stack is record
11
          Items : Items_Array (1 .. Max_Len);
12
          Len
               : Natural;
13
       end record;
14
       -- Growable Stack is a definite type, but
15
       -- size is not known at compile time.
16
17
       G : Growable Stack;
18
   end Var_Size_Record;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Var_Size_Record
MD5: 6fb0b3f2b685a72ec694640ce378f77c
```

It is completely fine to determine the size of your records at run time, but note that all objects of this type will have the same size.

10.2 Records with discriminant

In the example above, the size of the Items field is determined once, at run-time, but every Growable_Stack instance will be exactly the same size. But maybe that's not what you want to do. We saw that arrays in general offer this flexibility: for an unconstrained array type, different objects can have different sizes.

You can get analogous functionality for records, too, using a special kind of field that is called a discriminant:

Listing 3: var_size_record_2.ads

```
package Var Size Record 2 is
       type Items Array is
2
         array (Positive range <>) of Integer;
3
4
       type Growable Stack (Max Len : Natural) is
5
       record
6
                             ^ Discriminant. Cannot be
7
                               modified once
8
9
                               initialized.
          Items : Items Array (1 .. Max Len);
10
          Len : Natural := 0;
11
12
       end record;
       -- Growable Stack is an indefinite type
13
       -- (like an array)
14
  end Var Size Record 2;
15
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Var_Size_Record_2
MD5: 0c2ffe41b7553984e1ef48a50386559f
```

Discriminants, in their simple forms, are constant: You cannot modify them once you have initialized the object. This intuitively makes sense since they determine the size of the object.

Also, they make a type indefinite: Whether or not the discriminant is used to specify the size of an object, a type with a discriminant will be indefinite if the discriminant is not declared with an initialization:

Listing 4: test_discriminants.ads

```
package Test Discriminants is
1
      type Point (X, Y : Natural) is record
2
         null;
3
      end record;
4
      P : Point;
6
      -- ERROR: Point is indefinite, so you
7
      -- need to specify the discriminants
8
      -- or give a default value
9
10
      P2 : Point (1, 2);
11
      P3 : Point := (1, 2);
12
         Those two declarations are equivalent.
13
14
   end Test_Discriminants;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Test_Discriminants
MD5: c3ec81ccae0d4144fe952ad99482be81
```

Build output

```
test_discriminants.ads:6:08: error: unconstrained subtype not allowed (need_ initialization)
test_discriminants.ads:6:08: error: provide initial value or explicit discriminant_ values
test_discriminants.ads:6:08: error: or give default discriminant values for type
    "Point"
gprbuild: *** compilation phase failed
```

This also means that, in the example above, you cannot declare an array of Point values, because the size of a Point is not known.

As mentioned in the example above, we could provide a default value for the discriminants, so that we could legally declare Point values without specifying the discriminants. For the example above, this is how it would look:

Listing 5: test discriminants.ads

```
package Test Discriminants is
1
      type Point (X, Y : Natural := 0) is record
2
3
         null:
      end record;
4
5
      P : Point;
6
      -- We can now simply declare a "Point"
      -- without further ado. In this case,
8
      -- we're using the default values (0)
9
      -- for X and Y.
10
11
      P2 : Point (1, 2);
12
      P3 : Point := (1, 2);
13
      -- We can still specify discriminants.
14
15
  end Test_Discriminants;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Test_Discriminants
MD5: 259f6cdf7fa857cc006dac6d1daedd73
```

Also note that, even though the Point type now has default discriminants, we can still specify discriminants, as we're doing in the declarations of P2 and P3.

In most other respects discriminants behave like regular fields: You have to specify their values in aggregates, as seen above, and you can access their values via the dot notation.

Listing 6: main.adb

```
with Ada.Text I0;
                             use Ada.Text I0;
   with Var_Size_Record_2; use Var_Size_Record_2;
3
   procedure Main is
5
      procedure Print_Stack (G : Growable_Stack) is
6
      begin
7
         Put ("<Stack, items: [");</pre>
8
          for I in G.Items'Range loop
9
             exit when I > G.Len;
10
                                                                          (continues on next page)
```

```
Put (" " & Integer'Image (G.Items (I)));
11
          end loop;
12
          Put_Line ("]>");
13
      end Print_Stack;
15
      S : Growable_Stack :=
16
         (Max_Len => 128,
17
          Items => (1, 2, 3, 4, others => <>),
18
          Len
                 => 4);
19
   begin
20
      Print Stack (S);
21
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Var_Size_Record_2
MD5: 4e8c102cd93dc5d8aa1b402589c5239b
```

Runtime output

```
<Stack, items: [ 1 2 3 4]>
```

Mote

In the examples above, we used a discriminant to determine the size of an array, but it is not limited to that, and could be used, for example, to determine the size of a nested discriminated record.

10.3 Variant records

The examples of discriminants thus far have illustrated the declaration of records of varying size, by having components whose size depends on the discriminant.

However, discriminants can also be used to obtain the functionality of what are sometimes called "variant records": records that can contain different sets of fields.

Listing 7: variant record.ads

```
package Variant Record is
1
      -- Forward declaration of Expr
2
3
      type Expr;
4
      -- Access to a Expr
      type Expr_Access is access Expr;
6
      type Expr_Kind_Type is (Bin_Op_Plus,
8
                               Bin_Op_Minus,
9
                               Num):
10
      -- A regular enumeration type
11
12
      type Expr (Kind : Expr_Kind_Type) is record
13
                  ^ The discriminant is an
14
15
                   enumeration value
         case Kind is
            when Bin_0p_Plus | Bin_0p_Minus =>
17
                Left, Right : Expr_Access;
18
            when Num =>
19
```

(continues on next page)

```
Val: Integer;
end case;
-- Variant part. Only one, at the end of
-- the record definition, but can be
-- nested
end record;
end Variant_Record;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Variant_Record MD5: af9cledca3ed6b2d938249c7258806b1
```

The fields that are in a when branch will be only available when the value of the discriminant is covered by the branch. In the example above, you will only be able to access the fields Left and Right when the Kind is Bin Op Plus or Bin Op Minus.

If you try to access a field that is not valid for your record, a Constraint_Error will be raised.

Listing 8: main.adb

```
with Variant_Record; use Variant_Record;

procedure Main is
    E : Expr := (Num, 12);

begin
    E.Left := new Expr'(Num, 15);
    -- Will compile but fail at runtime
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Variant_Record
MD5: d157d5f96db0825b9376ba7fca9613ed
```

Build output

Runtime output

```
raised CONSTRAINT_ERROR : main.adb:6 discriminant check failed
```

Here is how you could write an evaluator for expressions:

Listing 9: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;

with Variant_Record; use Variant_Record;

procedure Main is
function Eval_Expr (E : Expr) return Integer is
(case E.Kind is
when Bin_Op_Plus =>
Eval_Expr (E.Left.all)
(continues on next page)
```

```
+ Eval_Expr (E.Right.all),
10
          when Bin_Op_Minus =>
11
                 Eval_Expr (E.Left.all)
12
                 Eval_Expr (E.Right.all),
          when Num => E.Val);
14
15
      E : Expr := (Bin_Op_Plus,
16
                    new Expr'(Bin_Op_Minus,
17
                               new Expr'(Num, 12),
18
                               new Expr'(Num, 15)),
19
                    new Expr'(Num, 3));
20
   begin
21
      Put_Line (Integer'Image (Eval_Expr (E)));
22
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.More_About_Records.Variant_Record
MD5: 807dbb921b44b3eaeaf1baf6ffe1afaa
```

Runtime output

0

1 In other languages

Ada's variant records are very similar to Sum types in functional languages such as OCaml or Haskell. A major difference is that the discriminant is a separate field in Ada, whereas the 'tag' of a Sum type is kind of built in, and only accessible with pattern matching.

There are other differences (you can have several discriminants in a variant record in Ada). Nevertheless, they allow the same kind of type modeling as sum types in functional languages.

Compared to C/C++ unions, Ada variant records are more powerful in what they allow, and are also checked at run time, which makes them safer.

FIXED-POINT TYPES

11.1 Decimal fixed-point types

We have already seen how to specify floating-point types. However, in some applications floating-point is not appropriate since, for example, the roundoff error from binary arithmetic may be unacceptable or perhaps the hardware does not support floating-point instructions. Ada provides a category of types, the decimal fixed-point types, that allows the programmer to specify the required decimal precision (number of digits) as well as the scaling factor (a power of ten) and, optionally, a range. In effect the values will be represented as integers implicitly scaled by the specified power of 10. This is useful, for example, for financial applications.

The syntax for a simple decimal fixed-point type is

```
type <type-name> is delta <delta-value> digits <digits-value>;
```

In this case, the **delta** and the **digits** will be used by the compiler to derive a range.

Several attributes are useful for dealing with decimal types:

Attribute Name	Meaning
First	The first value of the type
Last	The last value of the type
Delta	The delta value of the type

In the example below, we declare two data types: T3_D3 and T6_D3. For both types, the delta value is the same: 0.001.

Listing 1: decimal fixed point types.adb

```
with Ada.Text_IO; use Ada.Text IO;
   procedure Decimal Fixed Point Types is
      type T3_D3 is delta \overline{10.0} ** (-3) digits 3;
      type T6 D3 is delta 10.0 ** (-3) digits 6;
   begin
                               value of T3 D3 is "
      Put Line ("The delta
7
                 & T3_D3'Image (T3_D3'Delta));
8
      Put_Line ("The minimum value of T3_D3 is "
9
                 & T3 D3'Image (T3 D3'First));
10
      Put Line ("The maximum value of T3 D3 is "
11
                 & T3 D3'Image (T3 D3'Last));
12
      New Line;
13
14
      Put Line ("The delta
                                value of T6 D3 is "
                 & T6 D3'Image (T6 D3'Delta));
```

```
Put_Line ("The minimum value of T6_D3 is "
& T6_D3'Image (T6_D3'First));

Put_Line ("The maximum value of T6_D3 is "
& T6_D3'Image (T6_D3'Last));

end Decimal_Fixed_Point_Types;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Fixed_Point_Types.Decimal_Fixed_Point_Types MD5: 6b1f6bfa555031b831aa872187c8bee9
```

Runtime output

```
The delta value of T3_D3 is 0.001
The minimum value of T3_D3 is -0.999
The maximum value of T3_D3 is 0.999

The delta value of T6_D3 is 0.001
The minimum value of T6_D3 is -999.999
The maximum value of T6_D3 is 999.999
```

When running the application, we see that the delta value of both types is indeed the same: 0.001. However, because T3_D3 is restricted to 3 digits, its range is -0.999 to 0.999. For the T6_D3, we have defined a precision of 6 digits, so the range is -999.999 to 999.999.

Similar to the type definition using the **range** syntax, because we have an implicit range, the compiled code will check that the variables contain values that are not out-of-range. Also, if the result of a multiplication or division on decimal fixed-point types is smaller than the delta value required for the context, the actual result will be zero. For example:

Listing 2: decimal fixed point smaller.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Decimal Fixed Point Smaller is
       type T3 D3 is delta \overline{10.0} ** (-3) digits 3;
       type T6_D6 is delta 10.0 ** (-6) digits 6;
5
      A : T3_D3 := T3_D3'Delta;
6
      B : T3_D3 := 0.5;
7
      C : T6_D6;
8
   begin
9
      Put Line ("The value of A
                                       is "
10
                  & T3 D3'Image (A));
11
12
      A := A * B;
13
      Put Line ("The value of A * B is "
14
                  & T3 D3'Image (A));
15
16
      A := T3_D3'Delta;
17
      C := A \times B:
18
      Put_Line ("The value of A * B is "
19
                  & T6 D6'Image (C));
20
   end Decimal Fixed Point Smaller;
21
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Fixed_Point_Types.Decimal_Fixed_Point_Smaller MD5: 6b0242caa4a79f9b3447a304002e6a3b
```

Runtime output

```
The value of A is 0.001
The value of A * B is 0.000
The value of A * B is 0.000500
```

In this example, the result of the operation 0.001 * 0.5 is 0.0005. Since this value is not representable for the T3_D3 type because the delta value is 0.001, the actual value stored in variable A is zero. However, accuracy is preserved during the arithmetic operations if the target has sufficient precision, and the value displayed for C is 0.000500.

11.2 Ordinary fixed-point types

Ordinary fixed-point types are similar to decimal fixed-point types in that the values are, in effect, scaled integers. The difference between them is in the scale factor: for a decimal fixed-point type, the scaling, given explicitly by the type's **delta**, is always a power of ten.

In contrast, for an ordinary fixed-point type, the scaling is defined by the type's small, which is derived from the specified **delta** and, by default, is a power of two. Therefore, ordinary fixed-point types are sometimes called binary fixed-point types.

1 Note

Ordinary fixed-point types can be thought of being closer to the actual representation on the machine, since hardware support for decimal fixed-point arithmetic is not widespread (rescalings by a power of ten), while ordinary fixed-point types make use of the available integer shift instructions.

The syntax for an ordinary fixed-point type is

```
type <type-name> is
  delta <delta-value>
  range <lower-bound> .. <upper-bound>;
```

By default the compiler will choose a scale factor, or small, that is a power of 2 no greater than <delta-value>.

For example, we may define a normalized range between -1.0 and 1.0 as following:

Listing 3: normalized fixed point type.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Normalized Fixed Point Type is
3
      D : constant := 2.0 ** (-31);
      type TQ31 is delta D range -1.0 .. 1.0 - D;
   begin
      Put Line ("TQ31 requires "
                & Integer'Image (TQ31'Size)
                & " bits");
9
      Put_Line ("The delta
                               value of TQ31 is "
10
                & TQ31'Image (TQ31'Delta));
11
      Put Line ("The minimum value of TQ31 is "
12
                & TQ31'Image (TQ31'First));
13
      Put Line ("The maximum value of TQ31 is "
14
                & TQ31'Image (TQ31'Last));
15
   end Normalized Fixed Point Type;
16
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Fixed_Point_Types.Normalized_Fixed_Point_Type MD5: 778dde401c7ff3dd42938dccfe6cf9d3
```

Runtime output

```
TQ31 requires 32 bits
The delta value of TQ31 is 0.0000000005
The minimum value of TQ31 is -1.0000000000
The maximum value of TQ31 is 0.9999999995
```

In this example, we are defining a 32-bit fixed-point data type for our normalized range. When running the application, we notice that the upper bound is close to one, but not exact one. This is a typical effect of fixed-point data types — you can find more details in this discussion about the Q format 15 .

We may also rewrite this code with an exact type definition:

Listing 4: normalized adapted fixed point type.adb

```
procedure Normalized_Adapted_Fixed_Point_Type is

type TQ31 is

delta 2.0 ** (-31)

range -1.0 .. 1.0 - 2.0 ** (-31);

begin
null;
end Normalized_Adapted_Fixed_Point_Type;
```

Code block metadata

Project: Courses.Intro_To_Ada.Fixed_Point_Types.Normalized_Adapted_Fixed_Point_Type MD5: 3421800bb47b282d601a51d276944f62

We may also use any other range. For example:

Listing 5: custom fixed point range.adb

```
with Ada.Text IO; use Ada.Text IO;
   with Ada.Numerics; use Ada.Numerics;
   procedure Custom Fixed Point Range is
      type T_Inv_Trig is
5
         delta 2.0 ** (-15) * Pi
6
         range -Pi / 2.0 .. Pi / 2.0;
7
   begin
8
      Put_Line ("T_Inv_Trig requires "
9
                 & Integer'Image (T Inv Trig'Size)
10
                 & " bits");
11
      Put_Line ("Delta
                           value of T Inv Trig: "
12
                 & T_Inv_Trig'Image
13
                      (T_Inv_Trig'Delta));
14
      Put Line ("Minimum value of T Inv Trig: "
15
                 & T_Inv_Trig'Image
16
                      (T_Inv_Trig'First));
17
      Put Line ("Maximum value of T Inv Trig: "
18
                 & T_Inv_Trig'Image
19
                      (T \overline{I}nv Trig'Last));
20
   end Custom Fixed Point Range;
21
```

Code block metadata

¹⁵ https://en.wikipedia.org/wiki/Q_(number_format)

```
Project: Courses.Intro_To_Ada.Fixed_Point_Types.Custom_Fixed_Point_Range
MD5: a3e6c549cb1070aa285857ae8813de27
```

Runtime output

```
T_Inv_Trig requires 16 bits
Delta value of T_Inv_Trig: 0.00006
Minimum value of T_Inv_Trig: -1.57080
Maximum value of T_Inv_Trig: 1.57080
```

In this example, we are defining a 16-bit type called T_Inv_Trig, which has a range from $-\pi/2$ to $\pi/2$.

All standard operations are available for fixed-point types. For example:

Listing 6: fixed point op.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Fixed_Point_Op is
3
      type TQ31 is
4
        delta 2.0 ** (-31)
5
         range -1.0 .. 1.0 - 2.0 ** (-31);
6
7
      A, B, R: TQ31;
8
   begin
      A := 0.25;
10
      B := 0.50;
11
      R := A + B;
12
      Put Line ("R is " & TQ31'Image (R));
13
  end Fixed_Point_Op;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Fixed_Point_Types.Fixed_Point_Op
MD5: cad218b70b7fb0621468027a807431b1
```

Runtime output

```
R is 0.7500000000
```

As expected, R contains 0.75 after the addition of A and B.

In fact the language is more general than these examples imply, since in practice it is typical to need to multiply or divide values from different fixed-point types, and obtain a result that may be of a third fixed-point type. The details are outside the scope of this introductory course.

It is also worth noting, although again the details are outside the scope of this course, that you can explicitly specify a value for an ordinary fixed-point type's small. This allows non-binary scaling, for example:

```
type Angle is
  delta 1.0/3600.0
  range 0.0 .. 360.0 - 1.0 / 3600.0;
for Angle'Small use Angle'Delta;
```

CHAPTER

TWELVE

PRIVACY

One of the main principles of modular programming, as well as object oriented programming, is encapsulation¹⁶.

Encapsulation, briefly, is the concept that the implementer of a piece of software will distinguish between the code's public interface and its private implementation.

This is not only applicable to software libraries but wherever abstraction is used.

In Ada, the granularity of encapsulation is a bit different from most object-oriented languages, because privacy is generally specified at the package level.

12.1 Basic encapsulation

Listing 1: encapsulate.ads

```
package Encapsulate is
procedure Hello;

private

procedure Hello2;
    -- Not visible from external units
end Encapsulate;
```

Listing 2: encapsulate.adb

```
with Ada. Text IO; use Ada. Text IO;
   package body Encapsulate is
      procedure Hello is
5
      begin
6
          Put Line ("Hello");
      end Hello:
8
9
      procedure Hello2 is
10
      begin
11
          Put Line ("Hello #2");
12
      end Hello2;
   end Encapsulate;
```

 $^{^{16}\} https://en.wikipedia.org/wiki/Encapsulation_(computer_programming)$

Listing 3: main.adb

```
with Encapsulate;

procedure Main is
begin

Encapsulate.Hello;
Encapsulate.Hello2;
-- Invalid: Hello2 is not visible
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Privacy.Encapsulate
MD5: cf56ee89481962d1e0a6d1e9ad888362
```

Build output

```
main.adb:6:15: error: "Hello2" is not a visible entity of "Encapsulate"
gprbuild: *** compilation phase failed
```

12.2 Abstract data types

With this high-level granularity, it might not seem obvious how to hide the implementation details of a type. Here is how it can be done in Ada:

Listing 4: stacks.ads

```
package Stacks is
      type Stack is private;
2
      -- Declare a private type: You cannot depend
3
      -- on its implementation. You can only assign
4
      -- and test for equality.
5
6
      procedure Push (S : in out Stack;
7
                       Val:
                                   Integer);
8
      procedure Pop (S : in out Stack;
9
                      Val : out Integer);
10
   private
11
12
      subtype Stack_Index is
13
        Natural range 1 .. 10;
14
15
      type Content_Type is
16
        array (Stack_Index) of Natural;
17
18
      type Stack is record
19
                : Stack_Index;
20
         Content : Content_Type;
21
      end record;
   end Stacks;
```

Listing 5: stacks.adb

```
package body Stacks is

procedure Push (S : in out Stack;

Val : Integer) is

begin

(continues on next page)
```

```
-- Missing implementation!
6
         null;
7
      end Push;
8
9
      procedure Pop (S : in out Stack;
10
                     Val: out Integer) is
11
      begin
12
         -- Dummy implementation!
13
         Val := 0;
14
      end Pop;
15
16
   end Stacks;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Privacy.Stacks
MD5: 364df7c6806af4a1bc957c2c2d53b2cc
```

In the above example, we define a stack type in the public part (known as the *visible part* of the package spec in Ada), but the exact representation of that type is private.

Then, in the private part, we define the representation of that type. We can also declare other types that will be used as *helpers* for our main public type. This is useful since declaring helper types is common in Ada.

A few words about terminology:

- The Stack type as viewed from the public part is called the partial view of the type. This is what clients have access to.
- The Stack type as viewed from the private part or the body of the package is called the full view of the type. This is what implementers have access to.

From the point of view of the client (the *with*'ing unit), only the public (visible) part is important, and the private part could as well not exist. It makes it very easy to read linearly the part of the package that is important for you.

Here is how the Stacks package would be used:

```
-- Example of use
with Stacks; use Stacks;

procedure Test_Stack is
   S : Stack;
   Res : Integer;
begin
   Push (S, 5);
   Push (S, 7);
   Pop (S, Res);
end Test_Stack;
```

12.3 Limited types

Ada's *limited type* facility allows you to declare a type for which assignment and comparison operations are not automatically provided.

Listing 6: stacks.ads

```
package Stacks is
      type Stack is limited private;
2
      -- Limited type. Cannot assign nor compare.
3
      procedure Push (S : in out Stack;
5
                      Val: Integer):
6
      procedure Pop (S : in out Stack;
7
                     Val : out Integer);
   private
      subtype Stack_Index is
10
       Natural range 1 .. 10;
11
12
      type Content_Type is
13
        array (Stack_Index) of Natural;
14
15
      type Stack is limited record
16
         Top : Stack Index;
17
         Content : Content Type;
18
      end record;
19
   end Stacks;
```

Listing 7: stacks.adb

```
package body Stacks is
      procedure Push (S : in out Stack;
3
                     Val: Integer) is
4
      begin
5
        -- Missing implementation!
6
         null:
7
      end Push;
8
9
      procedure Pop (S : in out Stack;
10
                     Val : out Integer) is
11
12
         -- Dummy implementation!
13
         Val := 0;
      end Pop;
15
16
   end Stacks;
17
```

Listing 8: main.adb

```
with Stacks; use Stacks;

procedure Main is
    S, S2 : Stack;

begin
    S := S2;
    -- Illegal: S is limited.
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Privacy.Limited_Stacks
MD5: 811343b46f20ac6af5e1bf26561f8d8d
```

Build output

```
main.adb:6:04: error: left hand of assignment must not be limited type gprbuild: *** compilation phase failed
```

This is useful because, for example, for some data types the built-in assignment operation might be incorrect (for example when a deep copy is required).

Ada does allow you to overload the comparison operators = and /= for limited types (and to override the built-in declarations for non-limited types).

Ada also allows you to implement special semantics for assignment via controlled types¹⁷. However, in some cases assignment is simply inappropriate; one example is the **File_Type** from the Ada.Text_IO package, which is declared as a limited type and thus attempts to assign one file to another would be detected as illegal.

12.4 Child packages & privacy

We've seen previously (in the *child packages section* (page 35)) that packages can have child packages. Privacy plays an important role in child packages. This section discusses some of the privacy rules that apply to child packages.

Although the private part of a package P is meant to encapsulate information, certain parts of a child package P.C can have access to this private part of P. In those cases, information from the private part of P can then be used as if it were declared in the public part of its specification. To be more specific, the body of P.C and the private part of the specification of P.C have access to the private part of P. However, the public part of the specification of P.C only has access to the public part of P's specification. The following table summarizes this:

Part of a child package	Access to the private part of its parent's specification
Specification: public part	
Specification: private part	\checkmark
Body	\checkmark

The rest of this section shows examples of how this access to private information actually works for child packages.

Let's first look at an example where the body of a child package P.C has access to the private part of the specification of its parent P. We've seen, in a previous source-code example,

¹⁷ http://www.ada-auth.org/standards/12rm/html/RM-7-6.html

that the Hello2 procedure declared in the private part of the Encapsulate package cannot be used in the Main procedure, since it's not visible there. This limitation doesn't apply, however, for parts of the child packages of the Encapsulate package. In fact, the body of its child package Encapsulate. Child has access to the Hello2 procedure and can call it there, as you can see in the implementation of the Hello3 procedure of the Child package:

Listing 9: encapsulate.ads

```
package Encapsulate is
procedure Hello;

private

procedure Hello2;
   -- Not visible from external units
   -- But visible in child packages
end Encapsulate;
```

Listing 10: encapsulate.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   package body Encapsulate is
3
4
      procedure Hello is
5
6
      begin
          Put_Line ("Hello");
      end Hello;
8
9
      procedure Hello2 is
10
      begin
11
          Put Line ("Hello #2");
12
      end Hello2;
13
14
   end Encapsulate;
15
```

Listing 11: encapsulate-child.ads

```
package Encapsulate.Child is

procedure Hello3;

end Encapsulate.Child;
```

Listing 12: encapsulate-child.adb

```
with Ada. Text IO; use Ada. Text IO;
   package body Encapsulate.Child is
3
      procedure Hello3 is
5
      begin
6
             Using private procedure Hello2
7
             from the parent package
8
         Hello2;
9
         Put Line ("Hello #3");
10
      end Hello3;
11
12
   end Encapsulate.Child;
```

Listing 13: main.adb

```
with Encapsulate.Child;

procedure Main is
begin
Encapsulate.Child.Hello3;
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Privacy.Encapsulate_Child
MD5: 1533f43eee8f8b4d14c9b2101f42f13a
```

Runtime output

```
Hello #2
Hello #3
```

The same mechanism applies to types declared in the private part of a parent package. For instance, the body of a child package can access components of a record declared in the private part of its parent package. Let's look at an example:

Listing 14: my_types.ads

```
package My_Types is

type Priv_Rec is private;

private

type Priv_Rec is record
    Number : Integer := 42;
end record;

end My_Types;
```

Listing 15: my_types-ops.ads

```
package My_Types.Ops is

procedure Display (E : Priv_Rec);

end My_Types.Ops;
```

Listing 16: my_types-ops.adb

```
with Ada.Text_IO; use Ada.Text_IO;

package body My_Types.Ops is

procedure Display (E : Priv_Rec) is
begin
    Put_Line ("Priv_Rec.Number: "
    & Integer'Image (E.Number));
end Display;

end My_Types.Ops;
```

Listing 17: main.adb

```
with Ada.Text_IO; use Ada.Text_IO;
2
   with My Types;
                       use My_Types;
3
   with My_Types.Ops; use My_Types.Ops;
4
   procedure Main is
6
      E : Priv_Rec;
   begin
      Put_Line ("Presenting information:");
9
10
          The following code would trigger a
11
          compilation error here:
12
13
         Put Line ("Priv Rec.Number: "
14
                     & Integer'Image (E.Number));
15
16
      Display (E);
17
   end Main;
18
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Privacy.Private_Type_Child
MD5: 9960611460bc1190b30949eca08fc02b
```

Runtime output

```
Presenting information:
Priv_Rec.Number: 42
```

In this example, we don't have access to the Number component of the record type Priv_Rec in the Main procedure. You can see this in the call to Put_Line that has been commented-out in the implementation of Main. Trying to access the Number component there would trigger a compilation error. But we do have access to this component in the body of the My_Types.Ops package, since it's a child package of the My_Types package. Therefore, Ops's body has access to the declaration of the Priv_Rec type — which is in the private part of its parent, the My_Types package. For this reason, the same call to Put_Line that would trigger a compilation error in the Main procedure works fine in the Display procedure of the My_Types.Ops package.

This kind of privacy rules for child packages allows for extending the functionality of a parent package and, at the same time, retain its encapsulation.

As we mentioned previously, in addition to the package body, the private part of the specification of a child package P.C also has access to the private part of the specification of its parent P. Let's look at an example where we declare an object of private type Priv_Rec in the private part of the child package My_Types.Child and initialize the Number component of the Priv Rec record directly:

```
package My_Types.Child is

private
    E : Priv_Rec := (Number => 99);
end My_Types.Ops;
```

As expected, we wouldn't be able to initialize this component if we moved this declaration to the public (visible) part of the same child package:

```
package My_Types.Child is
    E : Priv_Rec := (Number => 99);
end My_Types.Ops;
```

The declaration above triggers a compilation error, since type Priv_Rec is private. Because the public part of My_Types.Child is also visible outside the child package, Ada cannot allow accessing private information in this part of the specification.

CHAPTER

THIRTEEN

GENERICS

13.1 Introduction

Generics are used for metaprogramming in Ada. They are useful for abstract algorithms that share common properties with each other.

Either a subprogram or a package can be generic. A generic is declared by using the keyword **generic**. For example:

Listing 1: operator.ads

```
generic
type T is private;
-- Declaration of formal types and objects
-- Below, we could use one of the following:
-- cprocedure | function | package>
procedure Operator (Dummy : in out T);
```

Listing 2: operator.adb

```
procedure Operator (Dummy : in out T) is
begin
null;
end Operator;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Simple_Generic
MD5: 1321d437043dafdb725fad416e654318
```

13.2 Formal type declaration

Formal types are abstractions of a specific type. For example, we may want to create an algorithm that works on any integer type, or even on any type at all, whether a numeric type or not. The following example declares a formal type T for the Set procedure.

Listing 3: set.ads

```
generic
type T is private;
-- T is a formal type that indicates that
-- any type can be used, possibly a numeric
-- type or possibly even a record type.
procedure Set (Dummy : T);
```

Listing 4: set.adb

```
procedure Set (Dummy : T) is
begin
null;
end Set;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Formal_Type_Declaration
MD5: 668156f66b2479c4932d18b5ad35deba
```

The declaration of T as **private** indicates that you can map any definite type to it. But you can also restrict the declaration to allow only some types to be mapped to that formal type. Here are some examples:

Formal Type	Format
Any type	type T is private;
Any discrete type	type T is (<>);
Any floating-point type	type T is digits <>;

13.3 Formal object declaration

Formal objects are similar to subprogram parameters. They can reference formal types declared in the formal specification. For example:

Listing 5: set.ads

```
generic
type T is private;
X : in out T;
-- X can be used in the Set procedure
procedure Set (E : T);
```

Listing 6: set.adb

```
procedure Set (E : T) is
pragma Unreferenced (E, X);
begin
null;
end Set;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Formal_Object_Declaration
MD5: 1b88bc0e5b8f48a35394966e6af07ac0
```

Formal objects can be either input parameters or specified using the **in out** mode.

13.4 Generic body definition

We don't repeat the **generic** keyword for the body declaration of a generic subprogram or package. Instead, we start with the actual declaration and use the generic types and objects we declared. For example:

Listing 7: set.ads

```
generic
type T is private;
X : in out T;
procedure Set (E : T);
```

Listing 8: set.adb

```
procedure Set (E : T) is
-- Body definition: "generic" keyword
-- is not used
begin
X := E;
end Set;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Generic_Body_Definition
MD5: de61lef77b528543fd6bad82c53857f7
```

13.5 Generic instantiation

Generic subprograms or packages can't be used directly. Instead, they need to be instantiated, which we do using the **new** keyword, as shown in the following example:

Listing 9: set.ads

```
generic
type T is private;
X : in out T;
-- X can be used in the Set procedure
procedure Set (E : T);
```

Listing 10: set.adb

```
procedure Set (E : T) is
begin
    X := E;
end Set;
```

Listing 11: show generic instantiation.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   with Set;
2
3
   procedure Show_Generic_Instantiation is
4
5
               : Integer := 0;
6
      Current : Integer;
8
      procedure Set_Main is new Set (T => Integer,
9
                                       X \Rightarrow Main);
10
       -- Here, we map the formal parameters to
11
       -- actual types and objects.
12
13
       -- The same approach can be used to
14
          instantiate functions or packages, e.g.:
15
```

```
16
          function Get_Main is new ...
17
          package Integer_Queue is new ...
18
   begin
20
      Current := 10;
21
22
      Set_Main (Current);
23
      Put_Line ("Value of Main is "
24
                 & Integer'Image (Main));
25
   end Show_Generic_Instantiation;
26
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Generic_Instantiation
MD5: 13dc0692252496d954240952561e1c05
```

Runtime output

```
Value of Main is 10
```

In the example above, we instantiate the procedure Set by mapping the formal parameters T and X to actual existing elements, in this case the **Integer** type and the Main variable.

13.6 Generic packages

The previous examples focused on generic subprograms. In this section, we look at generic packages. The syntax is similar to that used for generic subprograms: we start with the **generic** keyword and continue with formal declarations. The only difference is that **package** is specified instead of a subprogram keyword.

Here's an example:

Listing 12: element.ads

```
generic
1
      type T is private;
   package Element is
      procedure Set (E : T);
      procedure Reset;
6
      function Get return T;
7
      function Is_Valid return Boolean;
8
9
      Invalid_Element : exception;
10
11
   private
12
      Value : T;
13
      Valid : Boolean := False;
14
   end Element;
```

Listing 13: element.adb

```
package body Element is

procedure Set (E : T) is
begin
Value := E;
Valid := True;

(continues on next page)
```

```
end Set;
7
8
      procedure Reset is
10
      begin
          Valid := False;
11
      end Reset;
12
13
      function Get return T is
14
      begin
15
          if not Valid then
16
             raise Invalid_Element;
17
          end if;
18
          return Value;
19
      end Get;
20
      function Is_Valid return Boolean is (Valid);
22
   end Element;
23
```

Listing 14: show_generic_package.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   with Element;
2
   procedure Show_Generic_Package is
4
5
      package I is new Element (T => Integer);
6
7
8
      procedure Display Initialized is
      begin
         if I.Is_Valid then
10
             Put_Line ("Value is initialized");
11
12
             Put_Line ("Value is not initialized");
13
         end if:
14
      end Display_Initialized;
15
16
   begin
17
      Display_Initialized;
18
19
      Put_Line ("Initializing...");
20
      I.Set (5);
21
      Display_Initialized;
22
      Put_Line ("Value is now set to "
23
                 & Integer'Image (I.Get));
24
25
      Put Line ("Resetting...");
26
      I.Reset;
27
      Display_Initialized;
28
29
   end Show_Generic_Package;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Generic_Package
MD5: c5278a06c6d06f1f37353ee0ca6686ec
```

Runtime output

```
Value is not initialized
Initializing...
Value is initialized

(continues on next page)
```

```
Value is now set to 5
Resetting...
Value is not initialized
```

In the example above, we created a simple container named Element, with just one single element. This container tracks whether the element has been initialized or not.

After writing the package definition, we create the instance I of the Element. We use the instance by calling the package subprograms (Set, Reset, and Get).

13.7 Formal subprograms

In addition to formal types and objects, we can also declare formal subprograms or packages. This course only describes formal subprograms; formal packages are discussed in the advanced course.

We use the with keyword to declare a formal subprogram. In the example below, we declare a formal function (Comparison) to be used by the generic procedure Check.

Listing 15: check.ads

```
generic
Description: String;
type T is private;
with function Comparison (X, Y : T)
return Boolean;
procedure Check (X, Y : T);
```

Listing 16: check.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Check (X, Y : T) is
3
      Result : Boolean;
4
5
      Result := Comparison (X, Y);
6
      if Result then
7
         Put Line
8
            ("Comparison ("
             & Description
10
             & ") between arguments is OK!");
11
      else
12
         Put_Line
13
            ("Comparison ("
14
             & Description
15
            & ") between arguments is not OK!");
16
      end if;
17
   end Check;
18
```

Listing 17: show formal subprogram.adb

```
with Check;

procedure Show_Formal_Subprogram is

A, B : Integer;

procedure Check_Is_Equal is new
Check (Description => "equality",

(continues on next page)
```

Chapter 13. Generics

```
=> Integer,
                Comparison => Standard."=");
10
      -- Here, we are mapping the standard
11
      -- equality operator for Integer types to
      -- the Comparison formal function
13
   begin
      A := 0;
15
      B := 1;
16
      Check_Is_Equal (A, B);
17
   end Show_Formal_Subprogram;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Formal_Subprogram
MD5: 1c463a47e9ce56b5afbca1da6acd116d
```

Runtime output

```
Comparison (equality) between arguments is not OK!
```

13.8 Example: I/O instances

Ada offers generic I/O packages that can be instantiated for standard and derived types. One example is the generic Float_I0 package, which provides procedures such as Put and Get. In fact, Float_Text_I0 — available from the standard library — is an instance of the Float I0 package, and it's defined as:

```
with Ada.Text_I0;
package Ada.Float_Text_I0 is new Ada.Text_I0.Float_I0 (Float);
```

You can use it directly with any object of floating-point type. For example:

Listing 18: show float text io.adb

```
with Ada.Float_Text_IO;

procedure Show_Float_Text_IO is
    X : constant Float := 2.5;

use Ada.Float_Text_IO;
begin
    Put (X);
end Show_Float_Text_IO;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Float_Text_I0
MD5: 7cc9b547ef301a2071e9fb65caa4631b
```

Runtime output

```
2.50000E+00
```

Instantiating generic I/O packages can be useful for derived types. For example, let's create a new type Price that must be displayed with two decimal digits after the point, and no exponent.

Listing 19: show_float_io_inst.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Show Float IO Inst is
3
      type Price is digits 3;
5
6
      package Price IO is new
7
        Ada. Text IO. Float IO (Price);
8
9
      P : Price;
10
   begin
11
      -- Set to zero => don't display exponent
12
      Price_IO.Default_Exp := 0;
13
14
      P := 2.5:
15
      Price IO.Put (P);
16
      New Line;
17
18
      P := 5.75;
19
      Price_IO.Put (P);
20
      New Line;
21
   end Show_Float_IO_Inst;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Float_IO_Inst
MD5: 583c761421d7fdb812dd2a183b676bae
```

Runtime output

```
2.50
5.75
```

By adjusting Default_Exp from the Price_IO instance to *remove* the exponent, we can control how variables of Price type are displayed. Just as a side note, we could also have written:

```
type Price is new Float;

package Price_IO is new
   Ada.Text_IO.Float_IO (Price);

begin
   Price_IO.Default_Aft := 2;
   Price_IO.Default_Exp := 0;
```

In this case, we're ajusting Default_Aft, too, to get two decimal digits after the point when calling Put.

In addition to the generic Float_I0 package, the following generic packages are available from Ada.Text_I0:

- Enumeration IO for enumeration types;
- · Integer IO for integer types;
- Modular I0 for modular types;
- · Fixed I0 for fixed-point types;

• Decimal_IO for decimal types.

In fact, we could rewrite the example above using decimal types:

Listing 20: show_decimal_io_inst.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Show Decimal IO Inst is
      type Price is delta 10.0 ** (-2) digits 12;
5
      package Price_IO is new
        Ada. Text IO. Decimal IO (Price);
9
      P : Price;
10
   begin
11
      Price_IO.Default_Exp := 0;
12
13
      P := 2.5;
14
      Price IO.Put (P);
15
      New Line;
16
17
      P := 5.75;
18
      Price IO.Put (P);
19
      New_Line;
20
   end Show_Decimal_IO_Inst;
21
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Decimal_IO_Inst
MD5: f413570759dcb32cc166078b3cee1a16
```

Runtime output

```
2.50
5.75
```

13.9 Example: ADTs

An important application of generics is to model abstract data types (ADTs). In fact, Ada includes a library with numerous ADTs using generics: Ada.Containers (described in the containers section (page 199)).

A typical example of an ADT is a stack:

Listing 21: stacks.ads

```
generic
Max: Positive;
type T is private;
package Stacks is

type Stack is limited private;

Stack_Underflow, Stack_Overflow: exception;

function Is_Empty (S: Stack) return Boolean;

function Pop (S: in out Stack) return T;

(continues on next page)
```

```
procedure Push (S : in out Stack;
14
                        V :
15
16
   private
17
18
      type Stack_Array is
19
         array (Natural range <>) of T;
20
21
      Min : constant := 1;
22
23
       type Stack is record
24
          Container : Stack_Array (Min .. Max);
25
          Top
                     : Natural := Min - 1;
26
      end record;
   end Stacks;
```

Listing 22: stacks.adb

```
package body Stacks is
2
       function Is_Empty (S : Stack) return Boolean is
3
         (S.Top < S.Container'First);</pre>
4
5
       function Is_Full (S : Stack) return Boolean is
6
         (S.Top >= S.Container'Last);
7
8
9
       function Pop (S : in out Stack) return T is
10
       begin
          if Is_Empty (S) then
11
              raise Stack_Underflow;
12
          else
13
              \textbf{return}\ X\ :\ T\ \textbf{do}
14
                 X := S.Container (S.Top);
15
                 S.Top := S.Top - 1;
16
              end return;
17
          end if;
18
       end Pop;
19
20
       procedure Push (S : in out Stack;
21
22
       begin
23
          if Is_Full (S) then
24
              raise Stack_Overflow;
25
26
              S.Top
                                    := S.Top + 1;
27
              S.Container (S.Top) := V;
28
          end if;
29
       end Push;
30
31
   end Stacks;
```

Listing 23: show stack.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Stacks;

procedure Show_Stack is
package Integer_Stacks is new
```

```
Stacks (Max => 10,
7
                 T => Integer);
8
      use Integer_Stacks;
10
      Values : Integer_Stacks.Stack;
11
12
   begin
13
      Push (Values, 10);
14
      Push (Values, 20);
15
16
      Put Line ("Last value was "
17
                 & Integer'Image (Pop (Values)));
18
   end Show_Stack;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Show_Stack
MD5: ee112d395552c1a02d211b9e5425dc71
```

Runtime output

```
Last value was 20
```

In this example, we first create a generic stack package (Stacks) and then instantiate it to create a stack of up to 10 integer values.

13.10 Example: Swap

Let's look at a simple procedure that swaps variables of type Color:

Listing 24: colors.ads

```
package Colors is
type Color is (Black, Red, Green,
Blue, White);

procedure Swap_Colors (X, Y : in out Color);
end Colors;
```

Listing 25: colors.adb

```
package body Colors is
1
2
      procedure Swap_Colors (X, Y : in out Color) is
3
         Tmp : constant Color := X;
4
      begin
5
         X := Y;
6
         Y := Tmp;
7
      end Swap Colors;
8
  end Colors;
```

Listing 26: test non generic swap colors.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Colors; use Colors;

procedure Test_Non_Generic_Swap_Colors is
```

```
A, B, C : Color;
   begin
6
      A := Blue;
      B := White;
8
      C := Red;
9
10
      Put_Line ("Value of A is "
11
                 & Color'Image (A));
12
      Put_Line ("Value of B is
13
                 & Color'Image (B));
14
      Put_Line ("Value of C is
15
                 & Color'Image (C));
16
17
      New_Line;
18
      Put_Line ("Swapping A and C...");
      New Line;
20
      Swap_Colors (A, C);
21
22
      Put_Line ("Value of A is "
23
                 & Color'Image (A));
24
      Put_Line ("Value of B is
25
                 & Color'Image (B));
26
      Put_Line ("Value of C is '
27
                 & Color'Image (C));
28
   end Test_Non_Generic_Swap_Colors;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Test_Non_Generic_Swap_Colors
MD5: 4d1cf826a1676c3750a8aabd484ac71f
```

Runtime output

130

```
Value of A is BLUE
Value of B is WHITE
Value of C is RED

Swapping A and C...

Value of A is RED
Value of B is WHITE
Value of C is BLUE
```

In this example, Swap_Colors can only be used for the Color type. However, this algorithm can theoretically be used for any type, whether an enumeration type or a complex record type with many elements. The algorithm itself is the same: it's only the type that differs. If, for example, we want to swap variables of **Integer** type, we don't want to duplicate the implementation. Therefore, such an algorithm is a perfect candidate for abstraction using generics.

In the example below, we create a generic version of Swap_Colors and name it **Generic_**Swap. This generic version can operate on any type due to the declaration of formal type T.

Listing 27: generic_swap.ads

```
generic
type T is private;
procedure Generic_Swap (X, Y : in out T);
```

Listing 28: generic_swap.adb

```
procedure Generic_Swap (X, Y : in out T) is

Tmp : constant T := X;

begin

X := Y;
Y := Tmp;
end Generic_Swap;
```

Listing 29: colors.ads

Listing 30: test_swap_colors.adb

```
with Ada.Text_IO; use Ada.Text_IO;
1
   with Colors;
                       use Colors;
2
3
   procedure Test Swap Colors is
4
      A, B, C : Color;
5
   begin
      A := Blue;
      B := White;
8
      C := Red;
9
10
      Put_Line ("Value of A is "
11
                 & Color'Image (A));
12
      Put_Line ("Value of B is "
13
                 & Color'Image (B));
14
      Put_Line ("Value of C is
15
                 & Color'Image (C));
16
17
      New_Line;
18
      Put_Line ("Swapping A and C...");
19
20
      New_Line;
      Swap_Colors (A, C);
21
22
      Put_Line ("Value of A is "
23
                 & Color'Image (A));
24
       Put Line ("Value of B is
25
                 & Color'Image (B));
26
       Put_Line ("Value of C is
27
                 & Color'Image (C));
28
   end Test_Swap_Colors;
29
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Test_Swap_Colors
MD5: a5d94a40bd9d1c6736cc873f8b58e867
```

Runtime output

```
Value of A is BLUE
Value of B is WHITE
Value of C is RED

Swapping A and C...

Value of A is RED
Value of B is WHITE
Value of C is BLUE
```

As we can see in the example, we can create the same Swap_Colors procedure as we had in the non-generic version of the algorithm by declaring it as an instance of the generic Generic_Swap procedure. We specify that the generic T type will be mapped to the Color type by passing it as an argument to the Generic_Swap instantiation.

13.11 Example: Reversing

The previous example, with an algorithm to swap two values, is one of the simplest examples of using generics. Next we study an algorithm for reversing elements of an array. First, let's start with a non-generic version of the algorithm, one that works specifically for the Color type:

Listing 31: colors.ads

Listing 32: colors.adb

```
package body Colors is
      procedure Reverse_It (X : in out Color_Array)
3
      is
      begin
5
          for I in X'First ...
6
                   (X'Last + X'First) / 2 loop
7
             declare
8
                Tmp
                         : Color;
9
                X_Left : Color
10
                  renames X (I);
11
                X_Right : Color
12
                  renames X (X'Last + X'First - I);
13
             begin
                         := X_Left;
                Tmp
15
                X_Left := X_Right;
16
                X_Right := Tmp;
17
             end;
18
          end loop;
19
      end Reverse_It;
20
21
   end Colors;
```

Listing 33: test_non_generic_reverse_colors.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   with Colors;
                      use Colors;
2
3
   procedure Test_Non_Generic_Reverse_Colors is
      My Colors : Color Array (1 \dots 5) :=
6
         (Black, Red, Green, Blue, White);
   begin
9
      for C of My_Colors loop
10
         Put_Line ("My_Color: " & Color'Image (C));
11
      end loop;
12
13
      New Line;
14
      Put_Line ("Reversing My_Color...");
15
      New Line;
16
      Reverse_It (My_Colors);
17
18
      for C of My_Colors loop
19
         Put_Line ("My_Color: " & Color'Image (C));
20
      end loop;
21
22
   end Test_Non_Generic_Reverse_Colors;
23
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Test_Non_Generic_Reverse_Colors
MD5: 9b3a489d0bc0ecd79de6ba99fd7cd44f
```

Runtime output

```
My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Reversing My_Color...

My_Color: WHITE
My_Color: BLUE
My_Color: BLUE
My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK
```

The procedure Reverse_It takes an array of colors, starts by swapping the first and last elements of the array, and continues doing that with successive elements until it reaches the middle of array. At that point, the entire array has been reversed, as we see from the output of the test program.

To abstract this procedure, we declare formal types for three components of the algorithm:

- the elements of the array (Color type in the example)
- the range used for the array (Integer range in the example)
- the actual array type (Color Array type in the example)

This is a generic version of the algorithm:

Listing 34: generic_reverse.ads

```
generic
type T is private;
type Index is range <>;
type Array_T is
array (Index range <>) of T;
procedure Generic_Reverse (X : in out Array_T);
```

Listing 35: generic reverse.adb

```
procedure Generic_Reverse (X : in out Array_T) is
1
    begin
2
         for I in X'First ..
3
                     (X'Last + X'First) / 2 loop
4
5
             declare
                 Tmp
                            : T;
                 X_Left : T
                    renames X (I);
9
                 X_Right : T
                    renames X (X'Last + X'First - I);
10
             begin
11
                 \begin{array}{ll} \mathsf{Tmp} & := \ \mathsf{X\_Left}; \\ \mathsf{X\_Left} & := \ \mathsf{X\_Right}; \end{array}
12
13
                 X_Right := Tmp;
14
             end;
15
         end loop;
16
    end Generic_Reverse;
17
```

Listing 36: colors.ads

```
with Generic Reverse;
   package Colors is
      type Color is (Black, Red, Green,
5
                      Blue, White);
6
7
      type Color_Array is
8
        array (Integer range <>) of Color;
9
10
      procedure Reverse It is new
11
         Generic Reverse (T
                                   => Color,
12
                           Index => Integer,
13
                           Array_T => Color_Array);
14
15
   end Colors;
16
```

Listing 37: test_reverse_colors.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Colors; use Colors;

procedure Test_Reverse_Colors is

My_Colors : Color_Array (1 .. 5) :=
    (Black, Red, Green, Blue, White);

begin
for C of My_Colors loop
```

```
Put_Line ("My_Color: "
11
                     & Color'Image (C));
12
       end loop;
13
      New_Line;
15
      Put_Line ("Reversing My_Color...");
16
      New_Line;
17
      Reverse_It (My_Colors);
18
19
       for C of My_Colors loop
20
          Put_Line ("My_Color: "
21
                     & Color'Image (C));
22
       end loop;
23
   end Test_Reverse_Colors;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Test_Reverse_Colors
MD5: 9ef175c517d7574b4b65b24ba0027f1f
```

Runtime output

```
My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Reversing My_Color...

My_Color: WHITE
My_Color: BLUE
My_Color: BLUE
My_Color: BLUE
My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK
```

As mentioned above, we're abstracting three components of the algorithm:

- the T type abstracts the elements of the array
- the Index type abstracts the range used for the array
- the Array_T type abstracts the array type and uses the formal declarations of the T and Index types.

13.12 Example: Test application

In the previous example we've focused only on abstracting the reversing algorithm itself. However, we could have decided to also abstract our small test application. This could be useful if we, for example, decide to test other procedures that change elements of an array.

In order to do this, we again have to choose the elements to abstract. We therefore declare the following formal parameters:

- S: the string containing the array name
- a function Image that converts an element of type T to a string
- a procedure Test that performs some operation on the array

Note that Image and Test are examples of formal subprograms and S is an example of a formal object.

Here is a version of the test application making use of the generic Perform_Test procedure:

Listing 38: generic reverse.ads

```
generic
type T is private;
type Index is range <>;
type Array_T is
array (Index range <>) of T;
procedure Generic_Reverse (X : in out Array_T);
```

Listing 39: generic reverse.adb

```
procedure Generic_Reverse (X : in out Array_T) is
   begin
2
      for I in X'First ..
3
                (X'Last + X'First) / 2 loop
4
         declare
5
                     : T;
             Tmp
6
             X_Left : T
7
               renames X (I);
8
             X_Right : T
               renames X (X'Last + X'First - I);
10
11
         begin
12
             Tmp
                     := X_Left;
             X_Left := X_Right;
13
             X_Right := Tmp;
14
         end;
15
      end loop;
16
   end Generic_Reverse;
17
```

Listing 40: perform test.ads

```
generic
1
      type T is private;
2
      type Index is range <>;
3
      type Array T is
4
        array (Index range <>) of T;
      S : String;
      with function Image (E : T)
8
                            return String is <>;
      with procedure Test (X : in out Array_T);
9
   procedure Perform_Test (X : in out Array_T);
10
```

Listing 41: perform_test.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Perform_Test (X : in out Array_T) is
3
   begin
4
      for C of X loop
5
         Put Line (S & ": " & Image (C));
6
      end loop;
7
8
9
      New_Line;
      Put_Line ("Testing " & S & "...");
10
11
      New_Line;
      Test (X);
12
13
      for C of X loop
14
         Put_Line (S & ": " & Image (C));
15
```

```
end loop;
end Perform_Test;
```

Listing 42: colors.ads

```
with Generic Reverse;
1
   package Colors is
3
      type Color is (Black, Red, Green,
5
                      Blue, White);
6
      type Color_Array is
8
        array (Integer range <>) of Color;
9
10
      procedure Reverse It is new
11
        Generic Reverse (T
                                  => Color,
12
                          Index => Integer,
13
                          Array_T => Color_Array);
14
   end Colors;
```

Listing 43: test_reverse_colors.adb

```
with Colors;
                       use Colors;
   with Perform_Test;
   procedure Test_Reverse_Colors is
4
      procedure Perform_Test_Reverse_It is new
6
        Perform_Test (T
                               => Color,
7
                       Index => Integer,
8
9
                       Array_T => Color_Array,
                               => "My_Color",
10
                       Image => Color'Image,
11
12
                       Test
                               => Reverse_It);
13
      My_Colors : Color_Array (1 .. 5) :=
14
        (Black, Red, Green, Blue, White);
15
16
   begin
17
      Perform_Test_Reverse_It (My_Colors);
18
   end Test_Reverse_Colors;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Generics.Test_Reverse_Colors_2
MD5: 04640309f4f7e9f8bcff137d1a6f8733
```

Runtime output

```
My_Color: BLACK
My_Color: RED
My_Color: GREEN
My_Color: BLUE
My_Color: WHITE

Testing My_Color...

My_Color: WHITE
```

My_Color: BLUE
My_Color: GREEN
My_Color: RED
My_Color: BLACK

In this example, we create the procedure Perform_Test_Reverse_It as an instance of the generic procedure (Perform Test). Note that:

- For the formal Image function, we use the 'Image attribute of the Color type
- For the formal Test procedure, we reference the Reverse_Array procedure from the package.

CHAPTER

FOURTEEN

EXCEPTIONS

Ada uses exceptions for error handling. Unlike many other languages, Ada speaks about raising, not throwing, an exception and handling, not catching, an exception.

14.1 Exception declaration

Ada exceptions are not types, but instead objects, which may be peculiar to you if you're used to the way Java or Python support exceptions. Here's how you declare an exception:

Listing 1: exceptions.ads

```
package Exceptions is

My_Except : exception;

-- Like an object. *NOT* a type !

end Exceptions;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Exceptions.Show_Exception
MD5: 6201faeca9b029c790023856d2c8c419
```

Even though they're objects, you're going to use each declared exception object as a "kind" or "family" of exceptions. Ada does not require that a subprogram declare every exception it can potentially raise.

14.2 Raising an exception

To raise an exception of our newly declared exception kind, do the following:

Listing 2: main.adb

```
with Exceptions; use Exceptions;

procedure Main is
begin

raise My_Except;
-- Execution of current control flow
-- abandoned; an exception of kind
-- "My_Except" will bubble up until it
-- is caught.
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Exceptions.Show_Exception
MD5: 24b40ae1509722adf51c3dd0d3ea4fbe
```

Runtime output

```
raised EXCEPTIONS.MY_EXCEPT : main.adb:5
```

Here, the My Except exception is raised. We can also specify a message:

Listing 3: main.adb

```
with Exceptions; use Exceptions;
  procedure Main is
3
  begin
4
      raise My_Except with "My exception message";
5
      -- Execution of current control flow
6
          abandoned; an exception of kind
7
          "My Except" with associated string will
8
      -- bub\overline{b}le up until it is caught.
9
  end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Exceptions.Show_Exception
MD5: 279299c9703c3ed4e51fdd7c3a5e1392
```

Runtime output

```
raised EXCEPTIONS.MY_EXCEPT : My exception message
```

In this case, we see an additional message when the exception is displayed.

14.3 Handling an exception

Next, we address how to handle exceptions that were raised by us or libraries that we call. The neat thing in Ada is that you can add an exception handler to any statement block as follows:

Listing 4: open_file.adb

```
with Ada.Text_IO; use Ada.Text_IO;
  with Ada.Exceptions; use Ada.Exceptions;
2
3
   procedure Open_File is
      File : File_Type;
5
   begin
6
          Block (sequence of statements)
      begin
8
         Open (File, In_File, "input.txt");
9
      exception
10
         when E : Name Error =>
11
                  ^ Exception to be handled
12
            Put ("Cannot open input file : ");
13
            Put_Line (Exception_Message (E));
14
15
             raise;
             -- Reraise current occurence
16
17
      end:
  end Open_File;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Exceptions.Show_Exception_Handling
MD5: 4eald5da684a6d7d7ee32908810e9c8f
```

Runtime output

```
Cannot open input file : input.txt: No such file or directory raised ADA.IO_EXCEPTIONS.NAME_ERROR : input.txt: No such file or directory
```

In the example above, we're using the Exception_Message function from the Ada. Exceptions package. This function returns the message associated with the exception as a string.

You don't need to introduce a block just to handle an exception: you can add it to the statements block of your current subprogram:

Listing 5: open file.adb

```
with Ada. Text IO; use Ada. Text IO;
   with Ada.Exceptions; use Ada.Exceptions;
3
   procedure Open File is
      File : File Type;
6
      Open (File, In File, "input.txt");
     Exception block can be added to any block
8
   exception
9
      when Name Error =>
10
         Put ("Cannot open input file");
11
  end Open File;
12
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Exceptions.Show_Exception_Message
MD5: 838e87ae416b3a717901cdc00eb71b40
```

Runtime output

```
Cannot open input file
```

6 Attention

Exception handlers have an important restriction that you need to be careful about: Exceptions raised in the declarative section are not caught by the handlers of that block. So for example, in the following code, the exception will not be caught.

```
Listing 6: be careful.adb
   with Ada. Text IO; use Ada. Text IO;
   with Ada. Exceptions; use Ada. Exceptions;
2
   procedure Be Careful is
      function Dangerous return Integer is
5
      begin
6
          raise Constraint Error;
          return 42;
      end Dangerous;
9
10
   begin
11
      declare
12
          A : Integer := Dangerous;
```

```
begin
14
         Put Line (Integer'Image (A));
15
      exception
16
         when Constraint_Error =>
17
            Put_Line ("error!");
18
      end:
19
   end Be Careful;
   Code block metadata
   Project: Courses.Intro_To_Ada.Exceptions.Be_Careful
   MD5: 6ea8a214bbbaca09d7444136d069e782
   Runtime output
   raised CONSTRAINT_ERROR : be_careful.adb:7 explicit raise
   This is also the case for the top-level exception block that is part of the current subpro-
   gram.
```

14.4 Predefined exceptions

Ada has a very small number of predefined exceptions:

- Constraint Error is the main one you might see. It's raised:
 - When bounds don't match or, in general, any violation of constraints.
 - In case of overflow
 - In case of null dereferences
 - In case of division by 0
- Program_Error might appear, but probably less often. It's raised in more arcane situations, such as for order of elaboration issues and some cases of detectable erroneous execution.
- Storage_Error will happen because of memory issues, such as:
 - Not enough memory (allocator)
 - Not enough stack
- Tasking_Error will happen with task related errors, such as any error happening during task activation.

You should not reuse predefined exceptions. If you do then, it won't be obvious when one is raised that it is because something went wrong in a built-in language operation.

CHAPTER

FIFTEEN

TASKING

Tasks and protected objects allow the implementation of concurrency in Ada. The following sections explain these concepts in more detail.

15.1 Tasks

A task can be thought as an application that runs *concurrently* with the main application. In other programming languages, a task might be called a thread¹⁸, and tasking might be called multithreading¹⁹.

Tasks may synchronize with the main application but may also process information completely independently from the main application. Here we show how this is accomplished.

15.1.1 Simple task

Tasks are declared using the keyword **task**. The task implementation is specified in a **task body** block. For example:

Listing 1: show simple task.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Show_Simple_Task is
3
      task T;
4
5
      task body T is
6
      begin
7
         Put Line ("In task T");
8
      end T;
9
   begin
10
      Put_Line ("In main");
11
   end Show Simple Task;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Simple_Task
MD5: b17d9b35b4b2b53bc59776749e1be219
```

Runtime output

```
In task T
In main
```

¹⁸ https://en.wikipedia.org/wiki/Thread (computing)

¹⁹ https://en.wikipedia.org/wiki/Thread_(computing)#Multithreading

Here, we're declaring and implementing the task T. As soon as the main application starts, task T starts automatically — it's not necessary to manually start this task. By running the application above, we can see that both calls to Put Line are performed.

Note that:

- The main application is itself a task (the main or "environment" task).
 - In this example, the subprogram Show_Simple_Task is the main task of the application.
- Task T is a subtask.
 - Each subtask has a master, which represents the program construct in which the subtask is declared. In this case, the main subprogram Show_Simple_Task is T 's master.
 - The master construct is executed by some enclosing task, which we will refer to as the "master task" of the subtask.
- The number of tasks is not limited to one: we could include a task T2 in the example above.
 - This task also starts automatically and runs *concurrently* with both task T and the main task. For example:

Listing 2: show_simple_tasks.adb

```
with Ada.Text IO; use Ada.Text IO;
1
2
   procedure Show_Simple_Tasks is
3
4
       task T;
       task T2;
5
6
       task body T is
7
       begin
8
          Put Line ("In task T");
9
       end T;
10
11
       task body T2 is
12
13
          Put Line ("In task T2");
14
       end T2;
15
16
   beain
17
       Put Line ("In main");
18
   end Show Simple Tasks;
19
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Multiple_Simple_Task
MD5: 5e24b797e742bec306ad498f4f40d2b4
```

Runtime output

```
In task T
In task T2
In main
```

15.1.2 Simple synchronization

As we've just seen, as soon as the master construct reaches its "begin", its subtasks also start automatically. The master continues its processing until it has nothing more to do. At that point, however, it will not terminate. Instead, the master waits until its subtasks have

finished before it allows itself to complete. In other words, this waiting process provides synchronization between the master task and its subtasks. After this synchronization, the master construct will complete. For example:

Listing 3: show_simple_sync.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_Simple_Sync is
3
      task T;
4
      task body T is
5
      begin
6
          for I in 1 .. 10 loop
7
             Put_Line ("hello");
         end loop;
      end T;
10
   begin
11
      null:
12
       -- Will wait here until all tasks
13
          have terminated
14
   end Show_Simple_Sync;
15
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Simple_Sync
MD5: 84afce465854f99f8cbe0b57714d8a5f
```

Runtime output

```
hello
```

The same mechanism is used for other subprograms that contain subtasks: the subprogram execution will wait for its subtasks to finish. So this mechanism is not limited to the main subprogram and also applies to any subprogram called by the main subprogram, directly or indirectly.

Synchronization also occurs if we move the task to a separate package. In the example below, we declare a task T in the package Simple Sync Pkg.

Listing 4: simple_sync_pkg.ads

```
package Simple_Sync_Pkg is
task T;
end Simple_Sync_Pkg;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Simple_Sync_Pkg
MD5: 2f9be044d04994240970f150e2293d5e
```

This is the corresponding package body:

15.1. Tasks 145

Listing 5: simple_sync_pkg.adb

```
with Ada.Text_IO; use Ada.Text_IO;
  package body Simple_Sync_Pkg is
3
      task body T is
4
      begin
5
         for I in 1 .. 10 loop
6
            Put_Line ("hello");
7
         end loop;
8
      end T;
9
  end Simple_Sync_Pkg;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Simple_Sync_Pkg
MD5: b668451e4fb10e802f619889bcd743ff
```

Because the package is with'ed by the main procedure, the task T defined in the package will become a subtask of the main task. For example:

Listing 6: test simple sync pkg.adb

```
with Simple_Sync_Pkg;

procedure Test_Simple_Sync_Pkg is
begin
null;
-- Will wait here until all tasks
-- have terminated
end Test_Simple_Sync_Pkg;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Simple_Sync_Pkg
MD5: e51565b91767ce198496ef3e9c582ac8
```

Runtime output

```
hello
```

As soon as the main subprogram returns, the main task synchronizes with any subtasks spawned by packages T from Simple_Sync_Pkg before finally terminating.

15.1.3 Delay

We can introduce a delay by using the keyword **delay**. This puts the current task to sleep for the length of time (in seconds) specified in the delay statement. For example:

Listing 7: show_delay.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Show Delay is
3
4
      task T;
5
6
       task body T is
7
      begin
8
          for I in 1 .. 5 loop
9
             Put_Line ("hello from task T");
10
             delay 1.0;
11
                    ^ Wait 1.0 seconds
12
          end loop;
13
      end T;
14
   begin
15
       delay 1.5;
16
       Put Line ("hello from main");
17
   end Show Delay;
18
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Delay
MD5: 4a6e8039744301a128e8fb2dd27902a5
```

Runtime output

```
hello from task T
hello from task T
hello from main
hello from task T
hello from task T
hello from task T
```

In this example, we're making the task T wait one second after each time it displays the "hello" message. In addition, the main task is waiting 1.5 seconds before displaying its own "hello" message

15.1.4 Synchronization: rendezvous

The only type of synchronization we've seen so far is the one that happens automatically at the end of a master construct with a subtask. You can also define custom synchronization points using the keyword **entry**. An *entry* can be viewed as a special kind of subprogram, which is called by another task using a similar syntax, as we will see later.

In the task body definition, you define which part of the task will accept the entries by using the keyword **accept**. A task proceeds until it reaches an **accept** statement and then waits for some other task to synchronize with it. Specifically,

- The task with the entry waits at that point (in the **accept** statement), ready to accept a call to the corresponding entry from the master task.
- The other task calls the task entry, in a manner similar to a procedure call, to synchronize with the entry.

This synchronization between tasks is called a rendezvous. Let's see an example:

15.1. Tasks 147

Listing 8: show_rendezvous.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Show Rendezvous is
3
       task T is
          entry Start;
      end T;
8
      task body T is
9
      begin
10
          accept Start;
11
                  ^ Waiting for somebody
12
                    to call the entry
13
14
          Put Line ("In T");
15
      end T;
16
17
   begin
18
      Put_Line ("In Main");
19
20
       -- Calling T's entry:
21
      T.Start;
22
   end Show Rendezvous;
23
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Rendezvous
MD5: 479eea7adc876ac359ad20ac6e3acf66
```

Runtime output

```
In Main
In T
```

In this example, we declare an entry Start for task T. In the task body, we implement this entry using **accept** Start. When task T reaches this point, it waits for some other task to call its entry. This synchronization occurs in the T.Start statement. After the rendezvous completes, the main task and task T again run concurrently until they synchronize one final time when the main subprogram Show_Rendezvous finishes.

An entry may be used to perform more than a simple task synchronization: it also may perform multiple statements during the time both tasks are synchronized. We do this with a do ... end block. For the previous example, we would simply write accept Start do <statements>; end;. We use this kind of block in the next example.

15.1.5 Select loop

There's no limit to the number of times an entry can be accepted. We could even create an infinite loop in the task and accept calls to the same entry over and over again. An infinite loop, however, prevents the subtask from finishing, so it blocks its master task when it reaches the end of its processing. Therefore, a loop containing accept statements in a task body can be used in conjunction with a select ... or terminate statement. In simple terms, this statement allows its master task to automatically terminate the subtask when the master construct reaches its end. For example:

Listing 9: show rendezvous loop.adb

```
with Ada.Text_I0; use Ada.Text_I0;
(continues on next page)
```

```
procedure Show_Rendezvous_Loop is
3
       task T is
5
          entry Reset;
          entry Increment;
7
       end T;
8
9
       task body T is
10
          Cnt : Integer := 0;
11
       begin
12
          loop
13
             select
14
15
                 accept Reset do
16
                   Cnt := 0;
                 end Reset;
                 Put_Line ("Reset");
18
             or
19
                 accept Increment do
20
                    Cnt := Cnt + 1;
21
                 end Increment;
22
                 Put_Line ("In T's loop ("
23
                            & Integer'Image (Cnt)
24
                            & ")");
25
              or
26
                 terminate;
27
             end select;
28
          end loop;
29
       end T;
30
31
   begin
32
       Put_Line ("In Main");
33
34
       for I in 1 .. 4 loop
35
              Calling T's entry multiple times
36
          T.Increment;
37
       end loop;
38
39
       T.Reset;
40
       for I in 1 \dots 4 loop
41
          -- Calling T's entry multiple times
42
          T.Increment;
43
       end loop;
44
45
   end Show_Rendezvous_Loop;
46
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Rendezvous_Loop
MD5: 0542dbc029cffb9f794d761bab9f3a9d
```

Runtime output

```
In Main
In T's loop ( 1)
In T's loop ( 2)
In T's loop ( 3)
In T's loop ( 4)
Reset
In T's loop ( 1)
In T's loop ( 2)
In T's loop ( 3)

(continues on next page)
```

15.1. Tasks 149

```
In T's loop (4)
```

In this example, the task body implements an infinite loop that accepts calls to the Reset and Increment entry. We make the following observations:

- The accept E do ... end block is used to increment a counter.
 - As long as task T is performing the **do ... end** block, the main task waits for the block to complete.
- The main task is calling the Increment entry multiple times in the loop from 1 .. 4. It is also calling the Reset entry before the second loop.
 - Because task T contains an infinite loop, it always accepts calls to the Reset and Increment entries.
 - When the master construct of the subtask (the Show_Rendezvous_Loop subprogram) completes, it checks the status of the T task. Even though task T could accept new calls to the Reset or Increment entries, the master construct is allowed to terminate task T due to the **or terminate** part of the **select** statement.

15.1.6 Cycling tasks

In a previous example, we saw how to delay a task a specified time by using the **delay** keyword. However, using delay statements in a loop is not enough to guarantee regular intervals between those delay statements. For example, we may have a call to a computationally intensive procedure between executions of successive delay statements:

In this case, we can't guarantee that exactly 10 seconds have elapsed after 10 calls to the delay statement because a time drift may be introduced by the Computational_Intensive_App procedure. In many cases, this time drift is not relevant, so using the **delay** keyword is good enough.

However, there are situations where a time drift isn't acceptable. In those cases, we need to use the **delay until** statement, which accepts a precise time for the end of the delay, allowing us to define a regular interval. This is useful, for example, in real-time applications.

We will soon see an example of how this time drift may be introduced and how the **delay until** statement circumvents the problem. But before we do that, we look at a package containing a procedure allowing us to measure the elapsed time (Show_Elapsed_Time) and a dummy Computational_Intensive_App procedure which is simulated by using a simple delay. This is the complete package:

Listing 10: delay_aux_pkg.ads

```
with Ada.Real_Time; use Ada.Real_Time;

package Delay_Aux_Pkg is

function Get_Start_Time return Time
with Inline;

procedure Show_Elapsed_Time
with Inline;
```

(continues on next page)

```
procedure Computational_Intensive_App;
private

Start_Time : Time := Clock;

function Get_Start_Time return Time is
    (Start_Time);

end Delay_Aux_Pkg;
```

Listing 11: delay_aux_pkg.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   package body Delay_Aux_Pkg is
3
      procedure Show Elapsed Time is
5
         Now Time
                     : Time;
6
         Elapsed Time : Time Span;
      begin
         Now Time
                      := Clock;
         Elapsed_Time := Now_Time - Start_Time;
10
         Put_Line ("Elapsed time "
11
                    & Duration'Image
12
                        (To_Duration (Elapsed_Time))
13
                    & " seconds");
14
      end Show_Elapsed_Time;
15
16
      procedure Computational Intensive App is
17
18
      begin
         delay 0.5;
19
      end Computational_Intensive_App;
20
21
   end Delay_Aux_Pkg;
22
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Time
MD5: 422a38clafa0bbd659ec81de88479e0a
```

Using this auxiliary package, we're now ready to write our time-drifting application:

Listing 12: show_time_task.adb

```
with Ada.Text_I0;
                         use Ada.Text_I0;
   with Ada.Real_Time; use Ada.Real_Time;
2
   with Delay_Aux_Pkg;
4
   procedure Show_Time_Task is
6
      package Aux renames Delay Aux Pkg;
7
8
      task T;
9
10
       task body T is
11
          Cnt
                : Integer := 1;
12
      begin
13
          for I in 1 .. 5 loop
14
             delay 1.0;
15
16
             Aux. Show Elapsed Time;
17
             Aux.Computational_Intensive_App;
18
                                                                           (continues on next page)
```

15.1. Tasks 151

```
19
              Put_Line ("Cycle # "
20
                        & Integer'Image (Cnt));
21
             Cnt := Cnt + 1;
          end loop;
23
          Put_Line ("Finished time-drifting loop");
24
       end T;
25
26
   begin
27
       null;
28
   end Show_Time_Task;
29
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Time
MD5: fe17c902fc127c0132677ea4005ff3f1
```

Runtime output

```
Elapsed time 1.001227158 seconds
Cycle # 1
Elapsed time 2.504245558 seconds
Cycle # 2
Elapsed time 4.004753280 seconds
Cycle # 3
Elapsed time 5.506642053 seconds
Cycle # 4
Elapsed time 7.007136602 seconds
Cycle # 5
Finished time-drifting loop
```

We can see by running the application that we already have a time difference of about four seconds after three iterations of the loop due to the drift introduced by Computational_Intensive_App. Using the **delay until** statement, however, we're able to avoid this time drift and have a regular interval of exactly one second:

Listing 13: show_time_task.adb

```
with Ada.Text_I0;
                         use Ada.Text_I0;
   with Ada.Real_Time; use Ada.Real_Time;
2
   with Delay_Aux_Pkg;
   procedure Show_Time_Task is
6
      package Aux renames Delay Aux Pkg;
7
8
      task T;
9
10
      task body T is
11
          Cycle : constant Time_Span :=
12
            Milliseconds (1000);
13
         Next : Time := Aux.Get_Start_Time
14
                           + Cycle;
15
16
          Cnt
                : Integer := 1;
17
      begin
18
          for I in 1 .. 5 loop
19
             delay until Next;
20
21
             Aux.Show_Elapsed_Time;
22
             Aux.Computational_Intensive_App;
23
```

(continues on next page)

```
24
             -- Calculate next execution time
25
             -- using a cycle of one second
26
             Next := Next + Cycle;
28
             Put_Line ("Cycle # "
29
                        & Integer'Image (Cnt));
30
             Cnt := Cnt + 1;
31
          end loop;
32
          Put_Line ("Finished cycling");
33
      end T;
34
35
   begin
36
37
      null;
   end Show_Time_Task;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Time
MD5: 1456c0feee6def8b370d994c0ab75a15
```

Runtime output

```
Elapsed time 1.001306057 seconds
Cycle # 1
Elapsed time 2.002074211 seconds
Cycle # 2
Elapsed time 3.001603021 seconds
Cycle # 3
Elapsed time 4.001294033 seconds
Cycle # 4
Elapsed time 5.000124181 seconds
Cycle # 5
Finished cycling
```

Now, as we can see by running the application, the **delay until** statement ensures that the Computational_Intensive_App doesn't disturb the regular interval of one second between iterations.

15.2 Protected objects

When multiple tasks are accessing shared data, corruption of that data may occur. For example, data may be inconsistent if one task overwrites parts of the information that's being read by another task at the same time. In order to avoid these kinds of problems and ensure information is accessed in a coordinated way, we use *protected objects*.

Protected objects encapsulate data and provide access to that data by means of *protected operations*, which may be subprograms or protected entries. Using protected objects ensures that data is not corrupted by race conditions or other concurrent access.

Important

Objects can be protected from concurrent access using Ada tasks. In fact, this was the *only* way of protecting objects from concurrent access in Ada 83 (the first version of the Ada language). However, the use of protected objects is much simpler than using similar mechanisms implemented using only tasks. Therefore, you should use protected objects when your main goal is only to protect data.

15.2.1 Simple object

You declare a protected object with the **protected** keyword. The syntax is similar to that used for packages: you can declare operations (e.g., procedures and functions) in the public part and data in the private part. The corresponding implementation of the operations is included in the **protected body** of the object. For example:

Listing 14: show protected objects.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_Protected_Objects is
      protected Obj is
5
          -- Operations go here (only subprograms)
6
          procedure Set (V : Integer);
7
          function Get return Integer;
8
      private
9
          -- Data goes here
10
         Local : Integer := 0;
11
      end Obj;
12
13
      protected body Obj is
14
          -- procedures can modify the data
15
          procedure Set (V : Integer) is
16
          begin
17
             Local := V;
18
         end Set;
19
20
          -- functions cannot modify the data
21
          function Get return Integer is
22
         begin
23
            return Local;
24
         end Get;
25
      end Obj;
26
27
   begin
28
      Obj.Set (5);
29
      Put_Line ("Number is: "
30
                 & Integer'Image (Obj.Get));
31
   end Show Protected Objects;
32
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Protected_Objects
MD5: dd97dd584ba2f13def3c04725d4e48a7
```

Runtime output

```
Number is: 5
```

In this example, we define two operations for Obj: Set and Get. The implementation of these operations is in the Obj body. The syntax used for writing these operations is the same as that for normal procedures and functions. The implementation of protected objects is straightforward — we simply access and update Local in these subprograms. To call these operations in the main application, we use prefixed notation, e.g., Obj.Get.

15.2.2 Entries

In addition to protected procedures and functions, you can also define protected entry points. Do this using the **entry** keyword. Protected entry points allow you to define barriers using the **when** keyword. Barriers are conditions that must be fulfilled before the entry can start performing its actual processing — we speak of *releasing* the barrier when the condition is fulfilled.

The previous example used procedures and functions to define operations on the protected objects. However, doing so permits reading protected information (via 0bj.Get) before it's set (via 0bj.Set). To allow that to be a defined operation, we specified a default value (0). Instead, by rewriting 0bj.Get using an *entry* instead of a function, we implement a barrier, ensuring no task can read the information before it's been set.

The following example implements the barrier for the Obj.Get operation. It also contains two concurrent subprograms (main task and task T) that try to access the protected object.

Listing 15: show protected objects entries.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Show Protected Objects Entries is
3
       protected Obj is
5
          procedure Set (V : Integer);
          entry Get (V : out Integer);
      private
8
                 : Integer;
          Local
a
          Is Set : Boolean := False;
10
      end Obj;
11
12
       protected body Obj is
13
          procedure Set (V : Integer) is
14
          begin
15
             Local := V;
16
             Is Set := True;
17
          end Set;
18
          entry Get (V : out Integer)
20
            when Is Set is
21
             -- Entry is blocked until the
22
             -- condition is true. The barrier
23
             -- is evaluated at call of entries
24
             -- and at exits of procedures and
25
             -- entries. The calling task sleeps
26
             -- until the barrier is released.
27
          begin
28
             V := Local;
29
             Is_Set := False;
30
          end Get:
31
      end Obj;
32
33
      N : Integer := 0;
34
35
       task T;
36
37
      task body T is
38
      begin
39
40
            ("Task T will delay for 4 seconds...");
41
          delay 4.0;
42
43
          Put Line
44
                                                                            (continues on next page)
```

```
("Task T will set Obj...");
45
          Obj.Set (5);
46
47
          Put Line
            ("Task T has just set Obj...");
49
      end T;
50
   beain
51
      Put Line
52
         ("Main application will get Obj...");
53
      Obj.Get (N);
54
55
      Put Line
56
         ("Main application has retrieved Obj...");
57
      Put_Line
58
         ("Number is: " & Integer'Image (N));
   end Show_Protected_Objects_Entries;
61
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Protected_Objects_Entries
MD5: c1134445a96700b871fb76c4d6342359
```

Runtime output

```
Task T will delay for 4 seconds...
Main application will get Obj...
Task T will set Obj...
Task T has just set Obj...
Main application has retrieved Obj...
Number is: 5
```

As we see by running it, the main application waits until the protected object is set (by the call to 0bj . Set in task T) before it reads the information (via 0bj . Get). Because a 4-second delay has been added in task T, the main application is also delayed by 4 seconds. Only after this delay does task T set the object and release the barrier in 0bj . Get so that the main application can then resume processing (after the information is retrieved from the protected object).

15.3 Task and protected types

In the previous examples, we defined single tasks and protected objects. We can, however, generalize tasks and protected objects using type definitions. This allows us, for example, to create multiple tasks based on just a single task type.

15.3.1 Task types

A task type is a generalization of a task. The declaration is similar to simple tasks: you replace **task** with **task type**. The difference between simple tasks and task types is that task types don't create actual tasks that automatically start. Instead, a task object declaration is needed. This is exactly the way normal variables and types work: objects are only created by variable definitions, not type definitions.

To illustrate this, we repeat our first example:

Listing 16: show_simple_task.adb

```
with Ada.Text_IO; use Ada.Text_IO;
2
   procedure Show_Simple_Task is
3
      task T;
4
5
      task body T is
6
      begin
7
         Put Line ("In task T");
8
      end T;
9
   begin
10
      Put_Line ("In main");
11
   end Show Simple Task;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Simple_Task
MD5: b17d9b35b4b2b53bc59776749e1be219
```

Runtime output

```
In task T
In main
```

We now rewrite it by replacing **task** T with **task type TT**. We declare a task (A_Task) based on the task type TT after its definition:

Listing 17: show simple task type.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_Simple_Task_Type is
      task type TT;
      task body TT is
6
      begin
7
         Put_Line ("In task type TT");
8
      end TT;
9
10
      A_Task : TT;
11
   begin
12
      Put_Line ("In main");
13
   end Show_Simple_Task_Type;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Simple_Task_Type
MD5: 24c26dcbba6f5c54f0a7d47c3c0da728
```

Runtime output

```
In task type TT
In main
```

We can extend this example and create an array of tasks. Since we're using the same syntax as for variable declarations, we use a similar syntax for task types: **array** (<>) **of Task**_Type. Also, we can pass information to the individual tasks by defining a Start entry. Here's the updated example:

Listing 18: show_task_type_array.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Show_Task_Type_Array is
3
       task type TT is
4
          entry Start (N : Integer);
5
      end TT;
6
7
      task body TT is
8
          Task N : Integer;
9
      begin
10
          accept Start (N : Integer) do
11
             Task_N := N;
12
          end Start;
13
          Put_Line ("In task T: "
14
                     & Integer'Image (Task_N));
15
      end TT;
16
17
      My Tasks: array (1 ... 5) of TT;
18
19
      Put_Line ("In main");
20
21
       for I in My_Tasks'Range loop
22
         My_Tasks (I).Start (I);
23
      end loop;
24
   end Show_Task_Type_Array;
25
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Task_Type_Array
MD5: bba072dfc52fb2bfbef6e7b9f8191464
```

Runtime output

```
In main
In task T: 1
In task T: 2
In task T: 3
In task T: 4
In task T: 5
```

In this example, we're declaring five tasks in the array My_Tasks. We pass the array index to the individual tasks in the entry point (Start). After the synchronization between the individual subtasks and the main task, each subtask calls Put_Line concurrently.

15.3.2 Protected types

A protected type is a generalization of a protected object. The declaration is similar to that for protected objects: you replace **protected** with **protected type**. Like task types, protected types require an object declaration to create actual objects. Again, this is similar to variable declarations and allows for creating arrays (or other composite objects) of protected objects.

We can reuse a previous example and rewrite it to use a protected type:

Listing 19: show_protected_object_type.adb

```
with Ada.Text_I0; use Ada.Text_I0;
procedure Show_Protected_Object_Type is
(continues on next page)
```

```
protected type P_Obj_Type is
5
          procedure Set (V : Integer);
6
          function Get return Integer;
      private
8
          Local : Integer := 0;
9
      end P_Obj_Type;
10
11
      protected body P_Obj_Type is
12
          procedure Set (V : Integer) is
13
          begin
14
             Local := V;
15
          end Set;
16
          function Get return Integer is
18
          begin
19
             return Local;
20
          end Get;
21
      end P_Obj_Type;
22
23
      Obj : P_Obj_Type;
24
   begin
25
      Obj.Set (5);
26
       Put_Line ("Number is: "
27
                 & Integer'Image (Obj.Get));
   end Show_Protected_Object_Type;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Tasking.Show_Protected_Object_Type
MD5: c50321e55afef0d72f263fee0669e55f
```

Runtime output

```
Number is: 5
```

In this example, instead of directly defining the protected object 0bj, we first define a protected type P_0bj_Type and then declare 0bj as an object of that protected type. Note that the main application hasn't changed: we still use 0bj.Set and 0bj.Get to access the protected object, just like in the original example.

DESIGN BY CONTRACTS

Contracts are used in programming to codify expectations. Parameter modes of a subprogram can be viewed as a simple form of contracts. When the specification of subprogram 0p declares a parameter using <code>in</code> mode, the caller of 0p knows that the <code>in</code> argument won't be changed by 0p. In other words, the caller expects that 0p doesn't modify the argument it's providing, but just reads the information stored in the argument. Constraints and subtypes are other examples of contracts. In general, these specifications improve the consistency of the application.

Design-by-contract programming refers to techniques that include pre- and postconditions, subtype predicates, and type invariants. We study those topics in this chapter.

16.1 Pre- and postconditions

Pre- and postconditions provide expectations regarding input and output parameters of subprograms and return value of functions. If we say that certain requirements must be met before calling a subprogram 0p, those are preconditions. Similarly, if certain requirements must be met after a call to the subprogram 0p, those are postconditions. We can think of preconditions and postconditions as promises between the subprogram caller and the callee: a precondition is a promise from the caller to the callee, and a postcondition is a promise in the other direction.

Pre- and postconditions are specified using an aspect clause in the subprogram declaration. A with Pre => <condition> clause specifies a precondition and a with Post => <condition> clause specifies a postcondition.

The following code shows an example of preconditions:

Listing 1: show simple precondition.adb

```
procedure Show_Simple_Precondition is
2
      procedure DB_Entry (Name : String;
3
                            Age : Natural)
4
        with Pre => Name'Length > 0
5
      is
6
      begin
7
             Missing implementation
8
         null;
10
      end DB Entry;
11
   begin
      DB_Entry ("John", 30);
12
13
      -- Precondition will fail!
14
      DB_Entry ("",
                          21);
15
   end Show Simple Precondition;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Contracts.Show_Simple_Precondition MD5: 87b6e080555603111801a0fcd2469acd
```

Runtime output

In this example, we want to prevent the name field in our database from containing an empty string. We implement this requirement by using a precondition requiring that the length of the string used for the Name parameter of the DB_Entry procedure is greater than zero. If the DB_Entry procedure is called with an empty string for the Name parameter, the call will fail because the precondition is not met.

1 In the GNAT toolchain

GNAT handles pre- and postconditions by generating runtime assertions for them. By default, however, assertions aren't enabled. Therefore, in order to check pre- and postconditions at runtime, you need to enable assertions by using the *-gnata* switch.

Before we get to our next example, let's briefly discuss quantified expressions, which are quite useful in concisely writing pre- and postconditions. Quantified expressions return a Boolean value indicating whether elements of an array or container match the expected condition. They have the form: (for all I in A'Range => <condition on A(I)>, where A is an array and I is an index. Quantified expressions using for all check whether the condition is true for every element. For example:

```
(for all I in A'Range => A (I) = 0)
```

This quantified expression is only true when all elements of the array A have a value of zero.

Another kind of quantified expressions uses **for some**. The form looks similar: (**for some** I **in** A'Range => <condition on A(I)>. However, in this case the qualified expression tests whether the condition is true only on *some* elements (hence the name) instead of all elements.

We illustrate postconditions using the following example:

Listing 2: show simple postcondition.adb

```
with Ada. Text IO; use Ada. Text IO;
   procedure Show Simple Postcondition is
      type Int_8 is range -2 ** 7 .. 2 ** 7 - 1;
5
6
      type Int 8 Array is
7
         array (Integer range <>) of Int_8;
8
9
      function Square (A : Int_8) return Int_8 is
10
11
           with Post \Rightarrow (if abs A in 0 | 1
12
                          then Square'Result = abs A
13
                          else Square'Result > A);
14
15
      procedure Square (A : in out Int_8_Array)
16
         with Post => (for all I in A'Range =>
17
                          A (I) = A'Old (I) *
18
                                   A'01d (I))
19
```

(continues on next page)

```
is
20
       begin
21
          for V of A loop
22
             V := Square (V);
          end loop;
24
       end Square;
25
26
       V : Int_8_Array := (-2, -1, 0, 1, 10, 11);
27
   begin
28
       for E of V loop
29
          Put_Line ("Original: "
30
                     & Int_8'Image (E));
31
       end loop;
32
       New_Line;
33
       Square (V);
35
       for E of V loop
36
          Put_Line ("Square:
37
                     & Int_8'Image (E));
38
       end loop;
39
   end Show_Simple_Postcondition;
40
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Contracts.Show_Simple_Postcondition MD5: b9bae9fe09cefcbe6769ad9cd6739e2a
```

Runtime output

```
Original: -2
Original: -1
Original: 0
Original:
           1
Original:
           10
Original: 11
Square:
           4
Square:
           1
Square:
           0
Square:
           1
           100
Square:
Square:
           121
```

We declare a signed 8-bit type Int_8 and an array of that type (Int_8_Array). We want to ensure each element of the array is squared after calling the procedure Square for an object of the Int_8_Array type. We do this with a postcondition using a **for all** expression. This postcondition also uses the 'Old attribute to refer to the original value of the parameter (before the call).

We also want to ensure that the result of calls to the Square function for the Int_8 type are greater than the input to that call. To do that, we write a postcondition using the 'Result attribute of the function and comparing it to the input value.

We can use both pre- and postconditions in the declaration of a single subprogram. For example:

```
Listing 3: show_simple_contract.adb
```

```
with Ada.Text_I0; use Ada.Text_I0;
procedure Show_Simple_Contract is
(continues on next page)
```

```
4
       type Int_8 is range -2 ** 7 .. 2 ** 7 - 1;
5
6
       function Square (A : Int_8) return Int_8 is
         (A * A)
8
           with
9
             Pre => (Integer'Size >= Int_8'Size * 2
10
                       and Integer (A) *
11
                              Integer (A) <=</pre>
12
                            Integer (Int_8'Last)),
13
             Post \Rightarrow (if abs A in 0 | 1
14
                       then Square'Result = abs A
15
                       else Square'Result > A);
16
17
      V : Int_8;
18
19
   begin
      V := Square (11);
20
      Put_Line ("Square of 11 is "
21
                  & Int_8'Image (V));
22
23
       -- Precondition will fail...
24
      V := Square (12);
25
      Put_Line ("Square of 12 is "
26
                  & Int_8'Image (V));
27
   end Show_Simple_Contract;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Contracts.Show_Simple_Contract
MD5: 1d928dd100704907c858562155f90ee2
```

Runtime output

```
Square of 11 is 121

raised ADA.ASSERTIONS.ASSERTION_ERROR : failed precondition from show_simple_
contract.adb:10
```

In this example, we want to ensure that the input value of calls to the Square function for the Int_8 type won't cause overflow in that function. We do this by converting the input value to the **Integer** type, which is used for the temporary calculation, and check if the result is in the appropriate range for the Int_8 type. We have the same postcondition in this example as in the previous one.

16.2 Predicates

Predicates specify expectations regarding types. They're similar to pre- and postconditions, but apply to types instead of subprograms. Their conditions are checked for each object of a given type, which allows verifying that an object of type T is conformant to the requirements of its type.

There are two kinds of predicates: static and dynamic. In simple terms, static predicates are used to check objects at compile-time, while dynamic predicates are used for checks at run time. Normally, static predicates are used for scalar types and dynamic predicates for the more complex types.

Static and dynamic predicates are specified using the following clauses, respectively:

- with Static_Predicate => <property>
- with Dynamic Predicate => <property>

Let's use the following example to illustrate dynamic predicates:

Listing 4: show dynamic predicate courses.adb

```
with Ada.Calendar; use Ada.Calendar;
   with Ada.Containers.Vectors;
   with Ada. Strings. Unbounded;
   use Ada.Strings.Unbounded;
6
   procedure Show_Dynamic_Predicate_Courses is
8
       package Courses is
10
          type Course_Container is private;
11
12
          type Course is record
13
                        : Unbounded_String;
14
             Start_Date : Time;
15
                         : Time;
             End_Date
16
          end record
17
            with Dynamic_Predicate =>
18
              Course.Start_Date <= Course.End_Date;
19
20
          procedure Add (CC : in out Course_Container;
21
                                       Course);
22
       private
23
          package Course_Vectors is new
24
            Ada.Containers.Vectors
25
              (Index_Type => Natural,
26
               Element_Type => Course);
27
28
          type Course_Container is record
29
             V : Course_Vectors.Vector;
30
          end record;
31
       end Courses;
32
33
       package body Courses is
34
          procedure Add (CC : in out Course_Container;
35
                                    Course) is
36
          begin
37
             CC.V.Append (C);
38
          end Add:
39
       end Courses;
40
41
       use Courses;
42
43
       CC : Course_Container;
44
   begin
45
       Add (CC,
46
            Course'(
47
              Name
48
                 To_Unbounded_String
49
                   ("Intro to Photography"),
50
              Start_Date =>
51
                Time_Of (2018, 5, 1),
52
              End Date
53
                Time_Of (2018, 5, 10)));
54
55
           This should trigger an error in the
56
           dynamic predicate check
57
       Add (CC,
58
            Course'(
                                                                           (continues on next page)
```

16.2. Predicates 165

```
Name
60
                To_Unbounded_String
61
                  ("Intro to Video Recording"),
62
              Start_Date =>
63
                Time_Of (2019, 5, 1),
              End_Date
                         =>
65
                Time_Of (2018, 5, 10)));
66
67
   end Show_Dynamic_Predicate_Courses;
68
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Contracts.Show_Dynamic_Predicate_Courses
MD5: 8bd6539e72995fececfcdf9666ffd04f
```

Runtime output

In this example, the package Courses defines a type Course and a type Course_Container, an object of which contains all courses. We want to ensure that the dates of each course are consistent, specifically that the start date is no later than the end date. To enforce this rule, we declare a dynamic predicate for the Course type that performs the check for each object. The predicate uses the type name where a variable of that type would normally be used: this is a reference to the instance of the object being tested.

Note that the example above makes use of unbounded strings and dates. Both types are available in Ada's standard library. Please refer to the following sections for more information about:

- the unbounded string type (Unbounded_String): Unbounded Strings (page 243) section:
- dates and times: Dates & Times (page 227) section.

Static predicates, as mentioned above, are mostly used for scalar types and checked during compilation. They're particularly useful for representing non-contiguous elements of an enumeration. A classic example is a list of week days:

```
type Week is (Mon, Tue, Wed, Thu, Fri, Sat, Sun);
```

We can easily create a sub-list of work days in the week by specifying a **subtype** with a range based on Week. For example:

```
subtype Work_Week is Week range Mon .. Fri;
```

Ranges in Ada can only be specified as contiguous lists: they don't allow us to pick specific days. However, we may want to create a list containing just the first, middle and last day of the work week. To do that, we use a static predicate:

```
subtype Check_Days is Work_Week
with Static_Predicate =>
    Check_Days in Mon | Wed | Fri;
```

Let's look at a complete example:

Listing 5: show_predicates.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Show Predicates is
3
       type Week is (Mon, Tue, Wed, Thu,
                       Fri, Sat, Sun);
       subtype Work_Week is Week range Mon .. Fri;
8
9
       subtype Test_Days is Work_Week
10
         with Static_Predicate =>
11
           Test_Days in Mon | Wed | Fri;
12
13
       type Tests Week is array (Week) of Natural
14
         with Dynamic_Predicate =>
15
            (for all I in Tests_Week'Range =>
16
               (case I is
17
                    when Test_Days =>
18
                        Tests_Week (I) > 0,
19
                    when others
20
                        Tests_Week (I) = 0);
21
22
       Num_Tests : Tests_Week :=
23
                       (Mon \Rightarrow 3, Tue \Rightarrow 0,
24
                        Wed \Rightarrow 4, Thu \Rightarrow 0,
25
                        Fri \Rightarrow 2, Sat \Rightarrow 0,
26
                        Sun \Rightarrow 0;
27
28
       procedure Display_Tests (N : Tests_Week) is
29
       begin
30
          for I in Test_Days loop
31
              Put_Line ("# tests on "
32
                         & Test_Days'Image (I)
33
34
                         & Integer'Image (N (I)));
35
          end loop;
36
37
       end Display_Tests;
   begin
39
       Display_Tests (Num_Tests);
40
41
       -- Assigning non-conformant values to
42
          individual elements of the Tests Week
43
          type does not trigger a predicate
44
           check:
45
       Num_Tests (Tue) := 2;
46
47
           However, assignments with the "complete"
48
49
           Tests_Week type trigger a predicate
50
           check. For example:
51
           Num_Tests := (others => 0);
52
53
           Also, calling any subprogram with
54
           parameters of Tests_Week type
55
           triggers a predicate check. Therefore,
56
           the following line will fail:
57
       Display_Tests (Num_Tests);
58
   end Show_Predicates;
```

Code block metadata

16.2. Predicates 167

```
Project: Courses.Intro_To_Ada.Contracts.Show_Predicates
MD5: 126c47033fc67fc8b6d7f6479205e752
```

Runtime output

Here we have an application that wants to perform tests only on three days of the work week. These days are specified in the Test_Days subtype. We want to track the number of tests that occur each day. We declare the type Tests_Week as an array, an object of which will contain the number of tests done each day. According to our requirements, these tests should happen only in the aforementioned three days; on other days, no tests should be performed. This requirement is implemented with a dynamic predicate of the type Tests_Week. Finally, the actual information about these tests is stored in the array Num_Tests, which is an instance of the Tests_Week type.

The dynamic predicate of the Tests_Week type is verified during the initialization of Num_Tests. If we have a non-conformant value there, the check will fail. However, as we can see in our example, individual assignments to elements of the array do not trigger a check. We can't check for consistency at this point because the initialization of the a complex data structure (such as arrays or records) may not be performed with a single assignment. However, as soon as the object is passed as an argument to a subprogram, the dynamic predicate is checked because the subprogram requires the object to be consistent. This happens in the last call to Display_Tests in our example. Here, the predicate check fails because the previous assignment has a non-conformant value.

16.3 Type invariants

Type invariants are another way of specifying expectations regarding types. While predicates are used for *non-private* types, type invariants are used exclusively to define expectations about private types. If a type T from a package P has a type invariant, the results of operations on objects of type T are always consistent with that invariant.

Type invariants are specified with a with Type_Invariant => roperty> clause. Like predicates, the property defines a condition that allows us to check if an object of type T is conformant to its requirements. In this sense, type invariants can be viewed as a sort of predicate for private types. However, there are some differences in terms of checks. The following table summarizes the differences:

Element	Subprogram parameter checks	Assignment checks
Predi- cates	On all in and out parameters	On assignments and explicit initializations
	On out parameters returned from subprograms declared in the same public scope	On all initializations

We could rewrite our previous example and replace dynamic predicates by type invariants. It would look like this:

Listing 6: show_type_invariant.adb

```
with Ada. Text IO; use Ada. Text IO;
   with Ada.Calendar; use Ada.Calendar;
2
   with Ada.Containers.Vectors;
   with Ada.Strings.Unbounded;
   use Ada.Strings.Unbounded;
   procedure Show_Type_Invariant is
q
10
       package Courses is
11
          type Course is private
12
            with Type_Invariant => Check (Course);
13
14
          type Course_Container is private;
15
16
          procedure Add (CC : in out Course_Container;
17
                                        Course);
18
19
          function Init
20
                                     : String;
            (Name
21
             Start Date, End Date : Time)
22
             return Course;
23
24
          function Check (C : Course)
25
                            return Boolean;
26
27
       private
28
          type Course is record
29
             Name
                        : Unbounded_String;
30
             Start_Date : Time;
31
             End Date
                         : Time;
32
          end record;
33
34
          function Check (C : Course)
35
                            return Boolean is
36
            (C.Start_Date <= C.End_Date);</pre>
37
38
          package Course_Vectors is new
39
            Ada. Containers. Vectors
40
               (Index_Type => Natural,
41
                Element_Type => Course);
42
43
          type Course_Container is record
44
             V : Course_Vectors.Vector;
45
          end record;
46
       end Courses;
47
48
       package body Courses is
49
          procedure Add (CC : in out Course_Container;
50
                                   Course) is
                          C :
51
          begin
52
             CC.V.Append (C);
53
          end Add;
54
55
          function Init
56
            (Name
                                     : String;
57
              Start_Date, End_Date : Time)
58
             return Course is
59
          begin
60
61
              return
                                                                            (continues on next page)
```

```
Course'(Name
62
                          To_Unbounded_String (Name),
63
                        Start_Date => Start_Date,
64
                        End_Date => End_Date);
65
          end Init;
66
      end Courses;
67
68
      use Courses;
69
70
      CC : Course_Container;
71
   begin
72
      Add (CC,
73
            Init (Name
74
                     "Intro to Photography",
                   Start_Date =>
76
                     Time_Of (2018, 5, 1),
77
                   End_Date
78
                     Time_Of (2018, 5, 10)));
79
80
          This should trigger an error in the
81
           type-invariant check
82
      Add (CC,
83
            Init (Name
84
                     "Intro to Video Recording",
85
                   Start_Date =>
86
87
                     Time_Of (2019, 5, 1),
88
                   End_Date
                              =>
                     Time_Of (2018, 5, 10)));
89
   end Show_Type_Invariant;
90
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Contracts.Show_Type_Invariant
MD5: c6ef863da94285f927dd106645af8650
```

Runtime output

```
raised ADA.ASSERTIONS.ASSERTION_ERROR : failed invariant from show_type_invariant. 

adb:13
```

The major difference is that the Course type was a visible (public) type of the Courses package in the previous example, but in this example is a private type.

CHAPTER

SEVENTEEN

INTERFACING WITH C

Ada allows us to interface with code in many languages, including C and C++. This section discusses how to interface with C.

17.1 Multi-language project

By default, when using **gprbuild** we only compile Ada source files. To compile C files as well, we need to modify the project file used by **gprbuild**. We use the Languages entry, as in the following example:

```
project Multilang is

for Languages use ("ada", "c");

for Source_Dirs use ("src");
 for Main use ("main.adb");
 for Object_Dir use "obj";

end Multilang;
```

17.2 Type convention

To interface with data types declared in a C application, you specify the Convention aspect on the corresponding Ada type declaration. In the following example, we interface with the C Enum enumeration declared in a C source file:

Listing 1: show c enum.adb

```
type C_Enum is

type C_Enum is (A, B, C)
with Convention => C;
-- Use C convention for C_Enum
begin
null;
end Show_C_Enum;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Enum
MD5: a14d7d981fd7d6d806cf3c55f35e19c8
```

To interface with C's built-in types, we use the Interfaces. C package, which contains most of the type definitions we need. For example:

Listing 2: show_c_struct.adb

```
with Interfaces.C; use Interfaces.C;
   procedure Show_C_Struct is
3
      type c_struct is record
5
          a : int;
6
          b : long;
7
          c : unsigned;
8
          d : double;
9
      end record
10
        with Convention => C;
11
12
   begin
13
      null;
14
   end Show_C_Struct;
15
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Struct
MD5: dda4d3f8e4ddf5c5138a990a9a8ac427
```

Here, we're interfacing with a C struct (C_Struct) and using the corresponding data types in C (int, long, unsigned and double). This is the declaration in C:

Listing 3: c struct.h

```
struct c_struct
{
    int     a;
    long     b;
    unsigned    c;
    double    d;
};
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Struct
MD5: 58709b6a9eea2606d7ec0aaca0a749ff
```

17.3 Foreign subprograms

17.3.1 Calling C subprograms in Ada

We use a similar approach when interfacing with subprograms written in C. Consider the following declaration in the C header file:

```
Listing 4: my func.h
```

```
int my_func (int a);
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Func
MD5: 37b9d7ba668f7ec83c2b27ee33637937
```

Here's the corresponding C definition:

Listing 5: my_func.c

```
#include "my_func.h"

int my_func (int a)
{
    return a * 2;
}
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Func MD5: 284b1639cb393fc14ed196d78429f3ba
```

We can interface this code in Ada using the Import aspect. For example:

```
Listing 6: show_c_func.adb
```

```
with Interfaces.C; use Interfaces.C;
   with Ada.Text_IO; use Ada.Text_IO;
2
   procedure Show C Func is
4
      function my func (a : int) return int
6
7
          Import
                         => True,
8
          Convention
                         => C;
10
      -- Imports function 'my_func' from C.
11
      -- You can now call it from Ada.
12
13
      V : int;
14
   begin
15
      V := my_func(2);
16
      Put_Line ("Result is " & int'Image (V));
17
   end Show_C_Func;
18
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Func MD5: 6c5d85c1debdeaa642946eacf413dfd2
```

If you want, you can use a different subprogram name in the Ada code. For example, we could call the C function Get_Value:

Listing 7: show_c_func.adb

```
with Interfaces.C; use Interfaces.C;
   with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_C_Func is
5
      function Get_Value (a : int) return int
6
        with
7
          Import
                         => True,
9
          Convention
                        => C,
          External_Name => "my_func";
10
11
      -- Imports function 'my_func' from C and
12
          renames it to 'Get_Value'
13
14
```

(continues on next page)

```
15     V : int;
16     begin
17     V := Get_Value (2);
18     Put_Line ("Result is " & int'Image (V));
19     end Show_C_Func;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Func
MD5: 856b4d99dfaa6946fb4597f254fd2f97
```

17.3.2 Calling Ada subprograms in C

You can also call Ada subprograms from C applications. You do this with the Export aspect. For example:

```
Listing 8: c_api.ads
```

```
with Interfaces.C; use Interfaces.C;
   package C_API is
4
      function My_Func (a : int) return int
5
        with
6
          Export
                         => True,
7
          Convention
                        => C,
8
          External Name => "my func";
10
   end C API;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Ada_Func
MD5: 00aa4ec29fc551e710900e2ee7d96bc9
```

This is the corresponding body that implements that function:

Listing 9: c_api.adb

```
package body C_API is

function My_Func (a : int) return int is
begin
    return a * 2;
end My_Func;

end C_API;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Ada_Func
MD5: 2b999ab431bbc1ee223a654ad84b8248
```

On the C side, we do the same as we would if the function were written in C: simply declare it using the **extern** keyword. For example:

```
Listing 10: main.c
```

```
#include <stdio.h>
(continues on next page)
```

```
extern int my_func (int a);

int main (int argc, char **argv) {
   int v = my_func(2);
   printf("Result is %d\n", v);

return 0;
}
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Ada_Func
MD5: 69301036be9be16ed45895c2a86bc352
```

17.4 Foreign variables

17.4.1 Using C global variables in Ada

To use global variables from C code, we use the same method as subprograms: we specify the Import and Convention aspects for each variable we want to import.

Let's reuse an example from the previous section. We'll add a global variable (func_cnt) to count the number of times the function (my func) is called:

Listing 11: test.h

```
extern int func_cnt;

int my_func (int a);
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Vars
MD5: 11ba8f7a72ce7058571994870a02b052
```

The variable is declared in the C file and incremented in my_func:

Listing 12: test.c

```
#include "test.h"

int func_cnt = 0;

int my_func (int a)
{
  func_cnt++;
  return a * 2;
}
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Vars
MD5: 23631537cb877a03d1243c94cb7b48e8
```

In the Ada application, we just reference the foreign variable:

Listing 13: show_c_func.adb

```
with Interfaces.C; use Interfaces.C;
   with Ada.Text_IO; use Ada.Text_IO;
2
3
   procedure Show_C_Func is
4
5
      function my func (a : int) return int
6
        with
7
           Import
                      => True,
8
           Convention => C;
9
10
      V : int;
11
12
      func cnt : int
13
        with
14
           Import
                         => True.
15
                       => C;
           Convention
16
         We can access the func cnt variable
17
       -- from test.c
18
19
   begin
20
      V := my_func(1);
21
      V := my_func(2);
22
      V := my\_func (3);
23
24
      Put_Line ("Result is "
25
                 & int'Image (V));
26
27
      Put Line ("Function was called "
28
                 & int'Image (func cnt)
29
                 & " times");
30
   end Show_C_Func;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.Ada_C_Vars
MD5: cf64a9dfbc6be853ba19729fe55f0ba4
```

As we see by running the application, the value of the counter is the number of times my func was called.

We can use the External_Name aspect to give a different name for the variable in the Ada application in the same way we do for subprograms.

17.4.2 Using Ada variables in C

You can also use variables declared in Ada files in C applications. In the same way as we did for subprograms, you do this with the Export aspect.

Let's reuse a past example and add a counter, as in the previous example, but this time have the counter incremented in Ada code:

Listing 14: c_api.ads

```
with Interfaces.C; use Interfaces.C;

package C_API is

func_cnt : int := 0
    with

(continues on next page)
```

```
Export
                    => True,
7
           Convention => C;
8
      function My_Func (a : int) return int
10
        with
11
                         => True,
           Export
12
           Convention
                         => C,
13
           External_Name => "my_func";
14
15
   end C_API;
16
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Ada_Vars
MD5: fc118cddd797b669d2c68e57f90f69b2
```

The variable is then incremented in My_Func:

Listing 15: c api.adb

```
package body C_API is

function My_Func (a : int) return int is
begin
func_cnt := func_cnt + 1;
return a * 2;
end My_Func;

end C_API;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Ada_Vars
MD5: adff5f3088da8b0dd853f1fb8b1e204f
```

In the C application, we just need to declare the variable and use it:

Listing 16: main.c

```
#include <stdio.h>
   extern int my_func (int a);
   extern int func_cnt;
   int main (int argc, char **argv) {
     int v;
9
10
     v = my_func(1);
11
     v = my_func(2);
12
     v = my func(3);
13
14
     printf("Result is %d\n", v);
15
16
      printf("Function was called %d times\n",
17
             func_cnt);
18
19
      return 0;
20
   }
21
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Ada_Vars
MD5: 07fb3fbadb8ed4c0543fbfd7b5ef5c57
```

Again, by running the application, we see that the value from the counter is the number of times that my func was called.

17.5 Generating bindings

In the examples above, we manually added aspects to our Ada code to correspond to the C source-code we're interfacing with. This is called creating a *binding*. We can automate this process by using the *Ada spec dump* compiler option: -fdump-ada-spec. We illustrate this by revisiting our previous example.

This was our C header file:

Listing 17: test.h

```
extern int func_cnt;
int my_func (int a);
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds
MD5: 11ba8f7a72ce7058571994870a02b052
```

To create Ada bindings, we'll call the compiler like this:

```
gcc -c -fdump-ada-spec -C ./test.h
```

The result is an Ada spec file called test h.ads:

Listing 18: test h.ads

```
pragma Ada 2005;
   pragma Style Checks (Off);
   with Interfaces.C; use Interfaces.C;
   package test h is
6
      func_cnt : aliased int; -- ./test.h:3
8
      pragma Import (C, func_cnt, "func_cnt");
9
10
      function my_func (arg1 : int) return int; -- ./test.h:5
11
      pragma Import (C, my_func, "my_func");
12
13
   end test h;
14
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds
MD5: 8d18aeae72dba3a9ab4f9f3943fab839
```

Now we simply refer to this test h package in our Ada application:

```
Listing 19: show c func.adb
```

```
with Interfaces.C; use Interfaces.C;
with Ada.Text_I0; use Ada.Text_I0;
(continues on next page)
```

```
with test_h;
                       use test_h;
   procedure Show_C_Func is
      V : int;
   begin
      V := my_func (1);
8
      V := my\_func(2);
9
      V := my_func(3);
10
11
      Put_Line ("Result is "
12
                 & int'Image (V));
13
14
      Put Line ("Function was called "
15
                 & int'Image (func_cnt)
16
                 & " times");
   end Show_C_Func;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds
MD5: 8a07aae87b9f36c3fce84b75e8388933
```

You can specify the name of the parent unit for the bindings you're creating as the operand to fdump-ada-spec:

```
gcc -c -fdump-ada-spec -fada-spec-parent=Ext_C_Code -C ./test.h
```

This creates the file ext c code-test h.ads:

Listing 20: ext_c_code-test_h.ads

```
package Ext_C_Code.test_h is

-- automatic generated bindings...

end Ext_C_Code.test_h;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds_2
MD5: 3bd4087edff145a70d2a6db8543859ad
```

17.5.1 Adapting bindings

The compiler does the best it can when creating bindings for a C header file. However, sometimes it has to guess about the translation and the generated bindings don't always match our expectations. For example, this can happen when creating bindings for functions that have pointers as arguments. In this case, the compiler may use System. Address as the type of one or more pointers. Although this approach works fine (as we'll see later), this is usually not how a human would interpret the C header file. The following example illustrates this issue.

Let's start with this C header file:

Listing 21: test.h

```
struct test;
struct test * test_create(void);

(continues on next page)
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds_3
MD5: af642d9ea995bf01f13f8ff41bb0f4f6
```

And the corresponding C implementation:

Listing 22: test.c

```
#include <stdlib.h>
   #include <string.h>
   #include <stdio.h>
   #include "test.h"
   struct test {
     char name[80];
     char address[120];
9
10
   };
11
   static size_t
12
                         *dst,
   strlcpy_stat(char
13
                  const char *src,
14
                  size t
                              dstsize)
15
16
      size t len = strlen(src);
17
      if (dstsize) {
18
        size_t bl = (len < dstsize-1 ?</pre>
19
                        len : dstsize-1);
20
        ((char*)memcpy(dst, src, bl))[bl] = 0;
21
      }
22
      return len;
23
   }
24
25
   struct test * test create(void)
26
27
      return malloc (sizeof (struct test));
28
   }
29
30
   void test_destroy(struct test *t)
31
32
     if (t != NULL) {
33
        free(t);
34
35
   }
36
37
   void test reset(struct test *t)
38
39
   {
                  = '\0';
     t -> name[0]
```

```
t->address[0] = '\setminus 0';
41
   }
42
43
   void test_set_name(struct test *t,
                         char
                                       *name)
45
46
      strlcpy_stat(t->name,
47
                    name,
48
                     sizeof(t->name));
49
   }
50
51
   void test_set_address(struct test *t,
52
                                          *address)
53
                             char
54
      strlcpy_stat(t->address,
55
                     address,
                     sizeof(t->address));
57
   }
58
59
   void test_display(const struct test *t)
60
61
     printf("Name:
                         %s\n", t->name);
62
      printf("Address: %s\n", t->address);
63
64
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds_3
MD5: 32652eb76ad92212609680d64e5687d3
```

Next, we'll create our bindings:

```
gcc -c -fdump-ada-spec -C ./test.h
```

This creates the following specification in test_h.ads:

Listing 23: test h.ads

```
pragma Ada 2005;
   pragma Style Checks (Off);
   with Interfaces.C; use Interfaces.C;
   with System;
   with Interfaces.C.Strings;
   package test h is
8
9
      -- skipped empty struct test
10
11
      function test create return System.Address; -- ./test.h:5
12
      pragma Import (C, test create, "test create");
13
      procedure test_destroy (arg1 : System.Address); -- ./test.h:7
15
      pragma Import (C, test destroy, "test destroy");
16
17
      procedure test_reset (arg1 : System.Address); -- ./test.h:9
18
      pragma Import (C, test_reset, "test_reset");
19
20
      procedure test_set_name (arg1 : System.Address; arg2 : Interfaces.C.Strings.
21
    ⇔chars ptr); -- ./test.h:11
      pragma Import (C, test set name, "test set name");
22
                                                                       (continues on next page)
```

```
procedure test_set_address (arg1 : System.Address; arg2 : Interfaces.C.Strings.
chars_ptr); -- ./test.h:13
pragma Import (C, test_set_address, "test_set_address");

procedure test_display (arg1 : System.Address); -- ./test.h:15
pragma Import (C, test_display, "test_display");

end test_h;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds_3
MD5: 3bf8f0lb94fd28594e4121a6a36afdf7
```

As we can see, the binding generator completely ignores the declaration **struct test** and all references to the test struct are replaced by addresses (System.Address). Nevertheless, these bindings are good enough to allow us to create a test application in Ada:

Listing 24: show automatic c struct bindings.adb

```
with Ada. Text IO; use Ada. Text IO;
1
   with Interfaces.C;
3
   use Interfaces.C;
   with Interfaces.C.Strings;
   use Interfaces.C.Strings;
8
   with test_h; use test_h;
q
10
   with System;
11
12
   procedure Show Automatic C Struct Bindings is
13
14
      Name
               : constant chars ptr :=
15
         New String ("John Doe");
16
      Address : constant chars_ptr :=
17
        New_String ("Small Town");
18
19
      T : System. Address := test create;
20
21
   begin
22
      test reset (T);
23
      test set name (T, Name);
24
      test_set_address (T, Address);
25
26
      test_display (T);
27
      test_destroy (T);
28
   end Show Automatic C Struct Bindings;
29
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds_3
MD5: 99d64fb14d9c869d140dd2fb7d3888d7
```

We can successfully bind our C code with Ada using the automatically-generated bindings, but they aren't ideal. Instead, we would prefer Ada bindings that match our (human) interpretation of the C header file. This requires manual analysis of the header file. The good news is that we can use the automatic generated bindings as a starting point and adapt them to our needs. For example, we can:

- Define a Test type based on System. Address and use it in all relevant functions.
- 2. Remove the test prefix in all operations on the Test type.

This is the resulting specification:

Listing 25: adapted test h.ads

```
with System;
   with Interfaces.C; use Interfaces.C;
   with Interfaces.C.Strings;
4
   package adapted_test_h is
      type Test is new System.Address;
8
      function Create return Test;
10
11
      pragma Import (C, Create, "test_create");
12
13
      procedure Destroy (T : Test);
      pragma Import (C, Destroy, "test_destroy");
14
15
      procedure Reset (T : Test);
16
      pragma Import (C, Reset, "test_reset");
17
18
      procedure Set Name (T
                               : Test;
19
                            Name : Interfaces C.Strings.chars_ptr); -- ./test.h:11
20
      pragma Import (C, Set_Name, "test_set_name");
21
22
      procedure Set_Address (T
                                       : Test;
23
                               Address : Interfaces.C.Strings.chars_ptr);
24
      pragma Import (C, Set_Address, "test_set_address");
25
26
      procedure Display (T : Test); -- ./test.h:15
27
      pragma Import (C, Display, "test_display");
28
29
   end adapted_test_h;
30
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds_3
MD5: 5cc875e1b01af839141e5e623f6c5b7a
```

And this is the corresponding Ada body:

Listing 26: show_adapted_c_struct_bindings.adb

```
with Interfaces.C;
1
   use Interfaces.C;
2
3
   with Interfaces.C.Strings;
   use Interfaces.C.Strings;
   with adapted_test_h; use adapted_test_h;
   with System;
10
   procedure Show_Adapted_C_Struct_Bindings is
11
12
              : constant chars_ptr :=
      Name
13
        New String ("John Doe");
14
      Address : constant chars_ptr :=
15
```

```
New_String ("Small Town");
16
17
      T : Test := Create;
18
   begin
20
      Reset (T);
21
      Set_Name (T, Name);
22
      Set_Address (T, Address);
23
24
      Display (T);
25
      Destroy (T);
26
   end Show_Adapted_C_Struct_Bindings;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Interfacing_With_C.C_Binds_3
MD5: 626d07b080fbbd2bf1d5f9140b64955c
```

Now we can use the Test type and its operations in a clean, readable way.

OBJECT-ORIENTED PROGRAMMING

Object-oriented programming (OOP) is a large and ill-defined concept in programming languages and one that tends to encompass many different meanings because different languages often implement their own vision of it, with similarities and differences from the implementations in other languages.

However, one model mostly "won" the battle of what object-oriented means, if only by sheer popularity. It's the model used in the Java programming language, which is very similar to the one used by C++. Here are some defining characteristics:

- Type derivation and extension: Most object oriented languages allow the user to add fields to derived types.
- Subtyping: Objects of a type derived from a base type can, in some instances, be substituted for objects of the base type.
- Runtime polymorphism: Calling a subprogram, usually called a *method*, attached to an object type can dispatch at runtime depending on the exact type of the object.
- Encapsulation: Objects can hide some of their data.
- Extensibility: People from the "outside" of your package, or even your whole library, can derive from your object types and define their own behaviors.

Ada dates from before object-oriented programming was as popular as it is today. Some of the mechanisms and concepts from the above list were in the earliest version of Ada even before what we would call OOP was added:

- As we saw, encapsulation is not implemented at the type level in Ada, but instead at the package level.
- Subtyping can be implemented using, well, subtypes, which have a full and permissive static substitutability model. The substitution will fail at runtime if the dynamic constraints of the subtype are not fulfilled.
- Runtime polymorphism can be implemented using variant records.

However, this lists leaves out type extensions, if you don't consider variant records, and extensibility.

The 1995 revision of Ada added a feature filling the gaps, which allowed people to program following the object-oriented paradigm in an easier fashion. This feature is called *tagged types*.

Note

It's possible to program in Ada without ever creating tagged types. If that's your prefered style of programming or you have no specific use for tagged types, feel free to not use them, as is the case for many features of Ada.

However, they can be the best way to express solutions to certain problems and they may be the best way to solve your problem. If that's the case, read on!

18.1 Derived types

Before presenting tagged types, we should discuss a topic we have brushed on, but not really covered, up to now:

You can create one or more new types from every type in Ada. Type derivation is built into the language.

Listing 1: newtypes.ads

```
package Newtypes is
type Point is record
X, Y: Integer;
end record;

type New_Point is new Point;
end Newtypes;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Newtypes MD5: 0d45096755b4bfb08ba8db19ecba3f57
```

Type derivation is useful to enforce strong typing because the type system treats the two types as incompatible.

But the benefits are not limited to that: you can inherit things from the type you derive from. You not only inherit the representation of the data, but you can also inherit behavior.

When you inherit a type you also inherit what are called *primitive operations*. A primitive operation (or just a *primitive*) is a subprogram attached to a type. Ada defines primitives as subprograms defined in the same scope as the type.

A Attention

A subprogram will only become a primitive of the type if:

- 1. The subprogram is declared in the same scope as the type and
- 2. The type and the subprogram are declared in a package

Listing 2: primitives.adb

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Primitives is

package Week is
type Days is (Monday, Tuesday, Wednesday,
Thursday, Friday,
```

```
Saturday, Sunday);
7
8
              Print Day is a primitive
              of the type Days
10
         procedure Print_Day (D : Days);
11
       end Week;
12
13
       package body Week is
14
          procedure Print_Day (D : Days) is
15
16
             Put Line (Days'Image (D));
17
          end Print_Day;
18
       end Week;
19
20
      use Week;
21
       type Weekend_Days is new
22
         Days range Saturday .. Sunday;
23
24
          A procedure Print Day is automatically
25
          inherited here. I\bar{t} is as if the procedure
26
27
          procedure Print_Day (D : Weekend_Days);
28
29
       -- has been declared with the same body
30
31
      Sat : Weekend_Days := Saturday;
32
   begin
33
      Print_Day (Sat);
34
   end Primitives;
35
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Primitives MD5: eblb0eb66f03a4a17bd9686ec4e12e2e
```

Runtime output

```
SATURDAY
```

This kind of inheritance can be very useful, and is not limited to record types (you can use it on discrete types, as in the example above), but it's only superficially similar to object-oriented inheritance:

- Records can't be extended using this mechanism alone. You also can't specify a new representation for the new type: it will always have the same representation as the base type.
- There's no facility for dynamic dispatch or polymorphism. Objects are of a fixed, static type.

There are other differences, but it's not useful to list them all here. Just remember that this is a kind of inheritance you can use if you only want to statically inherit behavior without duplicating code or using composition, but a kind you can't use if you want any dynamic features that are usually associated with OOP.

18.2 Tagged types

The 1995 revision of the Ada language introduced tagged types to fullfil the need for an unified solution that allows programming in an object-oriented style similar to the one described at the beginning of this chapter.

Tagged types are very similar to normal records except that some functionality is added:

- Types have a *tag*, stored inside each object, that identifies the runtime type²⁰ of that object.
- Primitives can dispatch. A primitive on a tagged type is what you would call a method
 in Java or C++. If you derive a base type and override a primitive of it, you can often
 call it on an object with the result that which primitive is called depends on the exact
 runtime type of the object.
- Subtyping rules are introduced allowing a tagged type derived from a base type to be statically compatible with the base type.

Let's see our first tagged type declarations:

Listing 3: p.ads

```
package P is
1
      type My Class is tagged null record;
2
      -- Just like a regular record, but
3
      -- with tagged qualifier
4
5
      -- Methods are outside of the type
6
      -- definition:
7
8
      procedure Foo (Self : in out My Class);
9
      -- If you define a procedure taking a
10
      -- My_Class argument in the same package,
11
      -- it will be a method.
12
13
      -- Here's how you derive a tagged type:
14
15
      type Derived is new My_Class with record
16
         A : Integer;
17
         -- You can add fields in derived types.
18
      end record;
19
20
      overriding
21
      procedure Foo (Self : in out Derived);
22
      -- The "overriding" qualifier is optional,
23
      -- but if it is present, it must be valid.
24
   end P;
25
```

Listing 4: p.adb

```
with Ada.Text_IO; use Ada.Text_IO;
2
   package body P is
3
      procedure Foo (Self : in out My_Class) is
4
      begin
5
         Put_Line ("In My_Class.Foo");
6
      end Foo;
7
8
      procedure Foo (Self : in out Derived) is
9
      begin
10
         Put_Line ("In Derived.Foo, A = "
11
                    & Integer'Image (Self.A));
12
      end Foo:
13
   end P;
14
```

Code block metadata

²⁰ https://en.wikipedia.org/wiki/Run-time_type_information

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Types MD5: 45baaad66a1047358addb574d0fa00bc
```

18.3 Classwide types

To remain consistent with the rest of the language, a new notation needed to be introduced to say "This object is of this type or any descendant derives tagged type".

In Ada, we call this the *classwide type*. It's used in OOP as soon as you need polymorphism. For example, you can't do the following:

Listing 5: main.adb

```
with P; use P;
2
   procedure Main is
3
5
      01 : My Class;
      -- Declaring an object of type My_Class
6
      02 : Derived := (A => 12);
8
      -- Declaring an object of type Derived
9
10
      03 : My Class := 02;
11
      -- INVALID: Trying to assign a value
12
      -- of type derived to a variable of
13
      -- type My Class.
14
  begin
15
      null;
16
  end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Types MD5: c87ad8bb686cb1763740750846258357
```

Build output

```
main.adb:11:21: error: expected type "My_Class" defined at p.ads:2
main.adb:11:21: error: found type "Derived" defined at p.ads:16
gprbuild: *** compilation phase failed
```

This is because an object of a type T is exactly of the type T, whether T is tagged or not. What you want to say as a programmer is "I want O3 to be able to hold an object of type My Class or any type descending from My Class". Here's how you do that:

Listing 6: main.adb

```
with P; use P;
   procedure Main is
      01 : My Class;
4
      -- Declare an object of type My Class
5
6
      02 : Derived := (A => 12);
7
      -- Declare an object of type Derived
8
      03 : My_Class'Class := 02;
10
      -- Now valid: My_Class'Class designates
11
      -- the classwide type for My_Class,
```

```
-- which is the set of all types
-- descending from My_Class (including
-- My_Class).

begin
null;
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Types
MD5: 35412176a248015a26e507164ce526af
```

A Attention

Because an object of a classwide type can be the size of any descendant of its base type, it has an unknown size. It's therefore an indefinite type, with the expected restrictions:

- It can't be stored as a field/component of a record
- An object of a classwide type needs to be initialized immediately (you can't specify the constraints of such a type in any way other than by initializing it).

18.4 Dispatching operations

We saw that you can override operations in types derived from another tagged type. The eventual goal of OOP is to make a dispatching call: a call to a primitive (method) that depends on the exact type of the object.

But, if you think carefully about it, a variable of type My_Class always contains an object of exactly that type. If you want to have a variable that can contain a My_Class or any derived type, it has to be of type My_Class'Class.

In other words, to make a dispatching call, you must first have an object that can be either of a type or any type derived from this type, namely an object of a classwide type.

Listing 7: main.adb

```
with P; use P;
   procedure Main is
3
      01 : My Class;
      -- Declare an object of type My_Class
5
6
      02 : Derived := (A => 12):
7
      -- Declare an object of type Derived
8
      03 : My Class'Class := 02;
10
11
      04 : My Class'Class := 01;
12
   begin
13
      Foo (01);
      -- Non dispatching: Calls My_Class.Foo
15
      Foo (02);
16
      -- Non dispatching: Calls Derived.Foo
17
      Foo (03);
18
       -- Dispatching: Calls Derived.Foo
19
      Foo (04);
20
```

```
-- Dispatching: Calls My_Class.Foo
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Types MD5: 7631f823b0dd9e5474f6bb2dc35af2a2
```

Runtime output

```
In My_Class.Foo
In Derived.Foo, A = 12
In Derived.Foo, A = 12
In My_Class.Foo
```

4 Attention

You can convert an object of type Derived to an object of type My_Class. This is called a *view conversion* in Ada parlance and is useful, for example, if you want to call a parent method.

In that case, the object really is converted to a My_Class object, which means its tag is changed. Since tagged objects are always passed by reference, you can use this kind of conversion to modify the state of an object: changes to converted object will affect the original one.

Listing 8: main.adb

```
with P; use P;
   procedure Main is
3
      01 : Derived := (A => 12);
      -- Declare an object of type Derived
6
      02 : My_Class := My_Class (01);
7
8
      03 : My_Class'Class := 02;
q
   begin
10
      Foo (01);
11
      -- Non dispatching: Calls Derived.Foo
12
      Foo (02);
13
      -- Non dispatching: Calls My_Class.Foo
15
      Foo (03);
      -- Dispatching: Calls My_Class.Foo
17
   end Main:
```

Code block metadata

Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Types MD5: b92112b05201ff14789baca258fa0cbc

Runtime output

```
In Derived.Foo, A = 12
In My_Class.Foo
In My_Class.Foo
```

18.5 Dot notation

You can also call primitives of tagged types with a notation that's more familiar to object oriented programmers. Given the Foo primitive above, you can also write the above program this way:

Listing 9: main.adb

```
with P; use P;
   procedure Main is
      01 : My_Class;
4
      -- Declare an object of type My Class
5
6
      02 : Derived := (A => 12);
7
      -- Declare an object of type Derived
8
9
      03 : My Class'Class := 02;
10
11
      04 : My_Class'Class := 01;
12
   begin
13
      01.Foo;
14
      -- Non dispatching: Calls My_Class.Foo
15
      02.Foo;
16
      -- Non dispatching: Calls Derived.Foo
17
      03.Foo;
18
       -- Dispatching: Calls Derived.Foo
19
      04.Foo;
20
      -- Dispatching: Calls My Class.Foo
21
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Types MD5: 9c6ebdfec9ceeb986d92eb90ec9ff59b
```

Runtime output

```
In My_Class.Foo
In Derived.Foo, A = 12
In Derived.Foo, A = 12
In My_Class.Foo
```

If the dispatching parameter of a primitive is the first parameter, which is the case in our examples, you can call the primitive using the dot notation. Any remaining parameter are passed normally:

Listing 10: main.adb

```
with P; use P;
2
   procedure Main is
3
      package Extend is
4
         type D2 is new Derived with null record;
5
6
         procedure Bar (Self : in out D2;
                         Val :
                                       Integer);
      end Extend;
9
10
      package body Extend is
11
         procedure Bar (Self : in out D2;
12
```

```
Val:
                                           Integer) is
13
          begin
14
              Self.A := Self.A + Val;
15
          end Bar;
       end Extend;
17
18
       use Extend;
19
20
       0bj : D2 := (A => 15);
21
   begin
22
       Obj.Bar (2);
23
       Obj.Foo;
24
   end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Types MD5: fec4f5cc4213cc111708dcc276e870c2
```

Runtime output

```
In Derived.Foo, A = 17
```

18.6 Private & Limited

We've seen previously (in the *Privacy* (page 109) chapter) that types can be declared limited or private. These encapsulation techniques can also be applied to tagged types, as we'll see in this section.

This is an example of a tagged private type:

Listing 11: p.ads

```
package P is
type T is tagged private;
private
type T is tagged record
E : Integer;
end record;
end P;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Private_Types
MD5: 4cd4bcdla54d5f6407a500558b5da417
```

This is an example of a tagged limited type:

Listing 12: p.ads

```
package P is
type T is tagged limited record
E : Integer;
end record;
end P;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Tagged_Limited_Types MD5: 13228777133aa6db97da1c29f732459c
```

Naturally, you can combine both *limited* and *private* types and declare a tagged limited private type:

Listing 13: p.ads

```
package P is
type T is tagged limited private;

procedure Init (A : in out T);
private
type T is tagged limited record
E : Integer;
end record;
end P;
```

Listing 14: p.adb

```
package body P is

procedure Init (A : in out T) is
begin
    A.E := 0;
end Init;

end P;
```

Listing 15: main.adb

```
with P; use P;
1
   procedure Main is
3
      T1, T2 : T;
   begin
      T1.Init;
6
      T2.Init;
7
8
      -- The following line doesn't work
9
      -- because type T is private:
10
11
      -- T1.E := 0;
12
13
      -- The following line doesn't work
14
      -- because type T is limited:
15
16
      -- T2 := T1;
17
  end Main;
```

Code block metadata

Note that the code in the Main procedure above presents two assignments that trigger compilation errors because type T is limited private. In fact, you cannot:

- assign to T1.E directly because type T is private;
- assign T1 to T2 because type T is limited.

In this case, there's no distinction between tagged and non-tagged types: these compilation errors would also occur for non-tagged types.

18.7 Classwide access types

In this section, we'll discuss an useful pattern for object-oriented programming in Ada: classwide access type. Let's start with an example where we declare a tagged type T and a derived type T New:

Listing 16: p.ads

```
package P is
type T is tagged null record;

procedure Show (Dummy : T);

type T_New is new T with null record;

procedure Show (Dummy : T_New);
end P;
```

Listing 17: p.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   package body P is
3
      procedure Show (Dummy : T) is
5
      begin
6
          Put_Line ("Using type "
                    & T'External_Tag);
8
      end Show:
9
10
      procedure Show (Dummy : T New) is
11
      begin
12
          Put Line ("Using type "
13
                     & T_New'External_Tag);
14
      end Show;
15
16
   end P;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Classwide_Error MD5: fd5cb99925d3c88536546aa0be8104b7
```

Note that we're using null records for both types T and T_New. Although these types don't actually have any component, we can still use them to demonstrate dispatching. Also note that the example above makes use of the 'External_Tag attribute in the implementation of the Show procedure to get a string for the corresponding tagged type.

As we've seen before, we must use a classwide type to create objects that can make dispatching calls. In other words, objects of type T'Class will dispatch. For example:

Listing 18: dispatching example.adb

```
with P; use P;

procedure Dispatching_Example is

T2 : T_New;

T_Dispatch : constant T'Class := T2;

begin

T_Dispatch.Show;
end Dispatching_Example;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Classwide_Error
MD5: f8957b31c9c62db23759baad7b867a57
```

Runtime output

```
Using type P.T_NEW
```

A more useful application is to declare an array of objects that can dispatch. For example, we'd like to declare an array T_Arr, loop over this array and dispatch according to the actual type of each individual element:

```
for I in T_Arr'Range loop
   T_Arr (I).Show;
   -- Call Show procedure according
   -- to actual type of T_Arr (I)
end loop;
```

However, it's not possible to declare an array of type T'Class directly:

Listing 19: classwide compilation error.adb

```
with P; use P;
   procedure Classwide_Compilation_Error is
      T_Arr : array (1 .. 2) of T'Class;
4
5
                        Compilation Error!
6
   begin
7
      for I in T Arr'Range loop
8
         T_Arr (I).Show;
9
      end loop;
10
  end Classwide_Compilation_Error;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Classwide_Error MD5: e86f6c6ee35dced8f330bf6177d178fd
```

Build output

```
classwide_compilation_error.adb:4:32: error: unconstrained element type in array_declaration gprbuild: *** compilation phase failed
```

In fact, it's impossible for the compiler to know which type would actually be used for each element of the array. However, if we use dynamic allocation via access types, we can allocate objects of different types for the individual elements of an array T_Arr. We do this by using classwide access types, which have the following format:

```
type T_Class is access T'Class;
```

We can rewrite the previous example using the T_Class type. In this case, dynamically allocated objects of this type will dispatch according to the actual type used during the allocation. Also, let's introduce an Init procedure that won't be overridden for the derived T_New type. This is the adapted code:

Listing 20: p.ads

```
package P is
type T is tagged record

(continues on next page)
```

```
E : Integer;
3
      end record;
4
      type T_Class is access T'Class;
      procedure Init (A : in out T);
8
9
      procedure Show (Dummy : T);
10
11
      type T_New is new T with null record;
12
13
      procedure Show (Dummy : T_New);
14
15
   end P;
```

Listing 21: p.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   package body P is
4
      procedure Init (A : in out T) is
5
      begin
6
          Put_Line ("Initializing type T...");
          A.E := 0;
8
      end Init;
9
10
      procedure Show (Dummy : T) is
11
12
      begin
          Put_Line ("Using type "
13
                     & T'External_Tag);
14
      end Show;
15
16
      procedure Show (Dummy : T_New) is
17
      begin
18
          Put_Line ("Using type "
19
                     & T_New'External_Tag);
20
      end Show;
21
22
   end P;
```

Listing 22: main.adb

```
1
   with Ada.Text_IO; use Ada.Text_IO;
   with P;
                      use P;
2
4
   procedure Main is
      T_Arr : array (1 .. 2) of T_Class;
5
   begin
6
      T_Arr(1) := new T;
7
      T_Arr (2) := new T_New;
8
9
      for I in T Arr'Range loop
10
         Put_Line ("Element #
11
12
                    & Integer'Image (I));
13
         T_Arr (I).Init;
         T_Arr (I).Show;
15
16
         Put_Line ("----");
17
```

```
end loop;
end Main;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Object_Oriented_Programming.Classwide_Access MD5: 97c05a8f911d0a0e39c0cc90fae184a7
```

Runtime output

```
Element # 1
Initializing type T...
Using type P.T
------
Element # 2
Initializing type T...
Using type P.T_NEW
```

In this example, the first element (T_Arr (1)) is of type T, while the second element is of type T_New. When running the example, the Init procedure of type T is called for both elements of the T_Arr array, while the call to the Show procedure selects the corresponding procedure according to the type of each element of T_Arr.

STANDARD LIBRARY: CONTAINERS

In previous chapters, we've used arrays as the standard way to group multiple objects of a specific data type. In many cases, arrays are good enough for manipulating those objects. However, there are situations that require more flexibility and more advanced operations. For those cases, Ada provides support for containers — such as vectors and sets — in its standard library.

We present an introduction to containers here. For a list of all containers available in Ada, see *Appendix B* (page 267).

19.1 Vectors

In the following sections, we present a general overview of vectors, including instantiation, initialization, and operations on vector elements and vectors.

19.1.1 Instantiation

Here's an example showing the instantiation and declaration of a vector V:

Listing 1: show vector inst.adb

```
with Ada.Containers.Vectors;
   procedure Show_Vector_Inst is
      package Integer Vectors is new
        Ada. Containers. Vectors
6
           (Index_Type => Natural,
7
           Element Type => Integer);
8
q
      V : Integer Vectors. Vector;
10
   begin
11
      null;
12
   end Show Vector Inst;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Inst
MD5: 8b737842d2784f25502990f21e1cf6de
```

Containers are based on generic packages, so we can't simply declare a vector as we would declare an array of a specific type:

```
A : array (1 .. 10) of Integer;
```

Instead, we first need to instantiate one of those packages. We with the container package (Ada.Containers.Vectors in this case) and instantiate it to create an instance of the

generic package for the desired type. Only then can we declare the vector using the type from the instantiated package. This instantiation needs to be done for any container type from the standard library.

In the instantiation of Integer_Vectors, we indicate that the vector contains elements of **Integer** type by specifying it as the Element_Type. By setting Index_Type to **Natural**, we specify that the allowed range includes all natural numbers. We could have used a more restrictive range if desired.

19.1.2 Initialization

One way to initialize a vector is from a concatenation of elements. We use the & operator, as shown in the following example:

Listing 2: show_vector_init.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada. Containers. Vectors:
   with Ada. Text IO; use Ada. Text IO;
   procedure Show_Vector_Init is
      package Integer Vectors is new
        Ada. Containers. Vectors
9
           (Index_Type => Natural,
10
           Element_Type => Integer);
11
12
      use Integer Vectors;
13
14
      V : Vector := 20 & 10 & 0 & 13;
15
   begin
16
      Put_Line ("Vector has "
17
                 & Count Type'Image (V.Length)
18
                 & " elements");
19
   end Show_Vector_Init;
20
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Init
MD5: 0087b0a15e0c88b27ac36c3b27159a17
```

Runtime output

```
Vector has 4 elements
```

We specify **use** Integer_Vectors, so we have direct access to the types and operations from the instantiated package. Also, the example introduces another operation on the vector: Length, which retrieves the number of elements in the vector. We can use the dot notation because Vector is a tagged type, allowing us to write either V.Length or Length (V).

19.1.3 Appending and prepending elements

You add elements to a vector using the Prepend and Append operations. As the names suggest, these operations add elements to the beginning or end of a vector, respectively. For example:

Listing 3: show_vector_append.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada.Containers.Vectors;
2
3
   with Ada.Text_IO; use Ada.Text_IO;
4
   procedure Show Vector Append is
6
      package Integer Vectors is new
8
         Ada. Containers. Vectors
9
           (Index_Type => Natural,
10
            Element_Type => Integer);
11
12
      use Integer_Vectors;
13
14
      V : Vector;
15
   begin
16
      Put_Line ("Appending some elements "
17
                 & "to the vector...");
18
      V.Append (20);
19
      V.Append (10);
20
      V.Append (0);
21
      V.Append (13);
22
      Put_Line ("Finished appending.");
23
24
      Put_Line ("Prepending some elements"
25
                 & "to the vector...");
26
      V.Prepend (30);
27
      V.Prepend (40);
28
      V.Prepend (100);
29
      Put Line ("Finished prepending.");
30
31
      Put Line ("Vector has "
32
                 & Count_Type'Image (V.Length)
33
                 & " elements");
   end Show_Vector_Append;
35
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Append MD5: f88d393ba96a7950f58d9f1c0c74a021
```

Runtime output

```
Appending some elements to the vector...
Finished appending.
Prepending some elementsto the vector...
Finished prepending.
Vector has 7 elements
```

This example puts elements into the vector in the following sequence: (100, 40, 30, 20, 10, 0, 13).

The Reference Manual specifies that the worst-case complexity must be:

- O(log N) for the Append operation, and
- O(N log N) for the Prepend operation.

19.1. Vectors 201

19.1.4 Accessing first and last elements

We access the first and last elements of a vector using the First_Element and Last_Element functions. For example:

Listing 4: show_vector_first_last_element.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada.Containers.Vectors;
2
   with Ada. Text IO; use Ada. Text IO;
   procedure Show Vector First Last Element is
6
      package Integer_Vectors is new
8
        Ada. Containers. Vectors
9
           (Index_Type => Natural,
10
            Element_Type => Integer);
11
12
      use Integer Vectors;
13
14
      function Img (I : Integer)
                                       return String
15
         renames Integer'Image;
16
       function Img (I : Count_Type) return String
17
        renames Count_Type'Image;
18
19
      V : Vector := 20 & 10 & 0 & 13;
20
   begin
21
      Put_Line ("Vector has "
22
                 & Img (V.Length)
23
                 & " elements");
24
25
       -- Using V.First Element to
26
          retrieve first element
27
      Put_Line ("First element is "
28
                 & Img (V.First_Element));
29
30
       -- Using V.Last_Element to
31
          retrieve last element
32
      Put_Line ("Last element is "
33
                 & Img (V.Last_Element));
34
   end Show_Vector_First_Last_Element;
35
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_First_Last_Element MD5: 602255760d0017ced6b4115c845cd48d
```

Runtime output

```
Vector has 4 elements
First element is 20
Last element is 13
```

You can swap elements by calling the procedure Swap and retrieving a reference (a *cursor*) to the first and last elements of the vector by calling First and Last. A cursor allows us to iterate over a container and process individual elements from it.

With these operations, we're able to write code to swap the first and last elements of a vector:

Listing 5: show_vector_first_last_element.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada.Containers.Vectors;
2
   with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_Vector_First_Last_Element is
      package Integer_Vectors is new
8
        Ada.Containers.Vectors
9
           (Index_Type => Natural,
10
           Element_Type => Integer);
11
12
      use Integer_Vectors;
13
14
      function Img (I : Integer) return String
15
         renames Integer'Image;
16
17
      V : Vector := 20 & 10 & 0 & 13;
18
   beain
19
      -- We use V.First and V.Last to retrieve
20
          cursor for first and last elements.
21
      -- We use V.Swap to swap elements.
22
      V.Swap (V.First, V.Last);
23
24
      Put_Line ("First element is now "
25
                 & Img (V.First Element));
26
      Put_Line ("Last element is now "
27
                 & Img (V.Last_Element));
28
   end Show_Vector_First_Last_Element;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_First_Last_Element MD5: 1a0c0bf28bb661b3f328473ac3c2eb54
```

Runtime output

```
First element is now 13
Last element is now 20
```

19.1.5 Iterating

The easiest way to iterate over a container is to use a **for** E **of** Our_Container loop. This gives us a reference (E) to the element at the current position. We can then use E directly. For example:

Listing 6: show_vector_iteration.adb

```
with Ada.Containers.Vectors;
1
2
   with Ada. Text IO; use Ada. Text IO;
3
   procedure Show Vector Iteration is
5
6
      package Integer Vectors is new
7
         Ada. Containers. Vectors
8
           (Index Type => Natural,
9
            Element Type => Integer);
10
                                                                          (continues on next page)
```

19.1. Vectors 203

```
11
      use Integer_Vectors;
12
13
       function Img (I : Integer) return String
         renames Integer'Image;
15
16
      V : Vector := 20 & 10 & 0 & 13;
17
   begin
18
      Put_Line ("Vector elements are: ");
19
20
21
       -- Using for ... of loop to iterate:
22
23
       for E of V loop
         Put_Line ("- " & Img (E));
25
       end loop;
26
27
   end Show_Vector_Iteration;
28
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Iteration MD5: 4fc9a939aa822097d3a937646d3e2910
```

Runtime output

```
Vector elements are:
- 20
- 10
- 0
- 13
```

This code displays each element from the vector V.

Because we're given a reference, we can display not only the value of an element but also modify it. For example, we could easily write a loop to add one to each element of vector V:

```
for E of V loop
  E := E + 1;
end loop;
```

We can also use indices to access vector elements. The format is similar to a loop over array elements: we use a **for** I **in** <**range**> loop. The range is provided by V.First_Index and V.Last_Index. We can access the current element by using it as an array index: V (I). For example:

Listing 7: show_vector_index_iteration.adb

```
with Ada.Containers.Vectors;

with Ada.Text_IO; use Ada.Text_IO;

procedure Show_Vector_Index_Iteration is

package Integer_Vectors is new
Ada.Containers.Vectors
(Index_Type => Natural,
Element_Type => Integer);

use Integer_Vectors;
```

```
13
      V : Vector := 20 & 10 & 0 & 13;
14
   begin
15
      Put_Line ("Vector elements are: ");
16
17
18
       -- Using indices in a "for I in ..." loop
19
          to iterate:
20
21
       for I in V.First_Index .. V.Last_Index loop
22
          -- Displaying current index I
23
          Put ("- ["
24
               & Extended_Index'Image (I)
25
               & "] ");
26
          Put (Integer'Image (V (I)));
28
29
          -- We could also use the V.Element (I)
30
             function to retrieve the element at
31
             the current index I
32
33
          New_Line;
34
      end loop;
35
36
   end Show_Vector_Index_Iteration;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Index_Iteration MD5: f5600bbcc53d6d6887a771b1505676e9
```

Runtime output

```
Vector elements are:
- [ 0] 20
- [ 1] 10
- [ 2] 0
- [ 3] 13
```

Here, in addition to displaying the vector elements, we're also displaying each index, I, just like what we can do for array indices. Also, we can access the element by using either the short form V (I) or the longer form V.Element (I) but not V.I.

As mentioned in the previous section, you can use cursors to iterate over containers. For this, use the function Iterate, which retrieves a cursor for each position in the vector. The corresponding loop has the format **for** C **in** V.Iterate **loop**. Like the previous example using indices, you can again access the current element by using the cursor as an array index: V (C). For example:

Listing 8: show_vector_cursor_iteration.adb

19.1. Vectors 205

```
11
      use Integer_Vectors;
12
13
      V : Vector := 20 & 10 & 0 & 13;
15
   begin
      Put_Line ("Vector elements are: ");
16
17
18
          Use a cursor to iterate in a loop:
19
20
       for C in V.Iterate loop
21
             Using To_Index function to retrieve
22
23
              the index for the cursor position
          Put ("- ["
               & Extended_Index'Image (To_Index (C))
               & "] ");
26
27
          Put (Integer'Image (V (C)));
28
29
          -- We could use Element (C) to retrieve
30
          -- the vector element for the cursor
31
          -- position
32
33
          New_Line;
34
      end loop;
35
36
37
          Alternatively, we could iterate with a
          while-loop:
38
39
           declare
40
              C : Cursor := V.First;
41
           begin
42
              while C /= No_Element loop
43
                some processing here...
44
45
                 C := Next(C);
46
              end loop;
47
       -- end:
48
49
   end Show_Vector_Cursor_Iteration;
50
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Cursor_Iteration MD5: de789bbd2e1814aae3fb5213c99ac25c
```

Runtime output

```
Vector elements are:
- [ 0] 20
- [ 1] 10
- [ 2] 0
- [ 3] 13
```

Instead of accessing an element in the loop using V (C), we could also have used the longer form Element (C). In this example, we're using the function To_Index to retrieve the index corresponding to the current cursor.

As shown in the comments after the loop, we could also use a **while** ... **loop** to iterate over the vector. In this case, we would start with a cursor for the first element (retrieved by calling V.First) and then call Next (C) to retrieve a cursor for subsequent elements. Next (C) returns No Element when the cursor reaches the end of the vector.

You can directly modify the elements using a reference. This is what it looks like when using both indices and cursors:

```
-- Modify vector elements using index
for I in V.First_Index .. V.Last_Index loop
   V (I) := V (I) + 1;
end loop;

-- Modify vector elements using cursor
for C in V.Iterate loop
   V (C) := V (C) + 1;
end loop;
```

The Reference Manual requires that the worst-case complexity for accessing an element be O(log N).

Another way of modifying elements of a vector is using a *process procedure*, which takes an individual element and does some processing on it. You can call Update_Element and pass both a cursor and an access to the process procedure. For example:

Listing 9: show_vector_update.adb

```
with Ada.Containers.Vectors;
   with Ada.Text_IO; use Ada.Text_IO;
3
   procedure Show Vector Update is
5
      package Integer Vectors is new
7
         Ada.Containers.Vectors
8
           (Index_Type => Natural,
            Element_Type => Integer);
10
11
      use Integer_Vectors;
12
13
      procedure Add One (I : in out Integer) is
14
       begin
15
         I := I + 1;
16
      end Add One;
17
18
      V : Vector := 20 & 10 & 12;
19
20
   begin
21
          Use V.Update_Element to process elements
22
23
       for C in V.Iterate loop
24
          V.Update Element (C, Add One'Access);
25
      end loop;
26
27
   end Show Vector Update;
28
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Update
MD5: 5dcc3dd8020632a8ea2ce975ecd8f4da
```

19.1.6 Finding and changing elements

You can locate a specific element in a vector by retrieving its index. Find_Index retrieves the index of the first element matching the value you're looking for. Alternatively, you can use Find to retrieve a cursor referencing that element. For example:

19.1. Vectors 207

Listing 10: show_find_vector_element.adb

```
with Ada.Containers.Vectors;
   with Ada. Text IO; use Ada. Text IO;
3
   procedure Show_Find_Vector_Element is
5
      package Integer Vectors is new
7
        Ada. Containers. Vectors
8
           (Index Type => Natural,
9
           Element_Type => Integer);
10
11
      use Integer_Vectors;
12
13
      V : Vector := 20 & 10 & 0 & 13;
14
      Idx : Extended Index;
15
      C
          : Cursor;
16
   beain
17
      -- Using Find Index to retrieve the index
18
      -- of element with value 10
19
      Idx := V.Find Index (10);
20
      Put_Line ("Index of element with value 10 is "
21
                 & Extended_Index'Image (Idx));
22
23
      -- Using Find to retrieve the cursor for
24
      -- the element with value 13
25
          := V.Find (13);
      C
26
      Idx := To Index (C);
27
      Put_Line ("Index of element with value 13 is "
28
                 & Extended Index'Image (Idx));
29
   end Show Find Vector Element;
30
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Find_Vector_Element MD5: c3da01cd66c8705a7cbccae8390d5f81
```

Runtime output

```
Index of element with value 10 is 1
Index of element with value 13 is 3
```

As we saw in the previous section, we can directly access vector elements by using either an index or cursor. However, an exception is raised if we try to access an element with an invalid index or cursor, so we must check whether the index or cursor is valid before using it to access an element. In our example, Find_Index or Find might not have found the element in the vector. We check for this possibility by comparing the index to No_Index or the cursor to No_Element. For example:

```
-- Modify vector element using index
if Idx /= No_Index then
   V (Idx) := 11;
end if;
-- Modify vector element using cursor
if C /= No_Element then
   V (C) := 14;
end if;
```

Instead of writing V(C) := 14, we could use the longer form $V.Replace_Element(C, 14)$.

19.1.7 Inserting elements

In the previous sections, we've seen examples of how to add elements to a vector:

- using the concatenation operator (&) at the vector declaration, or
- calling the Prepend and Append procedures.

You may want to insert an element at a specific position, e.g. before a certain element in the vector. You do this by calling Insert. For example:

Listing 11: show_vector_insert.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada.Containers.Vectors;
   with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_Vector_Insert is
6
      package Integer Vectors is new
8
         Ada. Containers. Vectors
           (Index_Type => Natural,
10
            Element_Type => Integer);
11
12
      use Integer_Vectors;
13
14
      procedure Show Elements (V : Vector) is
15
       begin
16
          New Line;
17
          Put Line ("Vector has "
18
                     & Count_Type'Image (V.Length)
19
                     & " elements");
20
21
          if not V.Is_Empty then
22
             Put_Line ("Vector elements are: ");
23
             for E of V loop
24
                Put Line ("- " & Integer'Image (E));
25
             end loop;
26
          end if;
27
      end Show Elements;
28
29
      V : Vector := 20 & 10 & 12;
30
      C : Cursor;
31
32
   begin
      Show_Elements (V);
33
34
      New_Line;
35
      Put_Line ("Adding element with value 9");
36
      Put_Line (" (before 10)...");
37
38
39
       -- Using V.Insert to insert the element
40
          into the vector
41
42
      C := V.Find (10);
43
       if C /= No_Element then
44
          V.Insert (C, 9);
45
      end if:
46
47
      Show_Elements (V);
48
49
   end Show Vector Insert;
50
```

19.1. Vectors 209

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Insert
MD5: af49f390388896c51ab97541036fbcaf
```

Runtime output

In this example, we're looking for an element with the value of 10. If we find it, we insert an element with the value of 9 before it.

19.1.8 Removing elements

You can remove elements from a vector by passing either a valid index or cursor to the Delete procedure. If we combine this with the functions Find_Index and Find from the previous section, we can write a program that searches for a specific element and deletes it, if found:

Listing 12: show remove vector element.adb

```
with Ada.Containers.Vectors;
   with Ada.Text_IO; use Ada.Text_IO;
3
4
   procedure Show_Remove_Vector_Element is
5
      package Integer_Vectors is new
6
        Ada. Containers. Vectors
7
8
           (Index_Type => Natural,
9
            Element_Type => Integer);
10
      use Integer_Vectors;
11
12
      V : Vector := 20 & 10 & 0 & 13 & 10 & 13;
13
      Idx : Extended_Index;
14
      C
          : Cursor;
15
   begin
16
      -- Use Find Index to retrieve index of
17
      -- the element with value 10
18
      Idx := V.Find_Index (10);
19
20
21
      -- Checking whether index is valid
      if Idx /= No_Index then
22
         -- Removing element using V.Delete
         V.Delete (Idx);
24
      end if;
25
```

```
26
       -- Use Find to retrieve cursor for
27
      -- the element with value 13
28
      C := V.Find (13);
30
       -- Check whether index is valid
31
      if C /= No_Element then
32
          -- Remove element using V.Delete
33
         V.Delete (C);
34
      end if;
35
36
   end Show_Remove_Vector_Element;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Remove_Vector_Element MD5: 540d0dc5715e58926e9dc4600bd6ad5d
```

We can extend this approach to delete all elements matching a certain value. We just need to keep searching for the element in a loop until we get an invalid index or cursor. For example:

Listing 13: show_remove_vector_elements.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada.Containers.Vectors;
   with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_Remove_Vector_Elements is
      package Integer Vectors is new
8
         Ada. Containers. Vectors
           (Index Type => Natural,
10
            Element_Type => Integer);
11
12
13
      use Integer_Vectors;
14
      procedure Show_Elements (V : Vector) is
15
      begin
16
          New Line;
17
          Put_Line ("Vector has "
18
                     & Count_Type'Image (V.Length)
19
                     & " elements");
20
21
          if not V.Is_Empty then
22
             Put Line ("Vector elements are: ");
23
             for E of V loop
24
                Put_Line ("- " & Integer'Image (E));
25
             end loop;
26
          end if:
27
      end Show Elements;
28
29
      V : Vector := 20 & 10 & 0 & 13 & 10 & 14 & 13;
30
31
      Show_Elements (V);
32
33
          Remove elements using an index
35
36
      declare
37
                                                                          (continues on next page)
```

19.1. Vectors 211

```
E : constant Integer := 10;
38
          I : Extended_Index;
39
       begin
40
          New_Line;
41
          Put_Line
42
            ("Removing all elements with value of "
43
             & Integer'Image (E) & "...");
44
          loop
45
             I := V.Find_Index (E);
46
             exit when I = No_Index;
47
             V.Delete (I);
48
          end loop;
49
50
       end;
51
52
       -- Remove elements using a cursor
53
54
       declare
55
         E : constant Integer := 13;
56
          C : Cursor;
57
       begin
58
          New_Line;
59
          Put_Line
60
            ("Removing all elements with value of "
61
             & Integer'Image (E) & "...");
62
          loop
63
             C := V.Find(E);
64
             exit when C = No_Element;
65
             V.Delete (C);
66
          end loop;
67
       end;
68
69
       Show_Elements (V);
70
   end Show_Remove_Vector_Elements;
```

Code block metadata

Project: Courses.Intro_To_Ada.Standard_Library.Show_Remove_Vector_Elements MD5: 6e364843b9638224bd9a36eb9d45e446

Runtime output

```
Vector has 7 elements
Vector elements are:
- 20
- 10
- 0
- 13
- 10
- 14
- 13

Removing all elements with value of 10...

Removing all elements with value of 13...

Vector has 3 elements
Vector elements are:
- 20
- 0
```

- 14

In this example, we remove all elements with the value 10 from the vector by retrieving their index. Likewise, we remove all elements with the value 13 by retrieving their cursor.

19.1.9 Other Operations

We've seen some operations on vector elements. Here, we'll see operations on the vector as a whole. The most prominent is the concatenation of multiple vectors, but we'll also see operations on vectors, such as sorting and sorted merging operations, that view the vector as a sequence of elements and operate on the vector considering the element's relations to each other.

We do vector concatenation using the & operator on vectors. Let's consider two vectors V1 and V2. We can concatenate them by doing V := V1 & V2. V contains the resulting vector.

The generic package **Generic_**Sorting is a child package of Ada.Containers.Vectors. It contains sorting and merging operations. Because it's a generic package, you can't use it directly, but have to instantiate it. In order to use these operations on a vector of integer values (Integer_Vectors, in our example), you need to instantiate it directly as a child of Integer_Vectors. The next example makes it clear how to do this.

After instantiating **Generic_**Sorting, we make all the operations available to us with the **use** statement. We can then call Sort to sort the vector and Merge to merge one vector into another.

The following example presents code that manipulates three vectors (V1, V2, V3) using the concatenation, sorting and merging operations:

Listing 14: show_vector_ops.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada. Containers. Vectors:
   with Ada. Text IO; use Ada. Text IO;
   procedure Show Vector Ops is
6
      package Integer Vectors is new
8
         Ada.Containers.Vectors
           (Index Type => Natural,
10
            Element Type => Integer);
11
12
      package Integer Vectors Sorting is
13
         new Integer_Vectors.Generic_Sorting;
14
15
      use Integer Vectors;
16
      use Integer_Vectors_Sorting;
17
18
      procedure Show_Elements (V : Vector) is
19
      begin
20
          New Line;
21
          Put_Line ("Vector has "
22
                     & Count_Type'Image (V.Length)
23
                     & " elements");
24
25
          if not V.Is Empty then
26
             Put Line ("Vector elements are: ");
27
             for E of V loop
28
                Put Line ("- " & Integer'Image (E));
29
             end loop;
30
                                                                          (continues on next page)
```

19.1. Vectors 213

```
end if;
31
       end Show_Elements;
32
33
       V, V1, V2, V3 : Vector;
   begin
35
       V1 := 10 & 12 & 18;
36
       V2 := 11 & 13 & 19;
37
       V3 := 15 & 19;
38
39
       New_Line;
40
       Put_Line ("---- V1 ----");
41
       Show_Elements (V1);
42
43
       New_Line;
       Put_Line ("---- V2 ----");
45
       Show_Elements (V2);
46
47
       New Line;
48
       Put Line ("---- V3 ----");
49
       Show_Elements (V3);
50
51
       New_Line;
52
       Put_Line
53
         (\overline{\ }Concatenating V1, V2 and V3 into V:");
54
55
       V := V1 & V2 & V3;
56
57
       Show_Elements (V);
58
59
       New_Line;
60
       Put_Line ("Sorting V:");
61
62
       Sort (V);
63
64
       Show_Elements (V);
65
66
       New_Line;
67
       Put_Line ("Merging V2 into V1:");
68
69
       Merge (V1, V2);
70
71
       Show_Elements (V1);
72
73
   end Show_Vector_Ops;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Vector_Ops MD5: 3301513e4e7fd2f28488966e5b24e448
```

Runtime output

```
Vector has 3 elements
Vector elements are:
- 10
- 12
- 18
```

```
---- V2 ----
Vector has 3 elements
Vector elements are:
- 11
- 13
- 19
---- V3 ----
Vector has 2 elements
Vector elements are:
- 15
- 19
Concatenating V1, V2 and V3 into V:
Vector has 8 elements
Vector elements are:
- 10
- 12
- 18
- 11
- 13
  19
  15
- 19
Sorting V:
Vector has 8 elements
Vector elements are:
- 10
  11
- 12
- 13
- 15
- 18
- 19
- 19
Merging V2 into V1:
Vector has 6 elements
Vector elements are:
- 10
- 11
  12
  13
  18
   19
```

The Reference Manual requires that the worst-case complexity of a call to Sort be $O(N^2)$ and the average complexity be better than $O(N^2)$.

19.2 Sets

Sets are another class of containers. While vectors allow duplicated elements to be inserted, sets ensure that no duplicated elements exist.

19.2. Sets 215

In the following sections, we'll see operations you can perform on sets. However, since many of the operations on vectors are similar to the ones used for sets, we'll cover them more quickly here. Please refer back to the section on vectors for a more detailed discussion.

19.2.1 Initialization and iteration

To initialize a set, you can call the Insert procedure. However, if you do, you need to ensure no duplicate elements are being inserted: if you try to insert a duplicate, you'll get an exception. If you have less control over the elements to be inserted so that there may be duplicates, you can use another option instead:

- a version of Insert that returns a Boolean value indicating whether the insertion was successful;
- the Include procedure, which silently ignores any attempt to insert a duplicated element.

To iterate over a set, you can use a **for** E **of** S loop, as you saw for vectors. This gives you a reference to each element in the set.

Let's see an example:

Listing 15: show set init.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada.Containers.Ordered Sets;
   with Ada. Text IO; use Ada. Text IO;
   procedure Show_Set_Init is
      package Integer_Sets is new
8
         Ada.Containers.Ordered Sets
9
           (Element_Type => Integer);
10
11
      use Integer Sets;
12
13
      S : Set;
14
          Same as: S : Integer Sets.Set;
15
       C : Cursor;
16
17
      Ins : Boolean;
18
   begin
      S.Insert (20);
19
      S.Insert (10);
20
      S.Insert (0):
21
      S.Insert (13);
22
23
       -- Calling S.Insert(0) now would raise
24
       -- Constraint Error because this element
25
       -- is already in the set. We instead call a
26
       -- version of Insert that doesn't raise an
27
           exception but instead returns a Boolean
28
          indicating the status
29
30
      S.Insert (0, C, Ins);
31
       if not Ins then
32
          Put Line
33
            ("Error while inserting 0 into set");
34
      end if;
35
36
       -- We can also call S.Include instead
37
```

```
-- If the element is already present,
38
       -- the set remains unchanged
39
      S.Include (0);
40
      S.Include (13);
41
      S.Include (14);
42
43
      Put_Line ("Set has "
44
                 & Count_Type'Image (S.Length)
45
                 & " elements");
46
47
48
          Iterate over set using for .. of loop
49
50
      Put_Line ("Elements:");
51
       for E of S loop
52
           Put_Line ("- " & Integer'Image (E));
53
       end loop;
54
   end Show_Set_Init;
55
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Set_Init
MD5: b87f6729fea278396347248b95a30cb6
```

Runtime output

```
Error while inserting 0 into set
Set has 5 elements
Elements:
- 0
- 10
- 13
- 14
- 20
```

19.2.2 Operations on elements

In this section, we briefly explore the following operations on sets:

- Delete and Exclude to remove elements;
- Contains and Find to verify the existence of elements.

To delete elements, you call the procedure Delete. However, analogously to the Insert procedure above, Delete raises an exception if the element to be deleted isn't present in the set. If you want to permit the case where an element might not exist, you can call Exclude, which silently ignores any attempt to delete a non-existent element.

Contains returns a Boolean value indicating whether a value is contained in the set. Find also looks for an element in a set, but returns a cursor to the element or No_Element if the element doesn't exist. You can use either function to search for elements in a set.

Let's look at an example that makes use of these operations:

Listing 16: show set element ops.adb

```
with Ada.Containers; use Ada.Containers;
with Ada.Containers.Ordered_Sets;

with Ada.Text_IO; use Ada.Text_IO;
```

(continues on next page)

19.2. Sets 217

```
procedure Show_Set_Element_Ops is
      package Integer_Sets is new
8
         Ada.Containers.Ordered_Sets
9
           (Element_Type => Integer);
10
11
      use Integer_Sets;
12
13
      procedure Show_Elements (S : Set) is
14
       begin
15
          New Line;
16
          Put_Line ("Set has "
17
                     & Count_Type'Image (S.Length)
18
                     & " elements");
          Put_Line ("Elements:");
20
          for E of S loop
21
             Put_Line ("- " & Integer'Image (E));
22
          end loop;
23
      end Show_Elements;
24
25
      S : Set;
26
   begin
27
      S.Insert (20);
28
      S.Insert (10);
29
      S.Insert (0);
30
31
      S.Insert (13);
32
      S.Delete (13);
33
34
          Calling S.Delete (13) again raises
35
          Constraint Error because the element
36
          is no longer present in the set, so
37
           it can't be deleted. We can call
38
       -- V.Exclude instead:
39
      S.Exclude (13);
40
41
      if S.Contains (20) then
42
          Put_Line ("Found element 20 in set");
43
      end if;
44
45
       -- Alternatively, we could use S.Find
46
          instead of S.Contains
47
      if S.Find (0) /= No_Element then
48
          Put_Line ("Found element 0 in set");
49
      end if;
50
51
      Show_Elements (S);
52
   end Show Set Element Ops;
53
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Set_Element_Ops MD5: 77fb2aaba4221e337b0f90dd1a49c556
```

Runtime output

```
Found element 20 in set
Found element 0 in set

Set has 3 elements
Elements:

(continues on next page)
```

```
- 0
- 10
- 20
```

In addition to ordered sets used in the examples above, the standard library also offers hashed sets. The Reference Manual requires the following average complexity of each operation:

Operations	Ordered_Sets	Hashed_Sets
InsertIncludeReplaceDeleteExcludeFind	O((log N) ²) or better	O(log N)
Subprogram using cursor	O(1)	O(1)

19.2.3 Other Operations

The previous sections mostly dealt with operations on individual elements of a set. But Ada also provides typical set operations: union, intersection, difference and symmetric difference. In contrast to some vector operations we've seen before (e.g. Merge), here you can use built-in operators, such as -. The following table lists the operations and its associated operator:

Set Operation	Operator
Union	or
Intersection	and
Difference	-
Symmetric difference	xor

The following example makes use of these operators:

Listing 17: show_set_ops.adb

```
with Ada.Containers; use Ada.Containers;
   with Ada.Containers.Ordered_Sets;
   with Ada.Text_IO; use Ada.Text_IO;
   procedure Show Set Ops is
6
      package Integer Sets is new
8
         Ada.Containers.Ordered Sets
           (Element_Type => Integer);
10
11
      use Integer_Sets;
12
13
      procedure Show_Elements (S : Set) is
14
      begin
15
          Put_Line ("Elements:");
16
          for E of S loop
17
             Put Line ("- " & Integer'Image (E));
18
                                                                         (continues on next page)
```

19.2. Sets 219

```
end loop;
19
       end Show_Elements;
20
21
       procedure Show_0p (5
22
                                 : Set;
                            Op_Name : String) is
23
       begin
24
          New_Line;
25
          Put_Line (Op_Name
26
                     & \overline{\ }(set #1, set #2) has "
27
                     & Count_Type'Image (S.Length)
28
                     & " elements");
29
       end Show_Op;
30
31
       S1, S2, S3 : Set;
32
33
   begin
       S1.Insert (0);
34
       S1.Insert (10);
35
       S1.Insert (13);
36
37
       S2.Insert (0);
38
       S2.Insert (10);
39
       S2.Insert (14);
40
41
       S3.Insert (0);
42
43
       S3.Insert (10);
44
45
       New_Line;
       Put_Line ("---- Set #1 ----");
46
       Show_Elements (S1);
47
48
       New Line;
49
       Put Line ("---- Set #2 ----");
50
       Show_Elements (S2);
51
52
53
       New_Line;
       Put_Line ("---- Set #3 ----");
       Show_Elements (S3);
55
56
       New Line;
57
       if S3.Is_Subset (S1) then
58
          Put_Line ("S3 is a subset of S1");
59
60
          Put_Line ("S3 is not a subset of S1");
61
       end if;
62
63
       S3 := S1 and S2;
64
       Show_Op (S3, "Intersection");
65
       Show_Elements (S3);
66
67
       S3 := S1 or S2;
68
       Show_Op (S3, "Union");
69
       Show_Elements (S3);
70
71
       S3 := S1 - S2;
72
       Show Op (S3, "Difference");
73
       Show_Elements (S3);
       S3 := S1 xor S2;
76
       Show_Op (S3, "Symmetric difference");
77
       Show_Elements (S3);
78
79
```

```
end Show_Set_Ops;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Set_Ops
MD5: be9086591fc643e53facaf2ffea6c26d
```

Runtime output

```
---- Set #1 ----
Elements:
  10
 13
---- Set #2 ----
Elements:
- 0
- 10
- 14
---- Set #3 ----
Elements:
- 10
S3 is a subset of S1
Intersection(set #1, set #2) has 2 elements
Elements:
- 0
  10
Union(set #1, set #2) has 4 elements
Elements:
- 0
  10
  13
- 14
Difference(set #1, set #2) has 1 elements
Elements:
- 13
Symmetric difference(set #1, set #2) has 2 elements
Elements:
- 13
  14
```

19.3 Indefinite maps

The previous sections presented containers for elements of definite types. Although most examples in those sections presented **Integer** types as element type of the containers, containers can also be used with indefinite types, an example of which is the **String** type. However, indefinite types require a different kind of containers designed specially for them.

We'll also be exploring a different class of containers: maps. They associate a key with a specific value. An example of a map is the one-to-one association between a person and their age. If we consider a person's name to be the key, the value is the person's age.

19.3.1 Hashed maps

Hashed maps are maps that make use of a hash as a key. The hash itself is calculated by a function you provide.

1 In other languages

Hashed maps are similar to dictionaries in Python and hashes in Perl. One of the main differences is that these scripting languages allow using different types for the values contained in a single map, while in Ada, both the type of key and value are specified in the package instantiation and remains constant for that specific map. You can't have a map where two elements are of different types or two keys are of different types. If you want to use multiple types, you must create a different map for each and use only one type in each map.

When instantiating a hashed map from Ada.Containers.Indefinite_Hashed_Maps, we specify following elements:

- Key Type: type of the key
- Element Type: type of the element
- Hash: hash function for the Key_Type
- Equivalent_Keys: an equality operator (e.g. =) that indicates whether two keys are to be considered equal.
 - If the type specified in Key_Type has a standard operator, you can use it, which you do by specifying that operator as the value of Equivalent Keys.

In the next example, we'll use a string as a key type. We'll use the Hash function provided by the standard library for strings (in the Ada. Strings package) and the standard equality operator.

You add elements to a hashed map by calling Insert. If an element is already contained in a map M, you can access it directly by using its key. For example, you can change the value of an element by calling M (" My_Key ") := 10. If the key is not found, an exception is raised. To verify if a key is available, use the function Contains (as we've seen above in the section on sets).

Let's see an example:

Listing 18: show hashed map.adb

```
with Ada. Containers. Indefinite Hashed Maps;
   with Ada.Strings.Hash;
   with Ada.Text_IO; use Ada.Text_IO;
   procedure Show Hashed Map is
6
      package Integer Hashed Maps is new
        Ada.Containers.Indefinite Hashed Maps
           (Key_Type
                            => String,
10
           Element_Type
                            => Integer
11
                            => Ada.Strings.Hash,
12
           Equivalent_Keys => "=");
13
14
      use Integer_Hashed_Maps;
15
16
      M : Map;
17
          Same as:
18
```

```
19
          M : Integer_Hashed_Maps.Map;
20
   begin
21
      M.Include ("Alice", 24);
      M.Include ("John", 40);
23
      M.Include ("Bob", 28);
24
25
      if M.Contains ("Alice") then
26
         Put_Line ("Alice's age is
27
                    & Integer'Image (M ("Alice")));
28
      end if;
29
30
31
       -- Update Alice's age
       -- Key must already exist in M.
32
          Otherwise an exception is raised.
33
      M ("Alice") := 25;
34
35
      New_Line; Put_Line ("Name & Age:");
36
      for C in M.Iterate loop
37
          Put_Line (Key (C) & ": "
38
                    & Integer'Image (M (C)));
39
      end loop;
40
41
   end Show_Hashed_Map;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Hashed_Map
MD5: 6117775bd9ce2b1466f448b100117ded
```

Runtime output

```
Alice's age is 24

Name & Age:
John: 40
Bob: 28
Alice: 25
```

19.3.2 Ordered maps

Ordered maps share many features with hashed maps. The main differences are:

- A hash function isn't needed. Instead, you must provide an ordering function (< operator), which the ordered map will use to order elements and allow fast access, O(log N), using a binary search.
 - If the type specified in Key_Type has a standard < operator, you can use it in a similar way as we did for Equivalent_Keys above for hashed maps.

Let's see an example:

Listing 19: show_ordered_map.adb

```
with Ada.Containers.Indefinite_Ordered_Maps;

with Ada.Text_IO; use Ada.Text_IO;

procedure Show_Ordered_Map is

package Integer_Ordered_Maps is new

(continues on next page)
```

```
Ada.Containers.Indefinite_Ordered_Maps
8
                          => String,
           (Key_Type
            Element_Type => Integer);
10
11
      use Integer_Ordered_Maps;
12
13
      M : Map;
14
   begin
15
      M.Include ("Alice", 24);
16
      M.Include ("John", 40);
17
      M.Include ("Bob",
18
19
      if M.Contains ("Alice") then
20
          Put_Line ("Alice's age is "
21
                    & Integer'Image (M ("Alice")));
22
      end if;
23
24
       -- Update Alice's age
25
       -- Key must already exist in M
26
      M ("Alice") := 25;
27
28
      New_Line; Put_Line ("Name & Age:");
29
       for C in M.Iterate loop
30
          Put_Line (Key (C) & ": "
31
                    & Integer'Image (M (C)));
32
      end loop;
33
34
   end Show_Ordered_Map;
35
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Ordered_Map MD5: 3deb3c685e767cee271b06e87727b086
```

Runtime output

```
Alice's age is 24

Name & Age:
Alice: 25
Bob: 28
John: 40
```

You can see a great similarity between the examples above and from the previous section. In fact, since both kinds of maps share many operations, we didn't need to make extensive modifications when we changed our example to use ordered maps instead of hashed maps. The main difference is seen when we run the examples: the output of a hashed map is usually unordered, but the output of a ordered map is always ordered, as implied by its name.

19.3.3 Complexity

Hashed maps are generally the fastest data structure available to you in Ada if you need to associate heterogeneous keys to values and search for them quickly. In most cases, they are slightly faster than ordered maps. So if you don't need ordering, use hashed maps.

The Reference Manual requires the following average complexity of operations:

Operations	Ordered_Maps	Hashed_Maps	
InsertIncludeReplaceDeleteExcludeFind	O((log N) ²) or better	O(log N)	
Subprogram using cursor	O(1)	O(1)	

STANDARD LIBRARY: DATES & TIMES

The standard library supports processing of dates and times using two approaches:

- Calendar approach, which is suitable for handling dates and times in general;
- Real-time approach, which is better suited for real-time applications that require enhanced precision for example, by having access to an absolute clock and handling time spans. Note that this approach only supports times, not dates.

The following sections present these two approaches.

20.1 Date and time handling

The Ada. Calendar package supports handling of dates and times. Let's look at a simple example:

Listing 1: display current time.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Ada.Calendar; use Ada.Calendar;

with Ada.Calendar.Formatting;
use Ada.Calendar.Formatting;

procedure Display_Current_Time is
Now : Time := Clock;
begin
Put_Line ("Current time: " & Image (Now));
end Display_Current_Time;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Current_Time MD5: 4a88069b33ecf80314b0164a472ff606
```

Runtime output

```
Current time: 2025-04-27 01:02:27
```

This example displays the current date and time, which is retrieved by a call to the Clock function. We call the function Image from the Ada.Calendar.Formatting package to get a **String** for the current date and time. We could instead retrieve each component using the Split function. For example:

Listing 2: display_current_year.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Ada.Calendar; use Ada.Calendar;
(continues on next page)
```

```
3
   procedure Display_Current_Year is
                   : Time := Clock;
      Now
      Now Year
                   : Year_Number;
7
      Now_Month : Month_Number;
8
      Now Day
                    : Day_Number;
9
      Now_Seconds : Day_Duration;
10
   begin
11
       Split (Now,
12
              Now_Year,
13
              Now_Month,
14
15
              Now Day,
              Now_Seconds);
16
      Put_Line ("Current year is: "
18
                 & Year_Number'Image (Now_Year));
19
      Put_Line ("Current month is: "
20
                 \& <code>Month_Number'Image (Now_Month));</code>
21
      Put_Line ("Current day is: "
22
                 & Day_Number'Image (Now_Day));
23
   end Display_Current_Year;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Current_Year MD5: fdf298ee97f225261ce3839ebd833bbe
```

Runtime output

```
Current year is: 2025
Current month is: 4
Current day is: 27
```

Here, we're retrieving each element and displaying it separately.

20.1.1 Delaying using date

You can delay an application so that it restarts at a specific date and time. We saw something similar in the chapter on tasking. You do this using a **delay until** statement. For example:

Listing 3: display delay next specific time.adb

```
with Ada. Text IO; use Ada. Text IO;
   with Ada.Calendar; use Ada.Calendar;
   with Ada.Calendar.Formatting;
   use Ada.Calendar.Formatting;
   with Ada.Calendar.Time Zones;
   use Ada.Calendar.Time Zones;
8
9
   procedure Display Delay Next Specific Time is
10
      TZ : Time Offset := UTC Time Offset;
11
      Next : Time
12
        Ada.Calendar.Formatting.Time_Of
13
                       => 2018,
          (Year
                       => 5,
           Month
15
                       => 1,
           Day
```

```
Hour
                         => 15,
17
            Minute
                         => 0,
18
            Second
                         => 0,
19
            Sub\_Second => 0.0,
            Leap_Second => False,
21
            Time_Zone
                        => TZ);
22
23
          Next = 2018-05-01 \ 15:00:00.00
24
                   (local time-zone)
25
   begin
26
       Put Line ("Let's wait until...");
27
       Put_Line (Image (Next, True, TZ));
28
29
       delay until Next;
30
31
       Put_Line ("Enough waiting!");
32
   end Display_Delay_Next_Specific_Time;
33
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Delay_Next_Specific_Time MD5: 36ec2bdce7c1e8d107fae54ef9852d3f
```

Runtime output

```
Let's wait until...
2018-05-01 15:00:00.00
Enough waiting!
```

In this example, we specify the date and time by initializing Next using a call to Time_Of, a function taking the various components of a date (year, month, etc) and returning an element of the Time type. Because the date specified is in the past, the **delay until** statement won't produce any noticeable effect. However, if we passed a date in the future, the program would wait until that specific date and time arrived.

Here we're converting the time to the local timezone. If we don't specify a timezone, *Co-ordinated Universal Time* (abbreviated to UTC) is used by default. By retrieving the time offset to UTC with a call to UTC_Time_Offset from the Ada.Calendar.Time_Zones package, we can initialize TZ and use it in the call to Time_Of. This is all we need do to make the information provided to Time_Of relative to the local time zone.

We could achieve a similar result by initializing Next with a **String**. We can do this with a call to Value from the Ada. Calendar. Formatting package. This is the modified code:

Listing 4: display_delay_next_specific_time.adb

```
with Ada. Text IO; use Ada. Text IO;
   with Ada.Calendar; use Ada.Calendar;
   with Ada. Calendar. Formatting:
   use Ada.Calendar.Formatting;
   with Ada.Calendar.Time Zones;
   use Ada.Calendar.Time Zones;
   procedure Display_Delay_Next_Specific_Time is
10
      TZ : Time Offset := UTC Time Offset;
11
      Next : Time
                          :=
12
        Ada.Calendar.Formatting.Value
13
          ("2018-05-01 15:00:00.00", TZ);
14
15
```

```
-- Next = 2018-05-01 \ 15:00:00.00
16
      - -
                 (local time-zone)
17
18
      Put_Line ("Let's wait until...");
      Put_Line (Image (Next, True, TZ));
20
21
      delay until Next;
22
23
      Put_Line ("Enough waiting!");
24
   end Display_Delay_Next_Specific_Time;
25
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Delay_Next_Specific_Time MD5: fdf6ad7fca303d4d7bd444c23e11c7bd
```

Runtime output

```
Let's wait until...
2018-05-01 15:00:00.00
Enough waiting!
```

In this example, we're again using TZ in the call to Value to adjust the input time to the current time zone.

In the examples above, we were delaying to a specific date and time. Just like we saw in the tasking chapter, we could instead specify the delay relative to the current time. For example, we could delay by 5 seconds, using the current time:

Listing 5: display_delay_next.adb

```
with Ada.Calendar; use Ada.Calendar;
1
   with Ada.Text_IO; use Ada.Text_IO;
2
3
   procedure Display_Delay_Next is
4
      D : Duration := 5.0;
5
                            ^ seconds
6
      Now : Time := Clock;
Next : Time := Now + D;
8
                                   ^ use duration to
      - -
9
       - -
                                    specify next
10
                                     point in time
11
   begin
12
      Put Line ("Let's wait "
13
                 & Duration'Image (D)
14
                  & " seconds...");
15
       delay until Next;
16
      Put Line ("Enough waiting!");
17
   end Display_Delay_Next;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Delay_Next MD5: 58360d93388c3fe027c3d9d67389efc7
```

Runtime output

```
Let's wait 5.000000000 seconds...
Enough waiting!
```

Here, we're specifying a duration of 5 seconds in D, adding it to the current time from Now, and storing the sum in Next. We then use it in the **delay until** statement.

20.2 Real-time

In addition to Ada. Calendar, the standard library also supports time operations for real-time applications. These are included in the Ada. Real_Time package. This package also include a Time type. However, in the Ada. Real_Time package, the Time type is used to represent an absolute clock and handle a time span. This contrasts with the Ada. Calendar, which uses the Time type to represent dates and times.

In the previous section, we used the Time type from the Ada. Calendar and the **delay until** statement to delay an application by 5 seconds. We could have used the Ada. Real_Time package instead. Let's modify that example:

Listing 6: display_delay_next_real_time.adb

```
with Ada.Text IO:
                       use Ada. Text IO:
   with Ada.Real Time; use Ada.Real Time;
   procedure Display Delay Next Real Time is
           : Time Span := Seconds (5);
      Next : Time
                       := Clock + D;
   begin
      Put Line ("Let's wait "
                & Duration'Image (To_Duration (D))
q
                & " seconds...");
10
      delay until Next;
11
      Put Line ("Enough waiting!");
12
   end Display Delay Next Real Time;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Delay_Next_Real_Time
MD5: a80e96c4ac7bd3ba7813f983b10cb038
```

Runtime output

```
Let's wait 5.000000000 seconds...
Enough waiting!
```

The main difference is that D is now a variable of type Time_Span, defined in the Ada. Real_Time package. We call the function Seconds to initialize D, but could have gotten a finer granularity by calling Nanoseconds instead. Also, we need to first convert D to the **Duration** type using To Duration before we can display it.

20.2.1 Benchmarking

One interesting application using the Ada.Real_Time package is benchmarking. We've used that package before in a previous section when discussing tasking. Let's look at an example of benchmarking:

Listing 7: display benchmarking.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Ada.Real_Time; use Ada.Real_Time;

procedure Display_Benchmarking is

procedure Computational_Intensive_App is
begin
delay 5.0;
end Computational_Intensive_App;
```

20.2. Real-time 231

```
10
       Start_Time, Stop_Time : Time;
11
       Elapsed Time
                               : Time_Span;
12
13
   begin
14
       Start_Time := Clock;
15
16
       Computational_Intensive_App;
17
18
       Stop_Time
                     := Clock;
19
       Elapsed_Time := Stop_Time - Start_Time;
20
21
       Put_Line ("Elapsed time: "
22
                  & Duration'Image
23
                      (To_Duration (Elapsed_Time))
                  & " seconds");
25
   end Display_Benchmarking;
26
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Benchmarking MD5: 4b20940cb613d3f634be5224f409efeb
```

Runtime output

```
Elapsed time: 5.000309109 seconds
```

This example defines a dummy Computational_Intensive_App implemented using a simple **delay** statement. We initialize Start_Time and Stop_Time from the then-current clock and calculate the elapsed time. By running this program, we see that the time is roughly 5 seconds, which is expected due to the **delay** statement.

A similar application is benchmarking of CPU time. We can implement this using the Execution_Time package. Let's modify the previous example to measure CPU time:

Listing 8: display_benchmarking_cpu_time.adb

```
with Ada.Text_I0;
                              use Ada.Text I0;
1
   with Ada.Real Time;
                              use Ada.Real Time;
   with Ada. Execution Time; use Ada. Execution Time;
   procedure Display Benchmarking CPU Time is
      procedure Computational_Intensive_App is
7
      begin
8
          delay 5.0;
9
      end Computational Intensive App;
10
11
      Start_Time, Stop_Time : CPU_Time;
12
      Elapsed Time
                              : Time Span;
13
14
   begin
15
      Start Time := Clock;
16
17
      Computational_Intensive_App;
18
19
                     := Clock;
      Stop Time
20
      Elapsed_Time := Stop_Time - Start_Time;
21
22
      Put Line ("CPU time: "
23
                 & Duration'Image
24
```

```
(To_Duration (Elapsed_Time))
    & " seconds");
end Display_Benchmarking_CPU_Time;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Benchmarking_CPU_Time
MD5: ba83ddbd05db523479be5692c4134901
```

Runtime output

```
CPU time: 0.000137429 seconds
```

In this example, Start_Time and Stop_Time are of type CPU_Time instead of Time. However, we still call the Clock function to initialize both variables and calculate the elapsed time in the same way as before. By running this program, we see that the CPU time is significantly lower than the 5 seconds we've seen before. This is because the **delay** statement doesn't require much CPU time. The results will be different if we change the implementation of Computational_Intensive_App to use a mathematical function in a long loop. For example:

Listing 9: display_benchmarking_math.adb

```
with Ada. Text IO;
                               use Ada.Text I0;
   with Ada.Real Time;
                               use Ada.Real Time;
   with Ada.Execution_Time; use Ada.Execution_Time;
   with Ada.Numerics.Generic_Elementary_Functions;
   procedure Display_Benchmarking_Math is
8
       procedure Computational Intensive App is
9
          package Funcs is new
10
            Ada. Numerics. Generic Elementary Functions
11
              (Float Type => Long Long Float);
12
          use Funcs;
13
14
          X : Long_Long_Float;
15
      begin
16
          for I in 0 .. 1_000_000 loop
17
             X := Tan (Arctan
18
                     (Tan (Arctan
19
                       (Tan (Arctan
20
                          (Tan (Arctan
21
                            (Tan (Arctan
22
                              (Tan (Arctan
23
                                (0.577)))))))))))));
24
          end loop;
25
      end Computational_Intensive_App;
26
27
      procedure Benchm_Elapsed_Time is
28
          Start_Time, Stop_Time : Time;
29
          Elapsed Time
                                  : Time Span;
30
31
32
          Start Time := Clock;
33
34
          Computational_Intensive_App;
35
36
          Stop_Time
                        := Clock;
37
          Elapsed_Time := Stop_Time - Start_Time;
                                                                           (continues on next page)
```

20.2. Real-time 233

```
39
          Put_Line ("Elapsed time: "
40
                     & Duration'Image
41
                         (To_Duration (Elapsed_Time))
                    & " seconds");
43
      end Benchm_Elapsed_Time;
44
45
      procedure Benchm_CPU_Time is
46
          Start_Time, Stop_Time : CPU_Time;
47
          Elapsed_Time
                                 : Time_Span;
48
49
      begin
50
          Start_Time := Clock;
51
          Computational_Intensive_App;
53
          Stop_Time
                        := Clock;
55
          Elapsed_Time := Stop_Time - Start_Time;
56
57
          Put_Line ("CPU time: "
58
                    & Duration'Image
59
                         (To_Duration (Elapsed_Time))
60
                     & " seconds");
61
      end Benchm_CPU_Time;
62
   begin
63
      Benchm_Elapsed_Time;
      Benchm_CPU_Time;
65
   end Display_Benchmarking_Math;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Display_Benchmarking_Math MD5: 06fe96bf03321c248ddled843648cf0b
```

Runtime output

```
Elapsed time: 1.017993127 seconds
CPU time: 1.010350826 seconds
```

Now that our dummy Computational_Intensive_App involves mathematical operations requiring significant CPU time, the measured elapsed and CPU time are much closer to each other than before.

STANDARD LIBRARY: STRINGS

In previous chapters, we've seen source-code examples using the **String** type, which is a fixed-length string type — essentialy, it's an array of characters. In many cases, this data type is good enough to deal with textual information. However, there are situations that require more advanced text processing. Ada offers alternative approaches for these cases:

- Bounded strings: similar to fixed-length strings, bounded strings have a maximum length, which is set at its instantiation. However, bounded strings are not arrays of characters. At any time, they can contain a string of varied length provided this length is below or equal to the maximum length.
- *Unbounded strings*: similar to bounded strings, unbounded strings can contain strings of varied length. However, in addition to that, they don't require a maximum length to be specified at the declaration of a string. In this sense, they are very flexible.

1 For further reading...

Although we don't specify a maximum length for unbounded strings, the limit is defined by the Reference Manual²¹:

An object of type Unbounded_String represents a **String** whose low bound is 1 and whose length can vary conceptually between 0 and **Natural**'Last.

Therefore, the implicit maximum length is **Natural** 'Last. In contrast, bounded strings have an explicit maximum length that is specified when the **Generic_Bounded_Length** package is instantiated (as we'll see *later on* (page 240)).

Another difference between bounded and unbounded strings is the strategy that is used by the compiler to allocate memory for those strings. When using GNAT, bounded strings are allocated on the stack, while unbounded strings are allocated on the heap.

The following sections present an overview of the different string types and common operations for string types.

21.1 String operations

Operations on standard (fixed-length) strings are available in the Ada. Strings. Fixed package. As mentioned previously, standard strings are arrays of elements of **Character** type with a fixed-length. That's why this child package is called Fixed.

One of the simplest operations provided is counting the number of substrings available in a string (Count) and finding their corresponding indices (Index). Let's look at an example:

 $^{^{21}\} http://www.ada-auth.org/standards/12rm/html/RM-A-4-5.html$

Listing 1: show_find_substring.adb

```
with Ada. Strings. Fixed; use Ada. Strings. Fixed;
   with Ada.Text_I0;
                             use Ada.Text I0;
2
3
   procedure Show_Find_Substring is
4
5
           : String := "Hello" & 3 * " World";
6
      P : constant String := "World";
7
      Idx : Natural;
8
       Cnt : Natural;
9
   begin
10
      Cnt := Ada.Strings.Fixed.Count
11
         (Source => S,
12
          Pattern => P);
13
14
      Put Line ("String: " & S);
15
      Put_Line ("Count for '" & P & "': "
16
                 & Natural'Image (Cnt));
17
18
      Idx := 0;
19
       for I in 1 .. Cnt loop
20
          Idx := Index
21
            (Source => S,
22
             Pattern => P,
23
             From \Rightarrow Idx + 1);
24
25
          Put_Line ("Found instance of '"
26
                    & P & "' at position: "
27
                     & Natural'Image (Idx));
28
       end loop;
29
30
   end Show_Find_Substring;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Find_Substring MD5: faa8373bf9aec9f9f5507cf55590b0c0
```

Runtime output

```
String: Hello World World
Count for 'World': 3
Found instance of 'World' at position: 7
Found instance of 'World' at position: 13
Found instance of 'World' at position: 19
```

We initialize the string S using a multiplication. Writing "Hello" & 3 * " World" creates the string Hello World World. We then call the function Count to get the number of instances of the word World in S. Next we call the function Index in a loop to find the index of each instance of World in S.

That example looked for instances of a specific substring. In the next example, we retrieve all the words in the string. We do this using Find_Token and specifying whitespaces as separators. For example:

Listing 2: show find words.adb

```
with Ada.Strings; use Ada.Strings;
with Ada.Strings.Fixed;
with Ada.Strings.Maps;
with Ada.Text_I0; use Ada.Text_I0;
(continues on next page)

with Ada.Strings.Maps;
use Ada.Text_I0;
```

```
procedure Show_Find_Words is
6
           : String := "Hello" & 3 * " World";
      F
           : Positive;
9
           : Natural;
10
          : Natural := 1;
11
12
       Whitespace : constant Character_Set :=
13
         To_Set (' ');
14
   begin
15
       Put_Line ("String: " & S);
16
       Put_Line ("String length:
17
                 & Integer'Image (S'Length));
18
19
      while I in S'Range loop
20
          Find_Token
21
            (Source => S,
22
             Set
                      => Whitespace,
23
             From
                      => I,
24
                      => Outside,
             Test
25
                      => F,
             First
26
             Last
                      => L);
27
28
          exit when L = 0;
29
30
          Put_Line ("Found word instance at position "
31
                     & Natural'Image (F)
32
                     & ": '" & S (F .. L) & "'");
33
               & "-" & F'Img & "-" & L'Img
34
35
          I := L + 1;
36
       end loop;
37
   end Show_Find_Words;
38
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Find_Words
MD5: e622f489af5901e5d31f314efc3324d2
```

Runtime output

```
String: Hello World World
String length: 23
Found word instance at position 1: 'Hello'
Found word instance at position 7: 'World'
Found word instance at position 13: 'World'
Found word instance at position 19: 'World'
```

We pass a set of characters to be used as delimitators to the procedure Find_Token. This set is a member of the Character_Set type from the Ada.Strings.Maps package. We call the To_Set function (from the same package) to initialize the set to Whitespace and then call Find_Token to loop over each valid index and find the starting index of each word. We pass Outside to the Test parameter of the Find_Token procedure to indicate that we're looking for indices that are outside the Whitespace set, i.e. actual words. The First and Last parameters of Find_Token are output parameters that indicate the valid range of the substring. We use this information to display the string (S (F .. L)).

The operations we've looked at so far read strings, but don't modify them. We next discuss operations that change the content of strings:

Operation	Description
Insert	Insert substring in a string
Overwrite	Overwrite a string with a substring
Delete	Delete a substring
Trim	Remove whitespaces from a string

All these operations are available both as functions or procedures. Functions create a new string but procedures perform the operations in place. The procedure will raise an exception if the constraints of the string are not satisfied. For example, if we have a string S containing 10 characters, inserting a string with two characters (e.g. "!!") into it produces a string containing 12 characters. Since it has a fixed length, we can't increase its size. One possible solution in this case is to specify that truncation should be applied while inserting the substring. This keeps the length of S fixed. Let's see an example that makes use of both function and procedure versions of Insert, Overwrite, and Delete:

Listing 3: show adapted strings.adb

```
with Ada. Strings;
                              use Ada.Strings;
   with Ada. Strings. Fixed; use Ada. Strings. Fixed;
   with Ada.Text_I0;
                              use Ada.Text_I0;
   procedure Show Adapted Strings is
5
6
       S
           : String := "Hello World";
           : constant String := "World";
8
          : constant String := "Beautiful";
9
10
       procedure Display_Adapted_String
11
                  : String;
         (Source
12
          Before
                    : Positive;
13
          New Item : String;
14
          Pattern : String)
15
16
          S_Ins_In : String := Source;
17
          S Ovr In : String := Source;
18
          S_Del_In : String := Source;
19
20
          S_Ins : String :=
21
            Insert (Source,
22
                     Before,
23
                     New_Item & " ");
24
          S Ovr : String :=
25
            Overwrite (Source,
26
                        Before,
27
                        New Item);
28
          S Del : String :=
            Trim (Delete (Source,
30
                            Before,
31
                            Before +
32
                              Pattern'Length - 1),
33
                   Ada.Strings.Right);
34
       begin
35
          Insert (S_Ins In,
36
                   Before,
37
                   New Item,
38
                   Right);
39
40
          Overwrite (S Ovr In,
41
                      Before,
42
```

```
New_Item,
43
                      Right);
44
45
          Delete (S_Del_In,
                   Before,
47
                   Before + Pattern'Length - 1);
48
49
          Put_Line ("Original:
50
                     & Source & "'");
51
52
          Put Line ("Insert:
53
                     & S_Ins & "'");
54
          Put_Line ("Overwrite: '"
55
                     & S_0vr & "'");
56
                                 1.0
          Put_Line ("Delete:
57
                     & S_Del & "'");
59
          Put_Line ("Insert
                                 (in-place): '"
60
                     & S_Ins_In & "'");
61
          Put_Line ("Overwrite (in-place): '"
62
                     & S_0vr_In & "'");
63
          Put_Line ("Delete
                               (in-place): '"
64
                     & S_Del_In & "'");
65
       end Display_Adapted_String;
66
67
      Idx : Natural;
68
   begin
69
      Idx := Index
70
         (Source => S,
71
          Pattern => P);
72
73
       if Idx > 0 then
74
          Display_Adapted_String (S, Idx, N, P);
75
       end if;
76
   end Show_Adapted_Strings;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Adapted_Strings
MD5: b31b6bc94d8bdbec717c6b6b2534beb6
```

Runtime output

```
Original: 'Hello World'
Insert: 'Hello Beautiful World'
Overwrite: 'Hello Beautiful'
Delete: 'Hello'
Insert (in-place): 'Hello Beaut'
Overwrite (in-place): 'Hello Beaut'
Delete (in-place): 'Hello '
```

In this example, we look for the index of the substring World and perform operations on this substring within the outer string. The procedure Display_Adapted_String uses both versions of the operations. For the procedural version of Insert and Overwrite, we apply truncation to the right side of the string (Right). For the Delete procedure, we specify the range of the substring, which is replaced by whitespaces. For the function version of Delete, we also call Trim which trims the trailing whitespace.

21.2 Limitation of fixed-length strings

Using fixed-length strings is usually good enough for strings that are initialized when they are declared. However, as seen in the previous section, procedural operations on strings cause difficulties when done on fixed-length strings because fixed-length strings are arrays of characters. The following example shows how cumbersome the initialization of fixed-length strings can be when it's not performed in the declaration:

Listing 4: show_char_array.adb

```
use Ada.Text_I0;
   with Ada.Text_I0;
   procedure Show_Char_Array is
3
      S : String (1 .. 15);
4
         Strings are arrays of Character
5
   begin
6
      S := "Hello
7
      -- Alternatively:
8
9
      -- #1:
10
              S (1 .. 5) := "Hello";
11
              S (6 .. S'Last) := (others => ' ');
12
13
14
              S := ('H', 'e', 'l', 'l', 'o',
15
                    others => ' ');
16
17
      Put Line ("String: " & S);
18
      Put_Line ("String Length: "
19
                 & Integer'Image (S'Length));
20
   end Show_Char_Array;
21
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Char_Array MD5: 9f3df03c9c5336184139cf2a22f2cb7e
```

Runtime output

```
String: Hello
String Length: 15
```

In this case, we can't simply write S := "Hello" because the resulting array of characters for the Hello constant has a different length than the S string. Therefore, we need to include trailing whitespaces to match the length of S. As shown in the example, we could use an exact range for the initialization (S(1...5)) or use an explicit array of individual characters.

When strings are initialized or manipulated at run-time, it's usually better to use bounded or unbounded strings. An important feature of these types is that they aren't arrays, so the difficulties presented above don't apply. Let's start with bounded strings.

21.3 Bounded strings

Bounded strings are defined in the Ada.Strings.Bounded.Generic_Bounded_Length package. Because this is a generic package, you need to instantiate it and set the maximum length of the bounded string. You can then declare bounded strings of the Bounded_String type.

Both bounded and fixed-length strings have a maximum length that they can hold. How-

ever, bounded strings are not arrays, so initializing them at run-time is much easier. For example:

Listing 5: show bounded string.adb

```
with Ada.Strings;
                               use Ada.Strings;
   with Ada.Strings.Bounded;
   with Ada.Text_I0;
                               use Ada. Text IO;
3
   procedure Show_Bounded_String is
      package B_Str is new
         Ada.Strings.Bounded.Generic_Bounded_Length
           (Max => 15);
8
      use B_Str;
9
10
      S1, S2 : Bounded_String;
11
12
      procedure Display_String_Info
13
         (S : Bounded String)
14
15
      begin
16
          Put_Line ("String: " & To_String (S));
17
          Put_Line ("String Length:
18
                    & Integer'Image (Length (S)));
19
              String:
20
                       S'Length => ok
21
              Bounded String:
22
                       S'Length => compilation error:
23
                                    bounded strings are
24
                                    not arrays!
25
26
          Put_Line ("Max.
                             Length: "
27
                     & Integer'Image (Max_Length));
28
      end Display_String_Info;
29
30
   begin
31
      S1 := To_Bounded_String ("Hello");
32
      Display_String_Info (S1);
33
34
      S2 := To Bounded String ("Hello World");
35
      Display_String_Info (S2);
36
37
       S1 := To_Bounded_String
38
         ("Something longer to say here...",
39
          Right);
40
      Display_String_Info (S1);
41
   end Show_Bounded_String;
42
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Bounded_String MD5: a51fdeacfd43923145ee92bf5c72ecd6
```

Runtime output

```
String: Hello
String Length: 5
Max. Length: 15
String: Hello World
String Length: 11
Max. Length: 15
String: Something longe
```

```
String Length: 15
Max. Length: 15
```

By using bounded strings, we can easily assign to S1 and S2 multiple times during execution. We use the To_Bounded_String and To_String functions to convert, in the respective direction, between fixed-length and bounded strings. A call to To_Bounded_String raises an exception if the length of the input string is greater than the maximum capacity of the bounded string. To avoid this, we can use the truncation parameter (Right in our example).

Bounded strings are not arrays, so we can't use the 'Length attribute as we did for fixed-length strings. Instead, we call the Length function, which returns the length of the bounded string. The Max_Length constant represents the maximum length of the bounded string that we set when we instantiated the package.

After initializing a bounded string, we can manipulate it. For example, we can append a string to a bounded string using Append or concatenate bounded strings using the & operator. Like so:

Listing 6: show bounded string op.adb

```
with Ada.Strings;
                               use Ada.Strings;
   with Ada. Strings. Bounded;
   with Ada.Text I0;
                               use Ada. Text IO;
   procedure Show Bounded String Op is
5
       package B Str is new
6
         Ada.Strings.Bounded.Generic Bounded Length
7
           (Max => 30);
8
      use B Str;
9
10
      S1, S2 : Bounded String;
11
   begin
12
       S1 := To Bounded String ("Hello");
13
       -- Alternatively:
14
15
       -- A := Null Bounded String & "Hello";
16
17
       Append (S1, "World");
18
       -- Alternatively:
19
             Append (A, " World", Right);
20
21
      Put_Line ("String: " & To_String (S1));
22
23
      S2 := To Bounded String ("Hello!");
24
      S1 := S1 & " " & S2:
25
       Put Line ("String: " & To String (S1));
26
   end Show Bounded String Op;
27
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Bounded_String_Op
MD5: c7c6a840c314a9cd9f75aac082a63159
```

Runtime output

```
String: Hello World
String: Hello World Hello!
```

We can initialize a bounded string with an empty string using the Null_Bounded_String constant. Also, we can use the Append procedure and specify the truncation mode like we do with the To_Bounded_String function.

21.4 Unbounded strings

Unbounded strings are defined in the Ada.Strings.Unbounded package. This is *not* a generic package, so we don't need to instantiate it before using the Unbounded_String type. As you may recall from the previous section, bounded strings require a package instantiation.

Unbounded strings are similar to bounded strings. The main difference is that they can hold strings of any size and adjust according to the input string: if we assign, e.g., a 10-character string to an unbounded string and later assign a 50-character string, internal operations in the container ensure that memory is allocated to store the new string. In most cases, developers don't need to worry about these operations. Also, no truncation is necessary.

Initialization of unbounded strings is very similar to bounded strings. Let's look at an example:

Listing 7: show unbounded string.adb

```
with Ada. Text IO; use Ada. Text IO;
   with Ada.Strings; use Ada.Strings;
   with Ada. Strings. Unbounded;
   use Ada.Strings.Unbounded;
5
6
   procedure Show Unbounded String is
7
      S1, S2 : Unbounded_String;
8
9
      procedure Display_String_Info
10
         (S : Unbounded_String)
11
      is
12
      beain
13
          Put Line ("String: " & To String (S));
14
          Put_Line ("String Length: '
15
                    & Integer'Image (Length (S)));
16
      end Display String Info;
17
   begin
18
      S1 := To Unbounded String ("Hello");
19
       -- Alternatively:
20
21
       -- A := Null Unbounded String & "Hello";
22
23
      Display_String_Info (S1);
24
25
      S2 := To Unbounded String ("Hello World");
26
      Display String Info (S2);
27
28
      S1 := To Unbounded String
29
               ("Something longer to say here...");
      Display_String_Info (S1);
31
   end Show_Unbounded_String;
32
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Unbounded_String MD5: 904402992c96eb393b875d1b7cf49c1b
```

Runtime output

```
String: Hello
String Length: 5
String: Hello World

(continues on next page)
```

```
String Length: 11
String: Something longer to say here...
String Length: 31
```

Like bounded strings, we can assign to S1 and S2 multiple times during execution and use the To_Unbounded_String and To_String functions to convert back-and-forth between fixed-length strings and unbounded strings. However, in this case, truncation is not needed.

And, just like for bounded strings, you can use the Append procedure and the & operator for unbounded strings. For example:

Listing 8: show_unbounded_string_op.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   with Ada.Strings.Unbounded;
   use Ada.Strings.Unbounded;
   procedure Show Unbounded String Op is
6
       S1, S2 : Unbounded_String :=
                   Null_Unbounded_String;
   begin
9
      S1 := S1 & "Hello";
10
      S2 := S2 & "Hello!";
11
12
      Append (S1, "World");
13
      Put_Line ("String: " & To_String (S1));
14
15
      S1 := S1 \& " " \& S2;
Put_Line ("String: " & To_String (S1));
16
17
   end Show_Unbounded_String_Op;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Unbounded_String_Op MD5: 806e24a6dd0bc87e76f73a22e42ba390
```

Runtime output

```
String: Hello World
String: Hello World Hello!
```

In this example, we're concatenating the unbounded S1 and S2 strings with the "Hello" and "Hello!" strings, respectively. Also, we're using the Append procedure, just like we did with bounded strings.

STANDARD LIBRARY: FILES AND STREAMS

Ada provides different approaches for file input/output (I/O):

- Text I/O, which supports file I/O in text format, including the display of information on the console.
- Sequential I/O, which supports file I/O in binary format written in a sequential fashion for a specific data type.
- *Direct I/O*, which supports file I/O in binary format for a specific data type, but also supporting access to any position of a file.
- Stream I/O, which supports I/O of information for multiple data types, including objects of unbounded types, using files in binary format.

This table presents a summary of the features we've just seen:

File I/O option	Format	Random access	Data types
Text I/O	text		string type
Sequential I/O	binary		single type
Direct I/O	binary	\checkmark	single type
Stream I/O	binary	\checkmark	multiple types

In the following sections, we discuss details about these I/O approaches.

22.1 Text I/O

In most parts of this course, we used the Put_Line procedure to display information on the console. However, this procedure also accepts a **File_Type** parameter. For example, you can select between standard output and standard error by setting this parameter explicitly:

```
Listing 1: show_std_text_out.adb
```

```
with Ada.Text_IO; use Ada.Text_IO;

procedure Show_Std_Text_Out is
begin
Put_Line (Standard_Output, "Hello World #1");
Put_Line (Standard_Error, "Hello World #2");
end Show_Std_Text_Out;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Std_Text_Out
MD5: 4d75bd2906226897244e3d2a611c9725
```

Runtime output

```
Hello World #1
Hello World #2
```

You can also use this parameter to write information to any text file. To create a new file for writing, use the Create procedure, which initializes a **File_Type** element that you can later pass to Put_Line (instead of, e.g., Standard_Output). After you finish writing information, you can close the file by calling the Close procedure.

You use a similar method to read information from a text file. However, when opening the file, you must specify that it's an input file (In_File) instead of an output file. Also, instead of calling the Put_Line procedure, you call the Get_Line function to read information from the file.

Let's see an example that writes information into a new text file and then reads it back from the same file:

Listing 2: show_simple_text_file_io.adb

```
with Ada. Text IO; use Ada. Text IO;
1
2
   procedure Show Simple Text File IO is
3
                  : File Type;
       F
       File_Name : constant String := "simple.txt";
6
   begin
       Create (F, Out_File, File_Name);
       Put_Line (F, "Hello World #1");
Put_Line (F, "Hello World #2");
8
9
       Put_Line (F, "Hello World #3");
10
       Close (F);
11
12
       Open (F, In File, File Name);
13
       while not End Of File (F) loop
          Put_Line (Get_Line (F));
15
       end loop;
16
       Close (F);
17
   end Show Simple Text File IO;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Simple_Text_File_I0 MD5: 7461e946eef18c93219fa4ce3afbb1ea
```

Runtime output

```
Hello World #1
Hello World #2
Hello World #3
```

In addition to the Create and Close procedures, the standard library also includes a Reset procedure, which, as the name implies, resets (erases) all the information from the file. For example:

Listing 3: show_text_file_reset.adb

(continues on next page)

```
Reset (F);
9
      Put_Line (F, "Hello World #2");
10
      Close (F);
11
12
      Open (F, In_File, File_Name);
13
      while not End_Of_File (F) loop
14
          Put_Line (Get_Line (F));
15
       end loop;
16
      Close (F);
17
   end Show_Text_File_Reset;
18
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Text_File_Reset MD5: 5e5498f03b2c829513af062c5959fc93
```

Runtime output

```
Hello World #2
```

By running this program, we notice that, although we've written the first string ("Hello World #1") to the file, it has been erased because of the call to Reset.

In addition to opening a file for reading or writing, you can also open an existing file and append to it. Do this by calling the Open procedure with the Append_File option.

When calling the Open procedure, an exception is raised if the specified file isn't found. Therefore, you should handle exceptions in that context. The following example deletes a file and then tries to open the same file for reading:

Listing 4: show_text_file_input_except.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   procedure Show_Text_File_Input_Except is
3
      F
                : File_Type;
      File_Name : constant String := "simple.txt";
   begin
       -- Open output file and delete it
      Create (F, Out_File, File_Name);
8
      Delete (F);
9
10
      -- Try to open deleted file
11
      Open (F, In_File, File_Name);
12
      Close (F);
13
   exception
14
      when Name_Error =>
15
         Put_Line ("File does not exist");
16
      when others =>
17
         Put Line
18
            ("Error while processing input file");
19
   end Show_Text_File_Input_Except;
20
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Text_File_Input_Except MD5: c8d257091831c48d10b6e70e34b4261b
```

Runtime output

```
File does not exist
```

22.1. Text I/O 247

In this example, we create the file by calling Create and then delete it by calling Delete. After the call to Delete, we can no longer use the **File_Type** element. After deleting the file, we try to open the non-existent file, which raises a Name Error exception.

22.2 Sequential I/O

The previous section presented details about text file I/O. Here, we discuss doing file I/O in binary format. The first package we'll explore is the Ada. Sequential_IO package. Because this package is a generic package, you need to instantiate it for the data type you want to use for file I/O. Once you've done that, you can use the same procedures we've seen in the previous section: Create, Open, Close, Reset and Delete. However, instead of calling the Get_Line and Put_Line procedures, you'd call the Read and Write procedures.

In the following example, we instantiate the Ada. Sequential_IO package for floating-point types:

Listing 5: show seq float io.adb

```
with Ada. Text IO;
   with Ada.Sequential IO;
2
   procedure Show_Seq_Float_IO is
      package Float_IO is
         new Ada.Sequential_IO (Float);
6
      use Float IO;
7
8
                 : Float IO.File Type;
9
      File Name : constant String :=
10
                      "float file.bin";
11
12
      Create (F, Out_File, File_Name);
13
      Write (F, 1.5);
14
15
      Write (F, 2.4);
      Write (F, 6.7);
      Close (F);
17
18
      declare
19
          Value : Float;
20
      begin
21
          Open (F, In File, File Name);
22
          while not End Of File (F) loop
23
             Read (F, Value);
24
             Ada.Text IO.Put Line
25
               (Float'Image (Value));
          end loop;
27
          Close (F);
28
      end:
29
   end Show_Seq_Float_I0;
30
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Seq_Float_IO MD5: 27aa5daf92cba5df23fdc55c3578aa34
```

Runtime output

```
1.50000E+00
2.40000E+00
6.70000E+00
```

We use the same approach to read and write complex information. The following example

uses a record that includes a Boolean and a floating-point value:

Listing 6: show seq rec io.adb

```
with Ada. Text IO;
   with Ada.Sequential_I0;
   procedure Show_Seq_Rec_IO is
       type Num_Info is record
5
          Valid : Boolean := False;
6
          Value : Float;
7
      end record;
8
9
      procedure Put_Line (N : Num_Info) is
10
      begin
11
          if N.Valid then
12
             Ada.Text_IO.Put_Line
13
                ("(ok,
14
                & Float'Image (N.Value) & ")");
15
          else
16
             Ada.Text_IO.Put_Line
17
                            -----)"):
                ("(not ok,
18
          end if;
19
      end Put_Line;
20
21
      package Num Info IO is new
22
         Ada.Sequential_IO (Num_Info);
23
       use Num_Info_IO;
24
25
                 : Num_Info_IO.File_Type;
26
      File_Name : constant String :=
27
                      "float_file.bin";
28
   beain
29
      Create (F, Out_File, File_Name);
30
      Write (F, (True, 1.5));
31
      Write (F,
                  (False, 2.4));
32
      Write (F,
                   (True, 6.7));
33
      Close (F);
34
35
      declare
36
          Value : Num_Info;
37
      begin
38
          Open (F, In_File, File_Name);
39
          while not End_Of_File (F) loop
40
             Read (F, Value);
41
             Put_Line (Value);
42
          end loop;
43
          Close (F);
44
       end;
45
   end Show_Seq_Rec_IO;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Seq_Rec_IO
MD5: a88b1428cc50745dce0509087e74adb7
```

Runtime output

```
(ok, 1.50000E+00)
(not ok, -----)
(ok, 6.70000E+00)
```

As the example shows, we can use the same approach we used for floating-point types to

perform file I/O for this record. Once we instantiate the Ada.Sequential_IO package for the record type, file I/O operations are performed the same way.

22.3 Direct I/O

Direct I/O is available in the Ada.Direct_IO package. This mechanism is similar to the sequential I/O approach just presented, but allows us to access any position in the file. The package instantiation and most operations are very similar to sequential I/O. To rewrite the Show_Seq_Float_IO application presented in the previous section to use the Ada.Direct_IO package, we just need to replace the instances of the Ada.Sequential_IO package by the Ada.Direct_IO package. This is the new source code:

Listing 7: show_dir_float_io.adb

```
with Ada.Text IO;
1
   with Ada.Direct IO;
2
   procedure Show Dir Float IO is
      package Float IO is new Ada.Direct IO (Float);
      use Float_I0;
7
                 : Float_IO.File_Type;
8
      File_Name : constant String :=
9
                      "float_file.bin";
10
   begin
11
      Create (F, Out_File, File_Name);
12
      Write (F, 1.5);
13
      Write (F,
                  2.4);
14
      Write (F, 6.7);
15
      Close (F);
16
17
      declare
18
         Value : Float;
19
20
          Open (F, In_File, File Name);
21
         while not End Of File (F) loop
22
             Read (F, Value);
23
             Ada.Text_IO.Put_Line
24
               (Float'Image (Value));
25
         end loop;
26
         Close (F);
27
      end:
28
   end Show_Dir_Float_I0;
29
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Dir_Float_IO MD5: e4e5855976de44f53a821eb90dcbb206
```

Runtime output

```
1.50000E+00
2.40000E+00
6.70000E+00
```

Unlike sequential I/O, direct I/O allows you to access any position in the file. However, it doesn't offer an option to append information to a file. Instead, it provides an Inout_File mode allowing reading and writing to a file via the same File Type element.

To access any position in the file, call the Set_Index procedure to set the new position / index. You can use the Index function to retrieve the current index. Let's see an example:

Listing 8: show_dir_float_in_out_file.adb

```
with Ada. Text IO;
   with Ada.Direct_IO;
2
3
   procedure Show_Dir_Float_In_Out_File is
4
      package Float_IO is new Ada.Direct_IO (Float);
5
      use Float IO;
6
7
                 : Float_IO.File_Type;
8
      File Name : constant String :=
9
                      "float_file.bin";
10
   begin
11
           Open file for input / output
12
      Create (F, Inout_File, File_Name);
13
      Write (F, 1.5);
14
      Write (F, 2.4);
15
      Write (F, 6.7);
16
17
       -- Set index to previous position
18
      -- and overwrite value
19
      Set_Index (F, Index (F) - 1);
20
      Write (F, 7.7);
21
22
      declare
23
         Value : Float;
24
      begin
25
              Set index to start of file
26
          Set_Index (F, 1);
27
28
         while not End Of File (F) loop
29
             Read (F, Value);
30
             Ada.Text_IO.Put_Line
31
               (Float'Image (Value));
32
         end loop:
33
          Close (F);
      end:
35
   end Show_Dir_Float_In_Out_File;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Dir_Float_In_Out_File MD5: 17b83a16ab8fa30f07cf8a0bd54078a1
```

Runtime output

```
1.50000E+00
2.40000E+00
7.70000E+00
```

By running this example, we see that the file contains 7.7, rather than the previous 6.7 that we wrote. We overwrote the value by changing the index to the previous position before doing another write.

In this example we used the Inout_File mode. Using that mode, we just changed the index back to the initial position before reading from the file (Set_Index (F, 1)) instead of closing the file and reopening it for reading.

22.3. Direct I/O 251

22.4 Stream I/O

All the previous approaches for file I/O in binary format (sequential and direct I/O) are specific for a single data type (the one we instantiate them with). You can use these approaches to write objects of a single data type that may be an array or record (potentially with many fields), but if you need to create and process files that include different data types, or any objects of an unbounded type, these approaches are not sufficient. Instead, you should use stream I/O.

Stream I/O shares some similarities with the previous approaches. We still use the Create, Open and Close procedures. However, instead of accessing the file directly via a **File_Type** element, you use a Stream_Access element. To read and write information, you use the 'Read or 'Write attributes of the data types you're reading or writing.

Let's look at a version of the Show_Dir_Float_IO procedure from the previous section that makes use of stream I/O instead of direct I/O:

Listing 9: show float stream.adb

```
with Ada. Text IO;
1
2
   with Ada.Streams.Stream IO:
3
   use Ada.Streams.Stream IO;
4
5
   procedure Show Float Stream is
6
      F
                 : File Type;
      S
8
                 : Stream_Access;
      File Name : constant String :=
9
                      "float_file.bin";
10
   begin
11
      Create (F, Out_File, File_Name);
12
      S := Stream(F);
13
14
      Float'Write (S, 1.5);
15
      Float'Write (S, 2.4);
16
      Float'Write (S, 6.7);
17
18
      Close (F);
19
20
      declare
21
          Value : Float;
22
      begin
23
          Open (F, In_File, File_Name);
24
          S := Stream(F);
25
26
          while not End Of File (F) loop
27
             Float'Read (S, Value);
28
             Ada.Text_IO.Put_Line
29
                (Float'Image (Value));
30
          end loop:
31
          Close (F);
32
33
   end Show Float Stream;
34
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Float_Stream
MD5: 34ccf04b0821074a332019ac0e38bb3e
```

Runtime output

```
1.50000E+00
2.40000E+00
6.70000E+00
```

After the call to Create, we retrieve the corresponding Stream_Access element by calling the Stream function. We then use this stream to write information to the file via the 'Write attribute of the Float type. After closing the file and reopening it for reading, we again retrieve the corresponding Stream_Access element and processed to read information from the file via the 'Read attribute of the Float type.

You can use streams to create and process files containing different data types within the same file. You can also read and write unbounded data types such as strings. However, when using unbounded data types you must call the 'Input and 'Output attributes of the unbounded data type: these attributes write information about bounds or discriminants in addition to the object's actual data.

The following example shows file I/O that mixes both strings of different lengths and floating-point values:

Listing 10: show string stream.adb

```
with Ada. Text IO;
1
2
   with Ada.Streams.Stream IO:
3
   use Ada.Streams.Stream IO;
4
5
   procedure Show_String_Stream is
6
      F
                 : File Type;
7
      S
8
                 : Stream_Access;
      File Name : constant String :=
9
                      "float_file.bin";
10
11
      procedure Output (S : Stream Access;
12
                          FV : Float;
13
                          SV : String) is
14
      begin
15
          String'Output (S, SV);
16
          Float'Output (S, FV);
17
      end Output;
18
19
      procedure Input Display (S : Stream Access) is
20
          SV : String := String'Input (S);
21
          FV : Float := Float'Input (S);
22
23
          Ada.Text_IO.Put_Line (Float'Image (FV)
24
                                  & " --- " & SV);
25
      end Input Display;
26
27
   begin
28
      Create (F, Out_File, File_Name);
29
      S := Stream(F);
30
31
      Output (S, 1.5, "Hi!!");
32
      Output (S, 2.4, "Hello world!");
33
      Output (S, 6.7, "Something longer here...");
34
35
      Close (F);
36
37
      Open (F, In File, File Name);
38
      S := Stream(F);
39
40
      while not End_Of_File (F) loop
                                                                           (continues on next page)
```

22.4. Stream I/O 253

```
Input_Display (S);
end loop;
Close (F);

end Show_String_Stream;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_String_Stream MD5: 3ae8276ada5f24cab49994e368e0fa34
```

Runtime output

```
1.50000E+00 --- Hi!!
2.40000E+00 --- Hello world!
6.70000E+00 --- Something longer here...
```

When you use Stream I/O, no information is written into the file indicating the type of the data that you wrote. If a file contains data from different types, you must reference types in the same order when reading a file as when you wrote it. If not, the information you get will be corrupted. Unfortunately, strong data typing doesn't help you in this case. Writing simple procedures for file I/O (as in the example above) may help ensuring that the file format is consistent.

Like direct I/O, stream I/O support also allows you to access any location in the file. However, when doing so, you need to be extremely careful that the position of the new index is consistent with the data types you're expecting.

STANDARD LIBRARY: NUMERICS

The standard library provides support for common numeric operations on floating-point types as well as on complex types and matrices. In the sections below, we present a brief introduction to these numeric operations.

23.1 Elementary Functions

The Ada.Numerics.Elementary_Functions package provides common operations for floating-point types, such as square root, logarithm, and the trigonometric functions (e.g., sin, cos). For example:

Listing 1: show_elem_math.adb

```
with Ada.Text_IO; use Ada.Text_IO;
   with Ada. Numerics; use Ada. Numerics;
2
   with Ada.Numerics.Elementary_Functions;
4
   use Ada.Numerics.Elementary_Functions;
5
   procedure Show Elem Math is
7
       X : Float;
8
9
   begin
       X := 2.0;
10
       Put_Line ("Square root of "
11
                 & Float'Image (X)
12
                 & " is "
13
                 & Float'Image (Sqrt (X)));
14
15
       X := e;
16
       Put_Line ("Natural log of "
17
                 & Float'Image (X)
18
                 & " is "
19
                 & Float'Image (Log (X)));
20
21
       X := 10.0 ** 6.0;
22
       Put_Line ("Log_10
                               of "
23
                 & Float'Image (X)
24
                 & " is "
25
                 & Float'Image (Log (X, 10.0)));
26
27
       X := 2.0 ** 8.0;
28
                                of "
       Put Line ("Log 2
29
                 & Float'Image (X)
30
                 & " is "
31
                 & Float'Image (Log (X, 2.0)));
32
33
       X := Pi;
```

(continues on next page)

```
Put_Line ("Cos
                                of "
35
                  & Float'Image (X)
36
                  & " is "
37
                  & Float'Image (Cos (X)));
38
39
      X := -1.0;
40
      Put_Line ("Arccos
                                of "
41
                  & Float'Image (X)
42
                  & " is "
43
                  & Float'Image (Arccos (X)));
44
   end Show Elem Math;
45
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Elem_Math MD5: 17511d7e17cd98d4b6e49ad302d6dcb6
```

Runtime output

```
Square root of 2.00000E+00 is 1.41421E+00
Natural log of 2.71828E+00 is 1.00000E+00
Log_10 of 1.00000E+06 is 6.00000E+00
Log_2 of 2.56000E+02 is 8.00000E+00
Cos of 3.14159E+00 is -1.00000E+00
Arccos of -1.00000E+00 is 3.14159E+00
```

Here we use the standard e and Pi constants from the Ada. Numerics package.

The Ada.Numerics.Elementary_Functions package provides operations for the **Float** type. Similar packages are available for **Long_Float** and **Long_Long_Float** types. For example, the Ada.Numerics.Long_Elementary_Functions package offers the same set of operations for the **Long_Float** type. In addition, the Ada.Numerics. **Generic_**Elementary_Functions package is a generic version of the package that you can instantiate for custom floating-point types. In fact, the Elementary_Functions package can be defined as follows:

```
package Elementary_Functions is new
Ada.Numerics.Generic_Elementary_Functions (Float);
```

23.2 Random Number Generation

The Ada.Numerics.Float_Random package provides a simple random number generator for the range between 0.0 and 1.0. To use it, declare a generator G, which you pass to Random. For example:

Listing 2: show_float_random_num.adb

```
with Ada.Text_IO; use Ada.Text_IO;

with Ada.Numerics.Float_Random;
use Ada.Numerics.Float_Random;

procedure Show_Float_Random_Num is
    G : Generator;
    X : Uniformly_Distributed;
begin
    Reset (G);

(continues on next page)
```

```
Put_Line ("Some random numbers between "
12
                 & Float'Image
13
                      (Uniformly_Distributed'First)
14
                 & " and "
                 & Float'Image
16
                      (Uniformly_Distributed'Last)
17
                 & ":");
18
      for I in 1 .. 15 loop
19
         X := Random (G);
20
         Put_Line (Float'Image (X));
21
      end loop;
22
   end Show_Float_Random_Num;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Float_Random_Num MD5: cf38ab00e27bad4309010e678113dd36
```

Runtime output

```
Some random numbers between 0.00000E+00 and 1.00000E+00:
 8.13272E-01
 4.12954E-01
7.24715E-01
 5.94629E-01
 6.73457E-01
 3.78586E-01
 1.11984E-01
 2.71231E-01
 1.63945E-01
 2.29341E-01
 7.02121E-02
 5.94455E-01
 3.02021E-02
 9.42785E-01
 6.55720E-01
```

The standard library also includes a random number generator for discrete numbers, which is part of the Ada.Numerics.Discrete_Random package. Since it's a generic package, you have to instantiate it for the desired discrete type. This allows you to specify a range for the generator. In the following example, we create an application that displays random integers between 1 and 10:

Listing 3: show discrete random num.adb

```
with Ada.Text IO; use Ada.Text IO;
   with Ada.Numerics.Discrete_Random;
   procedure Show Discrete Random Num is
4
      subtype Random_Range is Integer range 1 .. 10;
6
      package R is new
8
        Ada.Numerics.Discrete_Random (Random_Range);
9
      use R:
10
11
      G : Generator;
12
      X : Random Range;
13
14
      Reset (G);
15
16
                                                                          (continues on next page)
```

```
Put_Line ("Some random numbers between "
17
                 & Integer'Image (Random_Range'First)
18
                 & " and "
19
                 & Integer'Image (Random_Range'Last)
20
                 & ":");
21
22
      for I in 1 .. 15 loop
23
         X := Random(G);
24
         Put_Line (Integer'Image (X));
25
      end loop;
26
   end Show Discrete Random Num;
27
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Discrete_Random_Num MD5: 892f6525477f9a2c56f88885de011fba
```

Runtime output

```
Some random numbers between 1 and 10:
7
8
4
1
1
3
10
10
8
4
4
2
6
1
3
1
```

Here, package R is instantiated with the Random_Range type, which has a constrained range between 1 and 10. This allows us to control the range used for the random numbers. We could easily modify the application to display random integers between 0 and 20 by changing the specification of the Random_Range type. We can also use floating-point or fixed-point types.

23.3 Complex Types

The Ada.Numerics.Complex_Types package provides support for complex number types and the Ada.Numerics.Complex_Elementary_Functions package provides support for common operations on complex number types, similar to the Ada.Numerics. Elementary_Functions package. Finally, you can use the Ada.Text_IO.Complex_IO package to perform I/O operations on complex numbers. In the following example, we declare variables of the Complex type and initialize them using an aggregate:

Listing 4: show_elem_math.adb

```
with Ada.Text_IO; use Ada.Text_IO;
with Ada.Numerics; use Ada.Numerics;

with Ada.Numerics.Complex_Types;
use Ada.Numerics.Complex_Types;
```

(continues on next page)

```
with Ada.Numerics.Complex_Elementary_Functions;
   use Ada.Numerics.Complex_Elementary_Functions;
   with Ada.Text_IO.Complex_IO;
10
11
   procedure Show_Elem_Math is
12
13
       package C_IO is new
14
         Ada.Text_IO.Complex_IO (Complex_Types);
15
       use C_IO;
16
17
       X, Y : Complex;
18
       R, Th : Float;
19
20
   begin
       X := (2.0, -1.0);
21
       Y := (3.0, 4.0);
22
23
       Put (X);
24
       Put (" * ");
25
       Put (Y);
26
       Put (" is ");
27
       Put (X * Y);
28
       New_Line;
29
       New_Line;
30
31
       R := 3.0;
32
       Th := Pi / 2.0;
33
       X := Compose_From_Polar (R, Th);
34
          Alternatively:
35
          X := R * Exp ((0.0, Th));
36
       -- X := R * e ** Complex'(0.0, Th);
37
38
       Put ("Polar form:
39
            & Float'Image (R) & " * e^{**}(i * "
40
            & Float'Image (Th) & ")");
41
       New_Line;
42
43
       Put ("Modulus
                          of ");
44
       Put (X);
45
       Put (" is ");
46
       Put (Float'Image (abs (X)));
47
       New_Line;
48
49
       Put ("Argument
                          of ");
50
       Put (X);
51
       Put (" is ");
52
       Put (Float'Image (Argument (X)));
53
       New_Line;
54
       New_Line;
55
56
       Put ("Sqrt
                          of ");
57
       Put (X);
58
       Put (" is ");
59
       Put (Sqrt (X));
60
       New_Line;
   end Show_Elem_Math;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Elem_Math (continues on next page)
```

MD5: 24fd48ab69aeac28286e6ec8065899c5

Runtime output

```
( 2.00000E+00,-1.00000E+00) * ( 3.00000E+00, 4.00000E+00) is ( 1.00000E+01, 5. -00000E+00)

Polar form: 3.00000E+00 * e**(i * 1.57080E+00)

Modulus of (-1.31134E-07, 3.00000E+00) is 3.00000E+00

Argument of (-1.31134E-07, 3.00000E+00) is 1.57080E+00

Sqrt of (-1.31134E-07, 3.00000E+00) is ( 1.22474E+00, 1.22474E+00)
```

As we can see from this example, all the common operators, such as \ast and +, are available for complex types. You also have typical operations on complex numbers, such as Argument and Exp. In addition to initializing complex numbers in the cartesian form using aggregates, you can do so from the polar form by calling the Compose From Polar function.

The Ada.Numerics.Complex_Types and Ada.Numerics.Complex_Elementary_Functions packages provide operations for the **Float** type. Similar packages are available for **Long_Float** and **Long_Long_Float** types. In addition, the Ada.Numerics. **Generic_Complex_Types** and Ada.Numerics.**Generic_Complex_Elementary_Functions** packages are generic versions that you can instantiate for custom or pre-defined floating-point types. For example:

```
with Ada.Numerics.Generic_Complex_Types;
with Ada.Numerics.Generic_Complex_Elementary_Functions;
with Ada.Text_IO.Complex_IO;
procedure Show Elem Math is
   package Complex Types is new
     Ada. Numerics. Generic Complex Types (Float);
   use Complex_Types;
  package Elementary_Functions is new
     Ada.Numerics.Generic_Complex_Elementary_Functions
       (Complex_Types);
  use Elementary_Functions;
  package C IO is new Ada. Text IO. Complex IO
     (Complex_Types);
  use C IO;
  X, Y : Complex;
  R, Th: Float;
```

23.4 Vector and Matrix Manipulation

The Ada.Numerics.Real_Arrays package provides support for vectors and matrices. It includes common matrix operations such as inverse, determinant, eigenvalues in addition to simpler operators such as matrix addition and multiplication. You can declare vectors and matrices using the Real Vector and Real Matrix types, respectively.

The following example uses some of the operations from the Ada.Numerics.Real_Arrays package:

Listing 5: show_matrix.adb

```
with Ada.Text_IO; use Ada.Text_IO;
2
   with Ada.Numerics.Real_Arrays;
3
   use Ada.Numerics.Real_Arrays;
4
   procedure Show Matrix is
6
       procedure Put_Vector (V : Real_Vector) is
8
      begin
9
          Put ("
10
          for I in V'Range loop
11
             Put (Float'Image (V (I)) & " ");
12
          end loop;
13
          Put_Line (")");
14
      end Put_Vector;
15
16
      procedure Put_Matrix (M : Real_Matrix) is
17
      begin
18
          for I in M'Range (1) loop
19
             Put ("
                       (");
20
             for J in M'Range (2) loop
21
                Put (Float'Image (M (I, J)) & " ");
22
             end loop;
23
             Put_Line (")");
24
          end loop;
25
      end Put_Matrix;
26
27
                 : Real_Vector := (1.0, 3.0);
28
      ٧2
                 : Real_Vector := (75.0, 11.0);
29
30
                 : Real_Matrix :=
      Μ1
31
                     ((1.0, 5.0, 1.0),
32
                      (2.0, 2.0, 1.0));
33
      M2
                 : Real_Matrix :=
34
                     ((31.0, 11.0, 10.0),
35
                      (34.0, 16.0, 11.0),
36
                      (32.0, 12.0, 10.0),
37
                      (31.0, 13.0, 10.0));
38
      М3
                 : Real_Matrix := ((1.0, 2.0),
39
                                     (2.0, 3.0));
40
   begin
41
      Put_Line ("V1");
42
      Put_Vector (V1);
43
      Put_Line ("V2");
44
45
      Put_Vector (V2);
      Put_Line ("V1 * V2 =");
46
      Put_Line ("
47
                 & Float'Image (V1 * V2));
48
      Put Line ("V1 * V2 =");
49
      Put_Matrix (V1 * V2);
50
      New_Line;
51
52
      Put_Line ("M1");
53
      Put_Matrix (M1);
54
      Put_Line ("M2");
55
      Put_Matrix (M2);
56
      Put_Line ("M2 * Transpose(M1) =");
57
      Put_Matrix (M2 * Transpose (M1));
58
      New Line;
59
60
                                                                           (continues on next page)
```

```
Put_Line ("M3");
61
       Put_Matrix (M3);
62
       Put_Line ("Inverse (M3) =");
63
       Put_Matrix (Inverse (M3));
       Put_Line ("abs Inverse (M3) =");
65
       Put_Matrix (abs Inverse (M3));
66
       Put_Line ("Determinant (M3) =");
67
       Put_Line ("
68
                  & Float'Image (Determinant (M3)));
69
       Put_Line ("Solve (M3, V1) =");
70
       Put_Vector (Solve (M3, V1));
Put_Line ("Eigenvalues (M3) =");
71
72
       Put_Vector (Eigenvalues (M3));
73
       New_Line;
   end Show_Matrix;
```

Code block metadata

```
Project: Courses.Intro_To_Ada.Standard_Library.Show_Matrix
MD5: c9df45a742a42bd47e03fbf2d0282238
```

Runtime output

```
٧1
    ( 1.00000E+00 3.00000E+00 )
٧2
    (7.50000E+01 1.10000E+01)
V1 * V2 =
     1.08000E+02
V1 * V2 =
    (7.50000E+01 1.10000E+01)
    ( 2.25000E+02 3.30000E+01 )
Μ1
    ( 1.00000E+00 5.00000E+00 1.00000E+00 )
    ( 2.00000E+00 2.00000E+00 1.00000E+00 )
M2
    ( 3.10000E+01 1.10000E+01 1.00000E+01 )
                   1.60000E+01 1.10000E+01 )
    ( 3.40000E+01
    ( 3.20000E+01 1.20000E+01 1.00000E+01 )
( 3.10000E+01 1.30000E+01 1.00000E+01 )
M2 * Transpose(M1) =
    ( 9.60000E+01 9.40000E+01 )
    ( 1.25000E+02 1.11000E+02 )
    ( 1.02000E+02  9.80000E+01 )
    ( 1.06000E+02  9.80000E+01 )
М3
    ( 1.00000E+00 2.00000E+00 )
    (2.00000E+00 3.00000E+00)
Inverse (M3) =
    (-3.00000E+00 2.00000E+00)
    ( 2.00000E+00 -1.00000E+00 )
abs Inverse (M3) =
    ( 3.00000E+00 2.00000E+00 )
    ( 2.00000E+00 1.00000E+00 )
Determinant (M3) =
    -1.00000E+00
Solve (M3, V1) =
    ( 3.00000E+00 -1.00000E+00 )
Eigenvalues (M3) =
```

(continues on next page)

```
( 4.23607E+00 -2.36068E-01 )
```

Matrix dimensions are automatically determined from the aggregate used for initialization when you don't specify them. You can, however, also use explicit ranges. For example:

```
M1 : Real_Matrix (1 .. 2, 1 .. 3) := ((1.0, 5.0, 1.0), (2.0, 2.0, 1.0));
```

The Ada.Numerics.Real_Arrays package implements operations for the **Float** type. Similar packages are available for **Long_Float** and **Long_Long_Float** types. In addition, the Ada.Numerics.**Generic_**Real_Arrays package is a generic version that you can instantiate with custom floating-point types. For example, the Real_Arrays package can be defined as follows:

```
package Real_Arrays is new
Ada.Numerics.Generic_Real_Arrays (Float);
```

APPENDICES

24.1 Appendix A: Generic Formal Types

The following tables contain examples of available formal types for generics:

Formal type	Actual type
Incomplete type	Any type
Format: type T;	
Discrete type	Any integer, modular or enumeration type
Format: type T is (<>);	
Range type	Any signed integer type
Format: type T is range <>;	A managed and a
Modular type	Any modular type
Format: type T is mod <>;	Any floating point type
Floating-point type Format: type T is digits <>;	Any floating-point type
Binary fixed-point type	Any binary fixed-point type
Format: type T is delta <>;	Any binary fixed point type
Decimal fixed-point type	Any decimal fixed-point type
<pre>Format: type T is delta <> digits <>;</pre>	, ,,
Definite nonlimited private type	Any nonlimited, definite type
Format: type T is private;	
Nonlimited Private type with discriminant	Any nonlimited type with discriminant
Format: type T (D : DT) is private;	
Access type	Any access type for type T
Format: type A is access T;	
Definite derived type	Any concrete type derived from base type
Format: type T is new B;	Any definite type limited or not
<pre>Limited private type Format: type T is limited private;</pre>	Any definite type, limited or not
Incomplete tagged type	Any concrete, definite, tagged type
Format: type T is tagged;	This concrete, definite, tagged type
Definite tagged private type	Any concrete, definite, tagged type
Format: type T is tagged private;	,
Definite tagged limited private type	Any concrete definite tagged type, limited
Format: type T is tagged limited pri-	or not
vate;	
Definite abstract tagged private type	Any nonlimited, definite tagged type, ab-
Format: type T is abstract tagged	stract or concrete
private;	

continues on next page

Table 1 - continued from previous page

lable 1 - continued from previous page				
Formal type	Actual type			
Definite abstract tagged limited private type Format: type T is abstract tagged limited private;	Any definite tagged type, limited or not, abstract or concrete			
Definite derived tagged type Format: type T is new B with private;	Any concrete tagged type derived from base type B			
Definite abstract derived tagged type Format: type T is abstract new B with private;	Any tagged type derived from base type B abstract or concrete			
Array type Format: type A is array (R) of T;	Any array type with range R containing elements of type T			
<pre>Interface type Format: type T is interface;</pre>	Any interface type T			
<pre>Limited interface type Format: type T is limited interface;</pre>	Any limited interface type T			
Task interface type Format: type T is task interface;	Any task interface type T			
<pre>Synchronized interface type Format: type T is synchronized in- terface;</pre>	Any synchronized interface type T			
<pre>Protected interface type Format: type T is protected inter- face;</pre>	Any protected interface type T			
Derived interface type Format: type T is new B and I with private;	Any type T derived from base type B and interface I			
Derived type with multiple interfaces Format: type T is new B and I1 and I2 with private;	Any type T derived from base type B and interfaces I1 and I2			
Abstract derived interface type Format: type T is abstract new B and I with private;	Any type T derived from abstract base type B and interface I			
Limited derived interface type Format: type T is limited new B and I with private;	Any type T derived from limited base type B and limited interface I			
Abstract limited derived interface type Format: type T is abstract limited new B and I with private;	Any type T derived from abstract limited base type B and limited interface I			
Synchronized interface type Format: type T is synchronized new SI with private;	Any type T derived from synchronized interface SI			
Abstract synchronized interface type Format: type T is abstract synchronized new SI with private;	Any type T derived from synchronized interface SI			

24.1.1 Indefinite version

Many of the examples above can be used for formal indefinite types:

Formal type	Actual type		
Indefinite incomplete type	Any type		
Format: type T (<>);			
Indefinite nonlimited private type	Any nonlimited type indefinite or def-		
<pre>Format: type T (<>) is private;</pre>	inite		
Indefinite limited private type	Any type, limited or not, indefinite or		
<pre>Format: type T (<>) is limited private;</pre>	definite		
Incomplete indefinite tagged private type	Any concrete tagged type, indefinite		
Format: type T (<>) is tagged;	or definite		
Indefinite tagged private type	Any concrete, nonlimited tagged		
Format: type T (<>) is tagged private;	type, indefinite or definite		
Indefinite tagged limited private type	Any concrete tagged type, limited or		
Format: type T (<>) is tagged limited pri-	not, indefinite or definite		
vate;			
Indefinite abstract tagged private type	Any nonlimited tagged type, indefi-		
<pre>Format: type T (<>) is abstract tagged private;</pre>	nite or definite, abstract or concrete		
Indefinite abstract tagged limited private type	Any tagged type, limited or not, in-		
<pre>Format: type T (<>) is abstract tagged limited private;</pre>	definite or definite abstract or concrete		
Indefinite derived tagged type	Any tagged type derived from base		
Format: type T (<>) is new B with private;	type B, indefinite or definite		
Indefinite abstract derived tagged type	Any tagged type derived from base		
<pre>Format: type T (<>) is abstract new B with private;</pre>	type B, indefinite or definite abstract or concrete		

The same examples could also contain discriminants. In this case, (<>) is replaced by a list of discriminants, e.g.: (D: DT).

24.2 Appendix B: Containers

The following table shows all containers available in Ada, including their versions (standard, bounded, unbounded, indefinite):

Cate- gory	Container	Std	Bounded	Un- bounded	Indefi- nite
Vector	Vectors	Υ	Υ		Υ
List	Doubly Linked Lists	Υ	Υ		Υ
Мар	Hashed Maps	Υ	Υ		Υ
Мар	Ordered Maps	Υ	Υ		Υ
Set	Hashed Sets	Υ	Υ		Υ
Set	Ordered Sets	Υ	Υ		Υ
Tree	Multiway Trees	Υ	Υ		Υ
Generic	Holders				Υ
Queue	Synchronized Queue Interfaces	Υ			
Queue	Synchronized Queues		Υ	Υ	
Queue	Priority Queues		Υ	Υ	

1 Note

To get the correct container name, replace the whitespace by _ in the names above. (For example, Hashed Maps becomes Hashed_Maps.)

Introduction to Ada

The following table presents the prefixing applied to the container name that depends on its version. As indicated in the table, the standard version does not have a prefix associated with it.

Version	Naming prefix
Std	
Bounded	Bounded_
Unbounded	Unbounded_
Indefinite	Indefinite_